

**THE CONSOLIDATED  
COMPARTMENT FIRE  
MODEL (CCFM)  
COMPUTER CODE  
APPLICATION CCFM.  
VENTS - PART III:  
CATALOG OF  
ALGORITHMS AND  
SUBROUTINES**

**Leonard Y. Cooper  
Glenn P. Forney  
Editors**

**U.S. DEPARTMENT OF COMMERCE  
National Institute of Standards  
and Technology  
Center for Fire Research  
Gaithersburg, MD 20899**

**Sponsored in part by:  
Naval Sea Systems Command  
Department of the Navy  
Washington, DC 20362**

**U.S. DEPARTMENT OF COMMERCE  
Robert A. Mosbacher, Secretary  
NATIONAL INSTITUTE OF STANDARDS  
AND TECHNOLOGY  
John W. Lyons, Director**

**NIST**

**THE CONSOLIDATED  
COMPARTMENT FIRE  
MODEL (CCFM)  
COMPUTER CODE  
APPLICATION CCFM.  
VENTS - PART III:  
CATALOG OF  
ALGORITHMS AND  
SUBROUTINES**

**Leonard Y. Cooper  
Glenn P. Forney  
Editors**

**U.S. DEPARTMENT OF COMMERCE  
National Institute of Standards  
and Technology  
Center for Fire Research  
Gaithersburg, MD 20899**

**Sponsored in part by:  
Naval Sea Systems Command  
Department of the Navy  
Washington, DC 20362**

**July 1990**



**U.S. DEPARTMENT OF COMMERCE  
Robert A. Mosbacher, Secretary  
NATIONAL INSTITUTE OF STANDARDS  
AND TECHNOLOGY  
John W. Lyons, Director**

## TABLE OF CONTENTS

	<u>Page</u>
TABLE OF CONTENTS . . . . .	iii
ABSTRACT . . . . .	1
1. Background . . . . .	2
2. Physical and Utility Algorithms/Subroutines - Modularity of the Catalog Entries . . . . .	3
3. The Format of the Catalog Entries . . . . .	4
4. Tabulation of the Catalog Entries . . . . .	5
REFERENCES . . . . .	6

### APPENDIX A - Physical Algorithms/Subroutines:

COMWL1 -	Setup for Calculation of the Flow Through Vents in the Wall Segment Common to Two Rooms
DELP -	Calculation of the Absolute Hydrostatic Pressures at a Specified Elevation in Each of Two Adjacent Rooms and the Pressure Difference
FANRES -	Calculate the Flow Through a Fan/Duct Forced Ventilation System
FLOGO2 -	Deposition of Mass, Enthalpy, Oxygen, and Other Product-of-Combustion Flows Passing Between Two Rooms Through a Vertical, Constant-Width Vent or Through a Forced Ventilation System
PLUGO -	Calculation of Mass, Enthalpy, Oxygen and Other Product of Combustion Flows to Upper and Lower Layers Due to a Fire-Driven Plume
VENTF -	Calculate Flow Through a Simple Fan/Resistance Forced Ventilation System Between Two Rooms
VENTHP -	Calculation of the Flow of Mass, Enthalpy, Oxygen, and Other Products of Combustion Through a Vertical, Constant-Width Vent in a Wall Segment Common to Two Rooms

### APPENDIX B - Utility Algorithms/Subroutines:

HSORT -	Arranges an Array in Ascending Order Using a Heap Sorting Algorithm
---------	---

INTERP - Interpolates a Table of Numbers  
RMVDUP - Remove Duplicate Entries From a List of Data

**The Consolidated Compartment Fire Model (CCFM) Computer Code  
Application CCFM.VENTS - Part III: Catalog of Algorithms and Subroutines**

**Leonard Y. Cooper and Glenn P. Forney, Editors**

**ABSTRACT**

A project was carried out at The National Institute of Standards and Technology (NIST) to study the feasibility of developing a new-generation, multi-room, compartment fire model computer code, called the Consolidated Compartment Fire Model (CCFM) computer code. The idea was that such a code would consolidate past progress in zone-type compartment fire modeling, and allow readily for integration of future advances with the greatest possible flexibility. Desired features of the CCFM would include: comprehensive documentation, user-friendliness, significant modularity, numerical robustness, and versatility in the sense that the code would provide a capability of analyzing a particular compartment fire problem by using any one of a range of physical-phenomena-modeling sophistication, from the most basic to the most comprehensive. The project led to the development of a prototype multi-room CCFM product called CCFM.VENTS. CCFM.VENTS involves a model formulation and code structure that allows for the required future CCFM growth flexibility. It has a relatively sophisticated and very general room-to-room forced and unforced vent flow capability. Finally, the CCFM.VENTS code uses the simplest possible, point-source-plume, smoke-filling fire physics in the rooms-of-fire-origin and a very simple heat transfer calculation there and in other spaces.

This is Part III of a four-part report which documents CCFM.VENTS. It is a catalog of all the modular algorithms and associated computer subroutines used to simulate the physical phenomena in CCFM.VENTS. The algorithms/subroutines are divided into two categories, those which describe the calculations for simulating the basic physical phenomena and those which are used to support these calculations. The latter catalog entries are called "utility" algorithms/subroutines.

Each physical algorithm entry of the catalog includes a brief description of the phenomenon simulated, a concise but complete presentation of the calculation procedure used, identification of all input and output parameters, and a listing of the subroutine. The catalog entries have been developed and are presented as modular, stand-alone products. The stand-alone design feature allows the catalog entries to be used both in CCFM and in any other modular, zone-type, compartment fire model computer code.

The other three parts of this report are: Part I: Physical Basis; Part II: Software Reference Guide; and Part IV: User Reference Guide.

**Keywords:** building fires; compartment fires; computer models; fire models; mathematical models; vents; zone models.

## 1. Background

A project was carried out at The National Institute of Standards and Technology (NIST) to study the feasibility of developing a new-generation, multi-room, compartment fire model computer code, called the Consolidated Compartment Fire Model (CCFM) computer code. The idea was that such a code would consolidate past progress in zone-type compartment fire modeling, and allow readily for integration of future advances with the greatest possible flexibility. Desired features of the CCFM would include: comprehensive documentation, user-friendliness, significant modularity, numerical robustness, and versatility in the sense that the code would provide a capability of analyzing a particular compartment fire problem by using any one of a range of physical-phenomena-modeling sophistication, from the most basic to the most comprehensive. The project led to the development of a prototype multi-room CCFM product called CCFM.VENTS. CCFM.VENTS involves a model formulation and code structure that allows for the required future CCFM growth flexibility.

This is the third of a four-part report which documents all aspects of CCFM.VENTS.

In Part I [1]<sup>1</sup>, introductory remarks discuss the generic features of CCFM, the CCFM development process, and the specific features of CCFM.VENTS. The main objective of Part I is to present a comprehensive description and technical basis of the governing equations used in CCFM.VENTS.

Part II [2] presents the generic and CCFM.VENTS-specific features of the framework of CCFM software. This includes features of both program and data structures; numerical considerations used to treat the solution to the CCFM equation set; and a presentation of methods of how one would use the CCFM structure to generate stages of the computer model beyond CCFM.VENTS.

Part IV [3] is the CCFM.VENTS user's reference guide.

The present Part III is a catalog of all the modular algorithms and associated modular computer subroutines used to simulate the physical phenomena in CCFM.VENTS. The purpose of this catalog is to allow the interested user to learn with relative ease how the code carries out any particular aspect of the overall CCFM.VENTS simulation. The catalog algorithm/subroutine entries are also available for use by people who may be interested in developing for their own particular needs a compartment fire model computer code other than the CCFM.

---

<sup>1</sup>Numbers in brackets always refer to the references at the end of this report.

## 2. Physical and Utility Algorithms/Subroutines - Modularity of the Catalog Entries

The CCFM.VENTS algorithms/subroutines presented here include two categories of catalog entries. The first includes catalog entries which describe the modeling of physical phenomena. The second category of catalog entry, those which do not involve the modeling of physical phenomena but which do support the calculations of the physical algorithms, are referred to as "utility" algorithms. Documentation of utility algorithms is limited to a listing of their appropriately well-commented subroutines.

Each physical algorithm/subroutine entry of the catalog includes a brief description of the phenomenon simulated, a concise presentation of the calculation procedure used, identification of all input and output parameters, and an actual listing of the computer subroutine. The catalog provides the bridge between the governing equations of CCFM.VENTS and their technical basis, as presented in detail in Part I, and the implementation of these mathematical submodels into the software of CCFM.

The entries of the catalog have been developed and are presented as stand-alone products. The stand-alone design feature allows the catalog entries to be used readily both in CCFM and in any other modular, zone-type, compartment fire model computer code.

Implementing the CCFM concept would lead to a process of development which would consolidate past progress in zone-type compartment fire modeling, and which would allow readily for integration of future advances with the greatest possible flexibility. It is envisaged that the modularity and stand-alone features of the catalog entries would help to provide this flexibility. Indeed, these features would be critical in insuring the required orderly and efficient future growth of and improvement to existing stages of the CCFM. For example, while some of the algorithms/subroutines presented here and used in CCFM.VENTS involve very simple mathematical models of the physical phenomena simulated, there is clear opportunity for improvement by the use of more sophisticated modeling ideas. The modular design of both the original and the new catalog entry and subroutine software would then allow easily for the replacement of the old product with the new, or for the option of using the product of either greater or less sophistication, say, for comparing the supposed benefit of using one or the other. Examples of this are the presently used plume algorithm, PLUGO, which implements a very simple point-source fire plume model, and the presently used vent-flow distribution model, FLOGO2, which extends and implements a very simple, well-tested rule for distributing vent flows and calculating heat transfer to the upper and lower layer and to the surfaces of the receiving room. More sophisticated alternatives to both of these algorithms are now under development.

The use of well-documented modular algorithms/subroutines in CCFM.VENTS would also be critical for successful future CCFM team development efforts. Indeed, by constructing CCFM products with modular components having well-defined inputs and outputs, a development-team effort that involves significant inter-institution participation would be possible.

### 3. The Format of the Catalog Entries

The development, technology transfer, and use of CCFM is enhanced by establishing and implementing guidelines for a uniform format of algorithm/subroutine documentation. In this regard a prototype format has been developed and used in all CCFM.VENTS algorithm/subroutine catalog entries presented here. It is hoped that this first CCFM catalog will provide a useful test of the proposed format, and that this will lead to effective, enriched, future editions of the CCFM catalog.

The adopted algorithm format includes the following elements:

<b>TITLE</b>	-	Should indicate the main purpose of the algorithm/subroutine.
<b>DESCRIPTION</b>	-	General description of the algorithm.
<b>OUTPUT</b>	-	List of output variables, including definitions and units.
<b>INPUT</b>	-	List of input variables, including definitions and units.
<b>CALCULATIONS</b>	-	Concise description of the rules for obtaining the output variables from the input variables. This would include or refer explicitly to all equations required in the calculation. If other algorithms/subroutines are required, then these should be readily available and referenced.
<b>SUBROUTINES USED</b>	-	Listing of or explicit reference to each algorithm/subroutine used to carry out the calculations.
<b>REFERENCES</b>	-	A list of references.
<b>SUBROUTINE VARIABLES</b>	-	A cross-reference of all nomenclature (including units) introduced in the above sections to the nomenclature used in the FORTRAN computer subroutine.
<b>PREPARED BY</b>	-	Names of those who prepared the algorithm/subroutine and date of preparation.
<b>SUBROUTINE</b>	-	Listing of the computer subroutine. This would be well-commented and would include a summary of the purpose of the subroutine and definitions (including units) of its input and output variables.



All algorithm/subroutines used in CCFM.VENTS are presented here in Appendices A and B. Appendix A is a compilation of the physical algorithm catalog entries. All of these are presented in the above format. Appendix B is a compilation of the supporting utility catalog entries.

#### 4. Tabulation of the Catalog Entries

The entries of each appendix are compiled in alphabetical order according to algorithm/subroutine name, with independent pagination for each entry. This presentation highlights the modular and stand-alone characteristic of the entries and the flexibility of the overall CCFM. Thus, at any time, new or improved entries can be sorted into or used to replace existing entries, thereby leading to an expanded and improved CCFM catalog. Also, as appropriate, the catalog entries can be lifted and used elsewhere to construct and/or enhance two-layer zone-type compartment fire models other than CCFM.

Listings of the catalog entries in both Appendices are presented in the Table of Contents.

## REFERENCES

- [1] Cooper, L.Y. and Forney, G.P., The Consolidated Compartment Fire Model (CCFM) Computer Code Application CCFM.VENTS - Part I: Physical Basis, NISTIR 90-4342, National Institute of Standards and Technology, Gaithersburg MD, 1990.
- [2] Forney, G.P., and Cooper, L.Y., The Consolidated Compartment Fire Model (CCFM) Computer Code Application CCFM.VENTS - Part II: Software Reference Guide, NISTIR 90-4343, National Institute of Standards and Technology, Gaithersburg MD, 1990.
- [3] Forney, G.P., Cooper, L.Y., and Moss, W.F., The Consolidated Compartment Fire Model (CCFM) Computer Code Application CCFM.VENTS - Part IV: User Reference Guide, NISTIR 90-4345, National Institute of Standards and Technology, Gaithersburg MD, 1990.

## **APPENDIX A - Physical Algorithms/Subroutines**



COMWL1 -     Setup for Calculation of the Flow Through Natural Vents in the  
Wall Segment Common to Two Rooms

DESCRIPTION

Consider an instant of time during the simulation of a compartment fire environment. This algorithm will be useful in setting up a procedure for calculating the instantaneous flow rate through one or more natural vents of arbitrary size and elevation located in the wall segment common to any two particular rooms of the compartment [1].

Designate the two rooms as rooms 1 and 2 and refer to the Figure 1. In general, within the elevation interval of the common wall segment the room 1-to-room 2 pressure difference can be described by at most three piecewise-continuous, linear functions of elevation. These will have endpoints at the bottom and top of the common wall segment and at the elevations of the interfaces which separate the upper and lower layers of the two rooms (provided these interfaces also lie in the elevation interval of the wall segment). It is possible for each of the pieces of the piecewise linear functions to change sign somewhere within its operable elevation range. The flow calculation for an arbitrary vent in the wall segment can be carried out conveniently once the elevations (no more than four) and corresponding room 1 and 2 absolute pressures and pressure differences at the endpoints of these pieces have been identified explicitly. (The absolute pressures will be important in cross-vent flow calculations only in relatively unusual circumstances where compressibility effects are significant.) Also required for the calculation are the elevations (no more than three in addition to the above) where the pieces change sign (i.e., elevations where the room 1-to-room 2 pressure difference is zero).

Pressure distributions in each room and possible cross-wall pressure-difference distributions are depicted in Figure 2.

This algorithm calculates the total number of these elevations (no more than seven unique elevations),  $N_{ELEV}$ ; the values of the elevations relative to a datum elevation (presented in an array, increasing uniformly with elevation),  $y_N$ ; and corresponding arrays of absolute pressure in rooms 1 and 2,  $p_{1,N}$  and  $p_{2,N}$ , respectively, the room 1-to-room 2 pressure differences,  $\Delta p_{1,2,N}$ .

OUTPUT

$N_{WELEV}$

Number of unique elevations,  $\leq 7$ , which identify 1) endpoints of the continuous, linear pieces of the room 1-to-room 2 pressure difference vs elevation and/or 2) elevations where the room 1-to-room 2 pressure difference is zero, where all these elevations are at or between the vertical extremes of the wall segment common to rooms 1 and 2.

$P_{1,N}$  [ $P_{2,N}$ ];  $N = 1$  to  $N_{WELEV}$

Absolute hydrostatic pressure in room 1 [room 2] at elevation  $y_N$ . [ $\text{Pa} = \text{kg}/(\text{m} \cdot \text{s}^2)$ ]

$y_N$ ;  $N = 1$  to  $N_{WELEV}$ ; and  $y_N < y_{N+1}$

Values of the elevations identified under the output variable  $N_{WELEV}$  above the datum elevation. [m]

$\Delta p_{1,2,N}$ ;  $N = 1$  to  $N_{WELEV}$

Pressure in room 1 - pressure in room 2 at elevation  $y_N$ . If  $\Delta p_{1,2,N} > 0$ ,  $= 0$ , or  $< 0$  then flow through a vent at this elevation will be from room 1 to room 2, zero, or from room 2 to room 1, respectively. [ $\text{Pa} = \text{kg}/(\text{m} \cdot \text{s}^2)$ ]

#### INPUT

$P_{\text{DATUM}}$

Datum absolute pressure. [ $\text{Pa} = \text{kg}/(\text{m} \cdot \text{s}^2)$ ]

$P_{\text{FLOOR},i}$ ;  $i = 1$  or  $2$

Hydrostatic pressure at floor in room 1 [room 2] above the datum pressure. [ $\text{Pa} = \text{kg}/(\text{m} \cdot \text{s}^2)$ ]

$y_{\text{FLOOR},i}$  [ $y_{\text{CEIL},i}$ ] ( $y_{\text{LAYER},i}$ );  $i = 1$  or  $2$

Elevation of the floor [ceiling] (layer interface) in room  $i$  above the datum elevation. [m]

$\rho_{L,i}$  [ $\rho_{U,i}$ ];  $i = 1$  or  $2$

Density of lower [upper] layer in room  $i$ . [ $\text{kg}/\text{m}^3$ ]

#### CALCULATIONS

1. Set  $y_{\text{MIN}}$  to maximum of two floor elevations and  $y_{\text{MAX}}$  to minimum of two ceiling elevations.

2. Store into the elevation list,  $y_N$ , the floor, ceiling and layer elevations of the two rooms that are between  $y_{MIN}$  and  $y_{MAX}$ . Use the algorithm/subroutine DELP to calculate the absolute pressure in each room and the difference in room pressures at each of these elevations and store the results in the arrays  $P_{1,N}$ ,  $P_{2,N}$ , and  $\Delta P_{1,2,N}$ .
3. Sort the elevation/pressure/pressure-difference list according to elevation and remove any duplicate entries.
4. Identify and add any additional neutral planes (elevations where pressure difference is zero) to the elevation/pressure/pressure-difference list.
5. Re-sort list, remove duplicate entries, and set  $N_{WELEV}$  to number of elevation entries in list. Return  $N_{WELEV}$ ,  $P_{1,N}$ ,  $P_{2,N}$ ,  $y_N$ ,  $\Delta P_{1,2,N}$  to calling routine.

#### SUBROUTINES USED

DELP [2]

HSORT [3,4]

RMVDUP [5]

#### REFERENCES

- [1] Cooper, L.Y. and Forney, G.P., Section 3.3 of Consolidated Compartment Fire Model (CCFM) Computer Code Application CCFM.VENTS - Part I: Physical Basis, NISTIR 90-4342, National Institute of Standards and Technology, Gaithersburg MD, 1990.
- [2] Cooper, L.Y. and Forney, G.P., DELP - Calculation of the Absolute Hydrostatic Pressures at a Specified Elevation in Each of Two Adjacent Rooms and the Presssure Difference, Appendix A of Consolidated Compartment Fire Model (CCFM) Computer Code Application CCFM.VENTS - Part III: Algorithms and Subroutines, Cooper, L.Y., and Forney, G.P., Eds., NISTIR 90-4344, National Institute of Standards and Technology, Gaithersburg MD, 1990.
- [3] Forney, G.P., Algorithm HSORT - Arranges an Array in Ascending Order Using a Heap Sorting Algorithm, Appendix B of Consolidated Compartment Fire Model (CCFM) Computer Code Application CCFM.VENTS - Part III: Algorithms and Subroutines, Cooper, L.Y., and Forney, G.P., Eds., NISTIR 90-4344, National Institute of Standards and Technology, Gaithersburg MD, 1990.
- [4] Press, W.H. et al, Numerical Recipes - the Art of Scientific Computing, Cambridge Press, pp. 231-232 (1986).

- [5] Forney, G.P., Algorithm RMVDUP - Remove Duplicate Entries From a List of Data, Appendix B of Consolidated Compartment Fire Model (CCFM) Computer Code Application CCFM.VENTS - Part III: Algorithms and Subroutines, Cooper, L.Y., and Forney, G.P., Eds., NISTIR 90-4344, National Institute of Standards and Technology, Gaithersburg MD, 1990.

#### SUBROUTINE VARIABLES

All nomenclature in the subroutine is identical to the nomenclature used above except for:

$N_{WELEV}$	-	NELEV
$P_{FLOOR,1}, P_{FLOOR,2}$	-	PLOR(1), PLOR(2) [Pa = kg/(m·s <sup>2</sup> )]
$P_{1,N} [P_{2,N}]$	-	PAB1(N), PAB2(N) [Pa = kg/(m·s <sup>2</sup> )]
$Y_N$	-	YELEV(N) [m]
$Y_{CEIL,1}, Y_{CEIL,2}$	-	YCEIL(1), YCEIL(2) [m]
$Y_{FLOOR,1}, Y_{FLOOR,2}$	-	YFLOR(1), YFLOR(2) [m]
$Y_{LAYER,1}, Y_{LAYER,2}$	-	YLAY(1), YLAY(2) [m]
$\Delta P_{1,2,N}$	-	DP1M2(N) [Pa = kg/(m·s <sup>2</sup> )]
$\rho_{L,1}, \rho_{L,2}$	-	DENL(1), DENL(2) [kg/m <sup>3</sup> ]
$\rho_{U,1}, \rho_{U,2}$	-	DENU(1), DENU(2) [kg/m <sup>3</sup> ]

#### PREPARED BY

Leonard Y. Cooper and Glenn P. Forney  
January 1988






Figure 1. The fire-generated environment in two adjacent rooms.




Figure 2. Possible distributions of  $\Delta p_{1,2}$  for a common-wall segment.

## SUBROUTINE

```
      SUBROUTINE COMWL1(
      I YFLOR,YCEIL,YLAY,DENU,DENL,PFLOR,
      I PDATM,
      O YELEV, DP1M2, PAB1, PAB2, NELEV
      + )
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
C*BEG
C
C***   COMWL1  GENERIC - SETUP FOR CALCULATION OF THE FLOW THROUGH VENTS
C           IN THE WALL SEGMENT COMMON TO TWO ROOMS.  THIS ROUTINE
C           CALCULATES ROOM PRESSURES AND ROOM-TO-ROOM PRESSURE
C           DIFFERENCES AT CERTAIN ELEVATIONS ALONG COMMON WALLS OF
C           ADJACENT ROOMS.  THESE ELEVATIONS ARE:
C
C           1.  YMIN - MAXIMUM OF TWO ROOMS FLOOR ELEVATIONS
C           2.  YMAX - MINIMUM OF TWO ROOMS CEILING ELEVATIONS
C           3.  LAYER ELEVATIONS BETWEEN YMIN AND YMAX
C           4.  NEUTRAL PLANES ELEVATIONS BETWEEN YMIN AND YMAX
C
C      THERE CAN BE UP TO 7 ELEVATIONS OF INTEREST (1 FLOOR, 1 CEILING,
C      2 LAYER AND 3 NEUTRAL PLANE ELEVATIONS).
C
C***   SUBROUTINE VARIABLES
C      INPUT
C      -----
C      YFLOR - ELEVATION OF FLOOR IN EACH ROOM [M]
C      YCEIL - ELEVATION OF CEILING IN EACH ROOM [M]
C      YLAY  - ELEVATION OF LAYER IN EACH ROOM [M]
C      DENU  - UPPER LAYER DENSITY IN EACH ROOM [KG/M**3]
C      DENL  - LOWER LAYER DENSITY IN EACH ROOM [KG/M**3]
C      PFLOR - PRESSURE AT FLOOR OF EACH ROOM [KG/(M*S**2) - PA]
C      PDATM - ABSOLUTE PRESSURE AT REFERENCE ELEVATION [PA]
C
C      OUTPUT
C      -----
C      NELEV - NUMBER OF ELEVATIONS (DESCRIBED IN 1 --> 4 ABOVE)
C      YELEV - ELEVATIONS OF ELEVATIONS (DESCRIBED IN 1 --> 4 ABOVE) [M]
C      DP1M2 - PRESSURE DIFFERENCES AT ELEVATIONS IN YELEV [PA]
C      PAB1  - ARRAY OF ABSOLUTE PRESSURES AT ELEVATIONS OF INTEREST [PA]
C      PAB2  - ARRAY OF ABSOLUTE PRESSURES AT ELEVATIONS OF INTEREST [PA]
C
C***   ROUTINES CALLED:
C           DELP,HSORT,RMVDUP
C
C*END

      DIMENSION PAB1(*), PAB2(*)
      DIMENSION YFLOR(*), YCEIL(*), YLAY(*), DENU(*), DENL(*)
      DIMENSION PFLOR(*), YELEV(*), DP1M2(*)
C
```

```

YMIN = MAX(YFLOR(1),YFLOR(2))
YMAX = MIN(YCEIL(1),YCEIL(2))
YELEV(1) = YFLOR(1)
YELEV(2) = YFLOR(2)
YELEV(3) = YLAY(1)
YELEV(4) = YLAY(2)
YELEV(5) = YCEIL(1)
YELEV(6) = YCEIL(2)

C
NELEV = 0
DO 10 I = 1, 6

C
C*** COPY ONLY THOSE VALUES OF Y(I) BETWEEN YMIN AND YMAX
C
      IF(YMIN.LE.YELEV(I).AND.YELEV(I).LE.YMAX)THEN
        NELEV = NELEV + 1
        YELEV(NELEV) = YELEV(I)
      CALL DELP(
I        YELEV(NELEV),YFLOR,YLAY,DENL,DENU,PFLOR,PDATM,
O        DP1M2(NELEV),PAB1(NELEV),PAB2(NELEV))
      ENDIF
10 CONTINUE

C
C*** SORT THE LIST OF HEIGHTS AND REMOVE ANY DUPLICATES THAT ARE FOUND
C
      CALL HSORT(YELEV,DP1M2,PAB1,PAB2,NELEV)
      CALL RMVDUP(YELEV,DP1M2,PAB1,PAB2,NELEV)
      N = NELEV
      DO 20 I = 2, NELEV

C
C*** LOOK FOR NEUTRAL PLANES.
C
      IF(DP1M2(I)*DP1M2(I-1).LT.0.)THEN

C
C*** DP1M2 HAS A CHANGE OF SIGN, SO INTERPOLATE TO FIND THE
C    HEIGHT WHERE DP1M2 IS ZERO.
C
C
C    SOLVE FOR YINTRP
C
      YINTRP - YELEV(I-1)          0. - DP1M2(I-1)
      ----- - -----
      YELEV(I) - YELEV(I-1)      DP1M2(I) - DP1M2(I-1)

C
      Y2 = YELEV(I)
      Y1 = YELEV(I-1)
      P2 = DP1M2(I)
      P1 = DP1M2(I-1)
      YINTRP = (Y2*P1-Y1*P2)/(P1-P2)

C
C*** CALL DELP TO CALCULATE ABSOLUTE PRESSURES AT NEUTRAL PLANE
C

```

```

        CALL DELP(
I          YINTRP, YFLOR, YLAY, DENL, DENU, PFLOR, PDATM,
O          DUMMY, PAB1(N+1), PAB2(N+1))
        YELEV(N+1) = YINTRP
        DP1M2(N+1) = 0.0D0
        N = N + 1
    ENDIF
20 CONTINUE
C
C*** RE-SORT LIST AND REMOVE DUPLICATES
C
    IF(NELEV.NE.N)THEN
        NELEV = N
        CALL HSORT(YELEV, DP1M2, PAB1, PAB2, NELEV)
        CALL RMVDUP(YELEV, DP1M2, PAB1, PAB2, NELEV)
    ENDIF
    RETURN
    END

```



**DELP - Calculation of the Absolute Hydrostatic Pressures at a Specified Elevation in Each of Two Adjacent Rooms and the Pressure Difference**

DESCRIPTION

Consider an instant of time during the simulation of a compartment fire environment. This algorithm calculates the absolute hydrostatic pressures at a specified elevation in each of two adjacent rooms of the compartment and the difference of these pressures [1]. These results are required in order to characterize the instantaneous rate of flow across a vent in a wall segment common to the two rooms.

Designate the two rooms as rooms 1 and 2. Using the pressure algorithm of [2] this algorithm calculates the absolute hydrostatic pressures in the rooms,  $p_1$  and  $p_2$ , respectively, at a specified elevation and then their difference,  $\Delta p_{1,2} = p_1 - p_2$ . Depending on whether  $\Delta p_{1,2}$  is positive or negative, the flow through a vent near the elevation in question will be from room 1 to room 2, or from room 2 to room 1, respectively.

OUTPUT

$p_1$  [ $p_2$ ]

Absolute hydrostatic pressure in room 1 [room 2] at elevation  $y$  [ $\text{Pa} = \text{kg}/(\text{m} \cdot \text{s}^2)$ ].

$\Delta p_{1,2}$

$p_1 - p_2$  [ $\text{Pa} = \text{kg}/(\text{m} \cdot \text{s}^2)$ ].

INPUT

$P_{\text{DATUM}}$

Datum absolute pressure [ $\text{Pa} = \text{kg}/(\text{m} \cdot \text{s}^2)$ ].

$P_{\text{FLOOR},i}$ ;  $i = 1$  or  $2$

Pressure at floor in room 1 [room 2] above datum pressure [ $\text{Pa} = \text{kg}/(\text{m} \cdot \text{s}^2)$ ].

y

Elevation above the datum elevation where  $\Delta p_{1,2}$  is to be calculated [m].

$y_{FLOOR,i}$  [ $y_{LAYER,i}$ ];  $i = 1$  or  $2$

Elevation of the floor [layer interface] in room  $i$  above the datum elevation [m].

$\rho_{L,i}$  [ $\rho_{U,i}$ ];  $i = 1$  or  $2$

Density of lower [upper] layer in room  $i$  [ $\text{kg/m}^3$ ].

### CALCULATIONS

For rooms  $i = 1, 2$  calculate the pressure and then the pressure difference at height  $y$  above the floor.

$$\delta p_i(y) = \begin{cases} -\rho_{L,i}g(y - y_{FLOOR,i}); & y_{FLOOR,i} \leq y \leq y_{LAYER,i} \\ -\rho_{L,i}g(y_{LAYER,i} - y_{FLOOR,i}) - \rho_{U,i}g(y - y_{LAYER,i}); & y_{LAYER,i} < y \end{cases} \quad (1)$$

$$p_i(y) = p_{FLOOR,i} + \delta p_i(y) + p_{DATUM} \quad (2)$$

$$\Delta p_{1,2} = p_1(y) - p_2(y) = (p_{FLOOR,1} - p_{FLOOR,2}) + [\delta p_1(y) - \delta p_2(y)] \quad (3)$$

where

$$g = 9.8 \text{ m/s}^2$$

is the acceleration of gravity.

### REFERENCES

- [1] Cooper, L.Y. and Forney, G.P., Section 3.3 of Consolidated Compartment Fire Model (CCFM) Computer Code Application CCFM.VENTS - Part I: Physical Basis, NISTIR 90-4342, National Institute of Standards and Technology, Gaithersburg MD, 1990.
- [2] Tanaka, T., A Model of Multiroom Fire Spread, NBSIR 83-2718, U.S. National Bureau of Standards, Gaithersburg, MD, 1978.



#### SUBROUTINE VARIABLES

$P_{\text{DATUM}}$	-	$\text{PDATM} \text{ [Pa} = \text{kg}/(\text{m} \cdot \text{s}^2)\text{]}$
$P_1, P_2$	-	$\text{PAB1, PAB2} \text{ [Pa} = \text{kg}/(\text{m} \cdot \text{s}^2)\text{]}$
$Y$	-	$Y \text{ [m]}$
$Y_{\text{FLOOR},1}, Y_{\text{FLOOR},2}$	-	$\text{YFLOR}(1), \text{YFLOR}(2) \text{ [m]}$
$Y_{\text{LAYER},1}, Y_{\text{LAYER},2}$	-	$\text{YLAY}(1), \text{YLAY}(2) \text{ [m]}$
$\Delta P_{1,2}$	-	$\text{DP} \text{ [Pa} = \text{kg}/(\text{m} \cdot \text{s}^2)\text{]}$
$\delta p_1, \delta p_2$	-	$\text{PROOM}(1), \text{PROOM}(2) \text{ [Pa} = \text{kg}/(\text{m} \cdot \text{s}^2)\text{]}$
$\rho_{\text{L},1}, \rho_{\text{L},2}$	-	$\text{DENL}(1), \text{DENL}(2) \text{ [kg}/\text{m}^3\text{]}$
$\rho_{\text{U},1}, \rho_{\text{U},2}$	-	$\text{DENU}(1), \text{DENU}(2) \text{ [kg}/\text{m}^3\text{]}$

#### PREPARED BY

Leonard Y. Cooper and Glenn P. Forney  
March 1988

## SUBROUTINE

```
      SUBROUTINE DELP(
        I Y,YFLOR,YLAY,DENL,DENU,PFLOR,PDATM,
        O DP, PAB1, PAB2)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
C*BEG
C***  DELP  GENERIC - CALCULATION OF THE ABSOLUTE HYDROSTATIC
C          PRESSURES AT A SPECIFIED ELEVATION IN EACH OF TWO ADJACENT
C          ROOMS AND THE PRESSURE DIFFERENCE.  THE BASIC CALCULATION
C          INVOLVES A DETERMINATION AND DIFFERENCING OF HYDROSTATIC
C          PRESSURES ABOVE A SPECIFIED DATUM PRESSURE.
C
C***  SUBROUTINE ARGUMENTS
C
C  INPUT
C  -----
C  Y      - HEIGHT ABOVE DATUM ELEVATION WHERE PRESSURE DIFFERENCE IS
C           TO BE CALCULATED [M]
C  YFLOR  - HEIGHT OF FLOOR IN EACH ROOM ABOVE DATUM ELEVATION
C           [M]
C  YLAY   - HEIGHT OF LAYER IN EACH ROOM ABOVE DATUM ELEVATION [M]
C  DENL   - LOWER LAYER DENSITY IN EACH ROOM [KG/M**3]
C  DENU   - UPPER LAYER DENSITY IN EACH ROOM [KG/M**3]
C  PFLOR  - PRESSURE AT BASE OF EACH ROOM ABOVE DATUM PRESSURE
C           [KG/(M*S**2) - PASCAL]
C  PDATM  - ABSOLUTE DATUM PRESSURE [KG/(M*S**2) - PASCAL]
C
C  OUTPUT
C  -----
C  DP     - CHANGE IN PRESSURE BETWEEN TWO ROOMS [KG/(M*S**2) - PASCAL]
C  PAB1   - ABSOLUTE PRESSURE AT ELEVATION Y IN ROOM 1 [KG/(M*S**2) -
C           PASCAL]
C  PAB2   - ABSOLUTE PRESSURE AT ELEVATION Y IN ROOM 2 [KG/(M*S**2) -
C           PASCAL]
C*END

      DIMENSION PROOM(2)
      DIMENSION DENL(*), DENU(*), YFLOR(*), YLAY(*), PFLOR(*)
      PARAMETER (G=9.80D0)
      DP = 0.0D0
      DO 10 I = 1, 2
        PROOM(I) = 0.0D0
        IF(YFLOR(I).LE.Y.AND.Y.LE.YLAY(I))THEN
C
C***  THE HEIGHT, Y, IS IN THE LOWER LAYER
C
          PROOM(I) = - (Y - YFLOR(I))*DENL(I)*G
C
          ELSE IF(Y.GT.YLAY(I))THEN
C
C***  THE HEIGHT, Y, IS IN THE UPPER LAYER
```

```

C          PROOM(I) = - (YLAY(I)-YFLOR(I))*DENL(I)*G
+          - (Y-YLAY(I))*DENU(I)*G
          ENDIF
10 CONTINUE
C
C*** CHANGE IN PRESSURE IS DIFFERENCE IN PRESSURES IN TWO ROOMS
C
      DP1 = PFLOR(1) + PROOM(1)
      DP2 = PFLOR(2) + PROOM(2)
      PAB1 = PDATM + DP1
      PAB2 = PDATM + DP2
C      IF(ABS(DP1-DP2).LT.MAX(1.0D-3,1.0D-3*(ABS(DP1)+ABS(DP2)))) THEN
C          DP = 0.0D0
C      ELSE
C          DP = DP1-DP2
C      ENDIF
      RETURN
      END

```



## FANRES - Calculate the Flow Through a Fan/Duct Forced Ventilation System

### DESCRIPTION

FANRES is a model for predicting flow through a simple fan/duct forced ventilation system connecting any two rooms of a multi-room facility, or connecting a room and the outside environment [1]. It is based on an extension to the ideas presented in [2]. The FANRES system consists of a fan and duct with a single inlet and outlet. The model predicts flow through the system under arbitrary end-point conditions. For example, if an appropriate fan curve is available, FANRES can provide an estimate of flow rate even when the room environments at the end-points lead to flow through the system in the direction opposite to normal fan operation.

Heat transfer is neglected and the fan/duct system is assumed to be adiabatic.

Refer to Figure 1. Designate the two rooms of interest, or the room and the outside environment as rooms 1 and 2. Define room 1 [room 2] as the room connected to the system inlet [outlet]. Here, the system inlet [outlet] is the end-point of the fan/duct flow system which corresponds to the flow inlet [outlet] under normal fan operating conditions.

Characterize the elevations of the inlet and outlet ducts relative to a datum elevation by the elevation of their midpoints,  $y_{\text{SYSTEM},1}$  and  $y_{\text{SYSTEM},2}$ , respectively. Designate also the hydrostatic pressures in the rooms at these elevations, relative to a datum pressure, as  $\delta p_{\text{SYSTEM},1}$  and  $\delta p_{\text{SYSTEM},2}$ , respectively. Then, it is assumed that the system elevation and pressure differences,  $\Delta y_{\text{SYSTEM}}$  and  $\Delta p_{\text{SYSTEM}}$ , respectively, are specified, where

$$\Delta y_{\text{SYSTEM}} = y_{\text{SYSTEM},1} - y_{\text{SYSTEM},2} \quad (1)$$

$$\Delta p_{\text{SYSTEM}} = \delta p_{\text{SYSTEM},1} - \delta p_{\text{SYSTEM},2} \quad (2)$$

The FANRES algorithm requires specified characteristics of the fan/duct system. These include the fan operating curve and the effective flow resistance of the ducting.

### The Fan Operating Curve

The fan curve provides the volume flow rate through the fan,  $\dot{V}_F$  as a function of cross-fan pressure difference,  $\Delta p_F$ , i.e.,

$$\dot{V}_F = f(\Delta p_F) \quad (3)$$

where

$$\Delta p_F = \delta p_{F,2} - \delta p_{F,1} \quad (4)$$

and  $\delta p_{F,1}$  and  $\delta p_{F,2}$  are the pressures that would be measured at the fan inlet and outlet sections, respectively. A sketch of a generic fan operating curve is presented in Figure 2. As indicated in the figure, a fan curve used in FANRES must be such that  $\Delta p_F$  is monotonically decreasing with increasing  $\dot{V}_F$ . In terms of the fan curve, forward [backward] flow, from room 1 to room 2 [room 2 to room 1], is associated with positive [negative] values of the volume flow-rate,  $\dot{V}_F$ , and corresponding positive [negative] values of the mass flow-rate,  $\dot{M}_F$ .

For the purpose of the present calculations, the fan curve of Eq. (3) is approximated by linear interpolation between and linear extrapolation beyond two or more pairs of specified values,  $[\dot{V}_{F,J}, \Delta p_{F,J}]$ ,  $J = 1$  to  $JTERM > 1$ , which identify  $[\dot{V}_F, \Delta p_F]$  points on the curve. If only one point is specified, i.e.,  $JTERM = 1$  with specified  $\dot{V}_{F,1}$  and  $\Delta p_{F,1}$ , then FANRES models the fan curve as a line of constant  $\dot{V}_F = \dot{V}_{F,1}$ . In the latter case, the fan/duct system simulated is assumed to be one which always delivers the specified volume flow-rate.

### Flow Resistance of the Duct

Since the environments of the two rooms 1 and 2 are generally different, the density of the flow through the system depends on flow direction, i.e., which of the two rooms is the source of the flow. Flow resistance through the duct, to be defined below, would also be dependent generally on flow direction. Define the density of the flow and the resistance of the duct as  $\rho_{DUCT,1}$  and  $R_1$ , respectively, if the flow is forward flow, from room 1, and as  $\rho_{DUCT,2}$  and  $R_2$ , respectively, if the flow is backward flow, from room 2. The  $\rho_{DUCT,i}$  and  $R_i$  are assumed to be specified.

From the above-specified parameters define and evaluate the adjusted pressure differences,  $\Delta p_{ADJ,i}$ , as

$$\Delta p_{ADJ,i} = \Delta p_{SYSTEM} + \rho_{DUCT,i} g \Delta y_{SYSTEM} \quad (5)$$

where  $i = 1$  or  $2$  for forward or backward flow, respectively. Then, for a particular value of  $\Delta p_F$  there could be

forward flow through systems with non-zero duct resistance if:

$$(\Delta p_F + \Delta p_{ADJ,1}) > 0 \quad (6)$$

Extending the ideas presented in [1],

when Eq. (6) is satisfied and there is forward flow, the duct flow resistance,  $R_1$ , is defined as the value which would lead to a mass flow-rate through the system duct,  $\dot{M}_{DUCT,1} > 0$ , and corresponding volume flow-rate,  $\dot{V}_{DUCT,1} > 0$ , which satisfies:

$$R_1 = (\Delta p_F + \Delta p_{ADJ,1})^{1/2} / \dot{M}_{DUCT,1} = (\Delta p_F + \Delta p_{ADJ,1})^{1/2} / \rho_{DUCT,1} \dot{V}_{DUCT,1} \quad (7)$$

Similarly, there could be

backward flow through systems with non-zero duct resistance if:

$$(\Delta p_F + \Delta p_{ADJ,2}) < 0 \quad (8)$$

and when Eq. (8) is satisfied and there is backward flow, the duct flow resistance,  $R_2$ , is defined as the value which would lead to a mass flow-rate through the system duct,  $\dot{M}_{DUCT,2} < 0$ , and corresponding volume flow-rate,  $\dot{V}_{DUCT,2} < 0$ , which satisfies:

$$R_2 = -(-\Delta p_F - \Delta p_{ADJ,2})^{1/2} / \dot{M}_{DUCT,2} = -(-\Delta p_F - \Delta p_{ADJ,2})^{1/2} / \rho_{DUCT,2} \dot{V}_{DUCT,2} \quad (9)$$

Note that for a particular value of  $\Delta p_F$  there is a possibility that the conditions of Eqs. (6) and (8) for forward and backward flow can be satisfied simultaneously. When such a circumstance occurs with the solution value of  $\Delta p_F$ , the correct one of the two possible solutions to the flow problem of Eqs. (7) and (9), would require other considerations.

If the duct flow resistance is negligible, i.e.,  $R_1 = R_2 = 0$  then  $\Delta p_F$  solutions to

$$\Delta p_F = -\Delta p_{ADJ,1} \text{ if } f(-\Delta p_{ADJ,1}) > 0$$

or

$$\Delta p_F = -\Delta p_{ADJ,2} \text{ if } f(-\Delta p_{ADJ,2}) > 0$$

(10)

and the corresponding  $\dot{V}_{DUCT} = \dot{V}_F$  solutions that would be obtained from Eq. (3) would represent valid  $[\dot{V}_F, \Delta p_F]$  solution pairs to the desired fan/duct flow problem.

#### The Problem of Determining Possible Solutions for the System Mass Flow-Rate and the Corresponding $\Delta p_F$ When Duct Resistance is Non-Zero

For a valid fan/duct system flow solution, the duct volume flow-rate of Eq. (7) or (9) and its corresponding mass flow rate must be identical to the fan volume flow-rate of Eq. (3) and its corresponding mass flow-rate. Equating these volume flow-rates leads to the following result:

For non-zero duct resistance,  $\Delta p_F$  solutions to

$$|\Delta p_F + \Delta p_{ADJ,1}|^{1/2} / (\rho_{DUCT,1} R_1) = f(\Delta p_F) \text{ if } f(\Delta p_F) > 0 \quad (11)$$

or

$$-|\Delta p_F + \Delta p_{ADJ,2}|^{1/2} / (\rho_{DUCT,2} R_2) = f(\Delta p_F) \text{ if } f(\Delta p_F) < 0$$

and the corresponding  $\dot{V}_{DUCT} = \dot{V}_E$  solutions that would be obtained from Eq. (3) would represent valid  $[\dot{V}_F, \Delta p_F]$  solution pairs to the desired fan/duct flow problem.

Define the  $\dot{V}_F = 0$  intercept of the simulated fan curve of Eq. (3) as  $\Delta p_{F,0}$ , i.e.,

$$f(\Delta p_{F,0}) = 0 \quad (12)$$

Then it can be shown that:

A forward-flow solution,  $\Delta p_{F,SOLN}$ , to Eqs. (10) or (11) exists if and only if  $\Delta p_{ADJ,1}$  satisfies  $-\Delta p_{ADJ,1} < \Delta p_{F,0}$ , and such a solution is bounded by  $-\Delta p_{ADJ,1} \leq \Delta p_{F,SOLN} < \Delta p_{F,0}$ . (13)

A backward-flow solution,  $\Delta p_{F,SOLN}$ , to Eqs. (10) or (11) exists if and only if  $\Delta p_{ADJ,2}$  satisfies  $-\Delta p_{ADJ,2} > \Delta p_{F,0}$ , and such a solution is bounded by  $\Delta p_{F,0} < \Delta p_{F,SOLN} \leq -\Delta p_{ADJ,2}$ . (14)

It also follows from this that

double solutions to Eqs. (10) or (11) require:

$$\Delta p_{ADJ,1} - \Delta p_{ADJ,2} > 0 \text{ or } (\rho_{DUCT,1} - \rho_{DUCT,2}) \Delta y_{SYSTEM} > 0 \quad (15)$$

The results of (13)-(15) are useful in the determining numerically the  $\Delta p_{F,SOLN}$  solutions.

If Eqs. (10) or (11) do not lead to any solutions, then the solution to the duct/fan flow problem is one of zero flow, i.e.,

$$\dot{V}_{DUCT} = \dot{V}_F = 0 \text{ and } \Delta p_F = \Delta p_{F,0} \text{ if there are no solutions to Eqs. (10) or (11)} \quad (16)$$

There are four possible classes of solution to Eqs. (10) or (11), namely: a single forward-flow solution and no backward-flow solution, a single backward-flow solution and no forward-flow solution, one forward-flow solution and one backward-flow solutions, and no solutions at all. In the first two classes, the unique solution obtained is taken to be the desired solution to the



fan/duct flow problem. When two solutions are obtained, both are returned as possible solutions to the fan/duct flow problem.

When no solutions to Eqs. (10) or (11) are obtained, the single solution of Eq. (16) is taken to be the solution to the fan/duct flow problem.

#### OUTPUT

$\dot{M}_{\text{SOLN},N}$ ,  $N = 1$  or  $2$

Solutions for the mass flow-rate through the system. If a flow solution is from room 1 to room 2 (normal inlet to normal outlet, i.e., in the normal-fan-operation flow direction), then  $\dot{M}_{\text{SOLN},N} > 0$ . If a flow is from room 2 to room 1 (normal outlet to normal inlet, i.e., a condition of fan backflow), then  $\dot{M}_{\text{SOLN},N} < 0$ .  $\dot{M}_{\text{SOLN},N} = 0$  indicates a zero flow-rate solution. If there is only one solution, i.e.,  $N_{\text{SOLN}} = 1$ , then it is  $\dot{M}_{\text{SOLN},1}$ . [kg/s]

$N_{\text{SOLN}} = 1$  or  $2$

Number of solutions.

#### INPUT

ACC

Absolute error tolerance for the fan/duct system's computed mass flow rate(s).

$\dot{V}_{F,J}$ ,  $J = 1$  to  $J\text{TERM}$

Values of the volume flow-rate at  $J\text{TERM}$  points used to approximate the fan operating curve (volume flow-rate vs cross-fan pressure difference). [ $\text{m}^3/\text{s}$ ]

$J\text{TERM}$

Number of points used to approximate the fan operating curve.

$R_1$  [ $R_2$ ]

Effective resistance of the duct when flow is in the forward [backward] direction. [ $\text{Pa}^{1/2}\text{kg}^{-1}\text{s} = (\text{kg}/(\text{m}\cdot\text{s}^2))^{1/2}\text{kg}^{-1}\text{s}$ ]

$\Delta p_{F,J}$ ,  $J = 1$  to  $J_{\text{TERM}}$

Values of the cross-fan pressure difference at  $J_{\text{TERM}}$  points used to approximate the fan operating curve (volume flow-rate vs cross-fan pressure difference). [ $\text{Pa} = \text{kg}/(\text{m}\cdot\text{s}^2)$ ]

$\Delta p_{\text{SYSTEM}}$

Pressure difference across the system end-points, see Eq. (2). [ $\text{Pa} = \text{kg}/(\text{m}\cdot\text{s}^2)$ ]

$\Delta y_{\text{SYSTEM}}$

Difference in elevation of the system endpoints, see Eq. (1). [m]

$\rho_{\text{DUCT},1}$  [ $\rho_{\text{DUCT},2}$ ]

Density of the system flow when flow is in the forward [backward] direction. [ $\text{kg}/\text{m}^3$ ]

### CALCULATIONS

1. Calculate  $\Delta p_{\text{ADJ},1}$  and  $\Delta p_{\text{ADJ},2}$  from Eq. (5).
2. Calculate the forward-flow solution if it exists.
3. Calculate the backward-flow solution if it exists.
4. Return to the calling routine.
  - a. If  $\dot{M}_{\text{DUCT},1}$  is non-zero and  $\dot{M}_{\text{DUCT},2} = 0$ , then there is a unique forward solution. Set  $\dot{M}_{\text{SOLN},1} = \dot{M}_{\text{DUCT},1}$ ,  $N_{\text{SOLN}} = 1$ , and return to the calling routine.
  - b. If  $\dot{M}_{\text{DUCT},1} = 0$  and  $\dot{M}_{\text{DUCT},2}$  is non-zero, then there is a unique backward solution. Set  $\dot{M}_{\text{SOLN},1} = \dot{M}_{\text{DUCT},2}$ ,  $N_{\text{SOLN}} = 1$ , and return to the calling routine.

- c. If  $\dot{M}_{DUCT,1} = 0$  and  $\dot{M}_{DUCT,2} = 0$ , then there is a zero-flow solution. Set  $\dot{M}_{SOLN,1} = 0$ ,  $N_{SOLN} = 1$ , and return to the calling routine.
- d. If  $\dot{M}_{DUCT,1}$  and  $\dot{M}_{DUCT,2}$  are both nonzero then there are two solutions, one for forward flow and one for backward flow. Set  $\dot{M}_{DUCT,1} = \dot{M}_{DUCT,1}$ ,  $\dot{M}_{SOLN,2} = \dot{M}_{DUCT,2}$ ,  $N_{SOLN} = 2$ , and return to the calling routine.

### SUBROUTINES USED

INTERP [3]

### REFERENCES

- [1] Cooper, L.Y. and Forney, G.P., Section 3.4.5 of Consolidated Compartment Fire Model (CCFM) Computer Code Application CCFM.VENTS - Part I: Physical Basis, NISTIR 90-4342, National Institute of Standards and Technology, Gaithersburg MD, 1990.
- [2] Klote, J.H. and Cooper, L.Y., Model of a Fan-Resistance Ventilation System and Its Application to Fire Modeling, NISTIR 89-4141, National Institute of Standards and Technology, Gaithersburg MD, September 1989.
- [3] Forney, G.P., Algorithm INTERP - Interpolates A Table of Numbers, Appendix B of Consolidated Compartment Fire Model (CCFM) Computer Code Application CCFM.VENTS - Part III: Algorithms and Subroutines, Cooper, L.Y., and Forney, G.P., Eds., NISTIR 90-4344, National Institute of Standards and Technology, Gaithersburg MD, 1990.

### SUBROUTINE VARIABLES

All nomenclature in the subroutine is identical to the nomenclature used above except for:

$\dot{M}_{DUCT,1}, \dot{M}_{DUCT,2}$	-	XMFAN1, XMFAN2 [kg/s]
$\dot{M}_{SOLN,1}, \dot{M}_{SOLN,2}$	-	FANMAS(1), FANMAS(2) [kg/s]
$N_{SOLN}$	-	NSOLN
$\dot{V}_{F,J}$	-	FANVOL(J) [m <sup>3</sup> /s]
$R_1, R_2$	-	RESIS(1), RESIS(2) [Pa <sup>1/2</sup> kg <sup>-1</sup> s = {kg/(m·s <sup>2</sup> )} <sup>1/2</sup> kg <sup>-1</sup> s]
$\Delta P_{F,J}$	-	FANPRS(J) [Pa = kg/(m·s <sup>2</sup> )]

FANRES - 7

$\Delta y_{\text{SYSTEM}}$	-	DYS [m]
$\Delta p_{\text{ADJ},1}, \Delta p_{\text{ADJ},2}$	-	DPFAJ [Pa = kg/(m·s <sup>2</sup> )]
$\Delta p_{\text{F},0}$	-	DPFAN [Pa = kg/(m·s <sup>2</sup> )]
$\Delta p_{\text{SYSTEM}}$	-	DPS [Pa = kg/(m·s <sup>2</sup> )]
$\rho_{\text{DUCT},1}, \rho_{\text{DUCT},2}$	-	RHOSYS(1), RHOSYS(2) [kg/m <sup>3</sup> ]

**PREPARED BY**

Leonard Y. Cooper, Glenn P. Forney, and John H. Klote  
 July 1989




Figure 1. The fire-generated environment in two rooms connected by a fan/duct system.




Figure 2. A sketch of a generic fan-operating curve which increases monotonically in volume-flow-rate with decreasing cross-fan pressure difference.

# SUBROUTINE

```
      SUBROUTINE FANRES(
1      I      ACC,FANPRS,FANVOL,JTERM,DPS,RESIS,DYS,RHOSYS,
O      FANMAS,NSOLN
+      )
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
C*BEG
C***  FANRES  GENERIC - FIND THE MASS THAT A FAN WILL DELIVER UNDER
C      GIVEN CONDITIONS IN THE INLET AND OUTLET ROOMS.
C
C***  SUBROUTINE VARIABLES
C
C      INPUT
C      -----
C      ACC      REQUESTED ACCURACY
C      FANPRS    LIST OF JTERM PRESSURES [P]
C      FANVOL    LIST OF JTERM VOLUME FLOWS [M**3/S]
C      JTERM     NUMBER OF POINTS IN FAN CURVE DEFINED BY (FANPRS,FANVOL)
C      DPS       PRESSURE DIFFERENCE ACROSS THE SYSTEM END POINTS [P]
C      RESIS     RESITANCE OF FAN SYSTEM FOR FORWARD AND REVERSE FLOW
C                CONDITIONS
C      DYS       DIFFERENCE IN ELEVATION OF THE SYSTEM ENDPOINTS [M]
C      RHOSYS    DENSITY AT INLET AND OUTLET OF FAN [KG/M**3]
C
C      OUTPUT
C      -----
C      FANMAS    MASS DELIVERED BY FAN [KG/S]
C      NSOLN     NUMBER OF SOLUTIONS
C*END

      DIMENSION FANPRS(*),FANVOL(*),RHOSYS(2)
      DIMENSION FANMAS(2), RESIS(2)
      EXTERNAL FANFOR,FANREV

      RSISFW = RESIS(1)
      RSISRV = RESIS(2)
C
C***  GET CANDIDATE FORWARD SOLUTION
C
C***  BRACKET ROOT
C
      DPMIN = FANPRS(1)
      IF(JTERM.GE.2)THEN
          DPMAX = FANPRS(JTERM)
      ELSE
          DPMAX = DPMIN + 1.0D0
      ENDIF
      IF(RSISFW.NE.0.0D0)THEN
          CALL BRACK(FANFOR,DPMIN,DPMAX,XA,XB,
1          RSISFW,RHOSYS,DYS,DPS,FANPRS,FANVOL,JTERM)
C
```

```

C*** REFINE ESTIMATE FOR ROOT
C
      CALL RTSAF2(FANFOR,XA,XB,ACC,PFAN1,
1        RSISFW,RHOSYS,DYS,DPS,FANPRS,FANVOL,JTERM)
C
C*** CALCULATE MASS FLOW THAT CORRESPONDS TO ROOT
C
      CALL MRES(PFAN1,1,RSISFW,RHOSYS,DYS,DPS,XMFAN1,XDUM)
      ENDIF
      IF(RSISRV.NE.0.0D0)THEN
C
C*** GET CANDIDATE REVERSE SOLUTION
C
C
C*** BRACKET ROOT
C
      CALL BRACK(FANREV,DPMIN,DPMAX,XA,XB,
1        RSISRV,RHOSYS,DYS,DPS,FANPRS,FANVOL,JTERM)
C
C*** REFINE ESTIMATE FOR ROOT
C
      CALL RTSAF2(FANREV,XA,XB,ACC,PFAN2,
1        RSISRV,RHOSYS,DYS,DPS,FANPRS,FANVOL,JTERM)
C
C*** CALCULATE MASS FLOW THAT CORRESPONDS TO ROOT
C
      CALL MRES(PFAN2,-1,RSISRV,RHOSYS,DYS,DPS,XMFAN2,XDUM)
      ENDIF
      IF(RSISRV.EQ.0.0D0.OR.RSISFW.EQ.0.0D0)THEN
C
C*** LOOK FOR SOLUTIONS WHEN RESISTANCE OF FAN SYSTEM IS ZERO
C
      G = 9.80D0
      DPFAJ = DPS + RHOSYS(1)*G*DYS
      CALL INTERP(FANPRS,FANVOL,JTERM,-DPFAJ,XMVOL,DXMVOL,2)
      IF(XMVOL.GT.0.0D0)THEN
        XMFAN1 = RHOSYS(1)*XMVOL
      ELSE
        XMFAN1 = 0.0D0
      ENDIF
      DPFAJ = DPS + RHOSYS(2)*G*DYS
      CALL INTERP(FANPRS,FANVOL,JTERM,-DPFAJ,XMVOL,DXMVOL,2)
      IF(XMVOL.LT.0.0D0)THEN
        XMFAN2 = RHOSYS(2)*XMVOL
      ELSE
        XMFAN2 = 0.0D0
      ENDIF
      ENDIF
C
C*** DETERMINE THE VALID SOLUTION(S)
C
      IF(XMFAN1.NE.0.0D0.AND.XMFAN2.EQ.0.0D0)THEN

```



```

C
C*** A UNIQUE SOLUTION IN THE FORWARD DIRECTION
C
      FANMAS(1) = XMFAN1
      NSOLN = 1
      ELSE IF(XMFAN1.EQ.0.0D0.AND.XMFAN2.NE.0.0D0)THEN
C
C*** A UNIQUE SOLUTION IN THE REVERSE DIRECTION
C
      FANMAS(1) = XMFAN2
      NSOLN = 1
      ELSE IF(XMFAN1.EQ.0.0D0.AND.XMFAN2.EQ.0.0D0)THEN
C
C*** A ZERO FLOW SOLUTION
C
      FANMAS(1) = 0.0D0
      NSOLN = 1
      ELSE
C
C*** TWO SOLUTIONS
C
      FANMAS(1) = XMFAN1
      FANMAS(2) = XMFAN2
      NSOLN = 2
      ENDIF
      RETURN
      END
      SUBROUTINE FANFOR(DPFAN,XMNET,DXMNET,
1      RESIS,RHO,DZ,DPS,FANPRS,FANVOL,NPTS)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
C*BEG
C*** FANFOR  GENERIC - CALCULATE THE NET MASS FLOW IN A FAN SYSTEM
C      ASSUMING THE FLOW IS IN A FORWARD OR NORMAL DIRECTION.
C      THIS ROUTINE IS CALLED BY A ZERO FINDER THAT LOCATES A
C      PRESSURE DIFFERENCE DPFAN, THAT CAUSES THE NET MASS FLOW
C      TO BE ZERO.
C
C*** SUBROUTINE ARGUMENTS
C
C      INPUT
C      -----
C      DPFAN      PRESSURE ACCROSS FAN [P]
C      RESIS      RESITANCE OF FAN SYSTEM
C      RHO        DENSITY AT INLET AND OUTLET OF FAN [KG/M**3]
C      DZ         DIFFERENCE IN ELEVATION OF THE SYSTEM ENDPOINTS [M]
C      DPS        PRESSURE DIFFERENCE ACROSS THE SYSTEM END POINTS [P]
C      FANPRS     LIST OF NPTS PRESSURES [P]
C      FANVOL     LIST OF NPTS VOLUME FLOWS [M**3/S]
C      NPTS       NUMBER OF POINTS IN FAN CURVE DEFINED BY (FANPRS,FANVOL)
C
C      OUTPUT
C      -----

```

```

C      XMNET      NET MASS FLOW [KG/(S*M**3)]
C      DXMET      DERIVATIVE OF NET FAN FLOW
C*END

      DIMENSION RHO(2), FANPRS(*),FANVOL(*)
C
C*** CALCULATE MASS RESISTANCE FLOW
C
      CALL MRES(DPFAN,1,RESIS,RHO,DZ,DPS,XMRES,DXMRES)
C
C*** CALCULATE VOLUME FLOW
C
      CALL INTERP(FANPRS,FANVOL,NPTS,DPFAN,XMVOL,DXMVOL,2)
C
C*** CONVERT VOLUME FLOW TO MASS FLOW ASSUMING FORWARD DIRECTION
C      (USE DENSITY IN ROOM 1)
C
      XMFAN = RHO(1)*XMVOL
      DXMFAN = RHO(1)*DXMVOL
C
C*** COMPUTE DIFFERENCE BETWEEN TWO FLOWS
C
      XMNET = XMFAN - XMRES
      DXMNET = DXMFAN - DXMRES
      RETURN
      END
      SUBROUTINE FANREV(DPFAN,XMNET,DXMNET,
1      RESIS,RHO,DZ,DPS,FANPRS,FANVOL,NPTS)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
C*BEG
C*** FANREV  GENERIC - CALCULATE THE NET MASS FLOW IN A FAN SYSTEM
C      ASSUMING THE FLOW IS IN A REVERSE DIRECTION.  THIS
C      ROUTINE IS CALLED BY A ZERO FINDER THAT LOCATES A
C      PRESSURE DIFFERENCE DPFAN, THAT CAUSES THE NET MASS
C      FLOW TO BE ZERO.
C
C*** SUBROUTINE VARIABLES
C
C      INPUT
C      -----
C      DPFAN      PRESSURE ACCROSS FAN [P]
C      RESIS      RESITANCE OF FAN SYSTEM
C      RHO        DENSITY AT INLET AND OUTLET OF FAN [KG/M**3]
C      DZ         ELEVATION OF INLET AND OUTLET OF FAN [M]
C      DPS        PRESSURE DIFFERENCE ACROSS THE SYSTEM END POINTS [P]
C      FANPRS     LIST OF NPTS PRESSURES [P]
C      FANVOL     LIST OF NPTS VOLUME FLOWS [M**3/S]
C      NPTS       NUMBER OF POINTS IN FAN CURVE DEFINED BY (FANPRS,FANVOL)
C
C      OUTPUT
C      -----
C      XMNET      NET MASS FLOW [M**3/S]

```

```

C      DXMET      DERIVATIVE OF NET FAN FLOW
C*END

      DIMENSION RHO(2), FANPRS(*),FANVOL(*)
C
C*** CALCULATE MASS RESISTANCE FLOW
C
      CALL MRES(DPFAN, -1, RESIS, RHO, DZ, DPS, XMRES, DXMRES)
C
C*** CALCULATE VOLUME FLOW
C
      CALL INTERP(FANPRS, FANVOL, NPTS, DPFAN, XMVOL, DXMVOL, 2)
C
C*** CONVERT VOLUME FLOW TO MASS FLOW ASSUMING REVERSE DIRECTION
C      (USE DENSITY IN ROOM 2)
C
      XMFAN = RHO(2)*XMVOL
      DXMFAN = RHO(2)*DXMVOL
C
C*** COMPUTE DIFFERENCE BETWEEN TWO FLOWS
C
      XMNET = XMFAN - XMRES
      DXMNET = DXMFAN - DXMRES
      RETURN
      END
      SUBROUTINE MRES(DPFAN, IDIR, RESIS, RHO, DZ, DPS, XMRES, DXMRES)
      IMPLICIT DOUBLE PRECISION (A-H, O-Z)
C*BEG
C*** MRES  GENERIC - THIS ROUTINE COMPUTES THE MASS FLOW IN A FAN
C          SYSTEM FOR A GIVEN RESITANCE AND PRESSURE
C
C*** SUBROUTINE ARGUMENTS
C
C      INPUT
C      -----
C      DPFAN  PRESSURE ACCROSS FAN [P]
C      IDIR   DIRECTION OF FLOW IDIR>0 FOR FORWARD FLOW
C              IDIR<0 FOR REVERSE FLOW
C      RESIS  RESISTANCE OF FAN SYSTEM
C      RHO    DENSITY AT INLET AND OUTLET OF FAN [KG/M**3]
C      DZ     DIFFERENCE IN ELEVATION OF THE SYSTEM ENDPOINTS [M]
C      DPS    PRESSURE DIFFERENCE ACROSS THE SYSTEM END POINTS [P]
C
C      OUTPUT
C      -----
C      XMRES  MASS RESISTANCE FLOW [KG/S]
C      DXMRES DERIVATIVE OF MASS RESISTANCE FLOW
C*END

```

```

      DIMENSION RHO(2), PROOM(2)
      PARAMETER (G=9.800D0)
      I = 1

```

```

      IF(IDIR.LT.0)I = 2
      DPAJ = DPS + RHO(I)*G*DZ
      DPNET = DPFAN + DPAJ
      SQRNET = SQRT(ABS(DPNET))
      XMRES = 0.0D0
      DXMRES = 0.0D0
      IF(I.EQ.1.AND.DPNET.GT.0.0D0)THEN
        XMRES = SQRNET/RESIS
        DXMRES = .5D0/(SQRNET*RESIS)
      ELSE IF(I.EQ.2.AND.DPNET.LT.0.0D0)THEN
        XMRES = -SQRNET/RESIS
        DXMRES = .5D0/(SQRNET*RESIS)
      ENDIF
      RETURN
      END
      SUBROUTINE BRACK(FUNCD,A,B,XA,XB,
1      RESIS,RHO,DZ,DP,FANPRS,FANVOL,NPTS)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
C*BEG
C
C***  BRACK  SPECIFIC - THIS ROUTINE BRACKETS THE ROOT OF THE
C          FUNCTION, FUNCD SO THAT THE ROOT LIES IN THE INTERVAL
C          [XA,XB].  THIS SUBROUTINE IS BASED ON A SUBROUTINE
C          NAMED ZBRAC FOUND IN:
C
C          NUMERICAL RECIPES, THE ART OF SCIENTIFIC COMPUTING, PRESS,
C          WILLIAM H., FLANNERY, BRAIN P., TEUKOLSKY, SAUL AND
C          VETTERLING, WILLIAM T., CAMBRIDGE, PP. 245, 1986
C
C  SUBROUTINE ARGUMENTS
C
C  INPUT
C  -----
C  FUNCD  FUNCTION TO BE BRACKETED
C  A      CANDIDATE LEFT BRACKET
C  B      CANDIDATE RIGHT BRACKET
C  RESIS  RESITANCE OF FAN SYSTEM
C  RHO    DENSITY AT INLET AND OUTLET OF FAN [KG/M**3]
C  DZ     DIFFERENCE IN ELEVATION OF THE SYSTEM ENDPOINTS [M]
C  DP     PRESSURE DIFFERENCE ACROSS THE SYSTEM END POINTS [P]
C  FANPRS LIST OF NPTS PRESSURES [P]
C  FANVOL LIST OF NPTS VOLUME FLOWS [M**3/S]
C  NPTS   NUMBER OF POINTS IN FAN CURVE DEFINED BY (FANPRS,FANVOL)
C
C  OUTPUT
C  -----
C  XA     ACTUAL LEFT BRACKET
C  XB     ACTUAL RIGHT BRACKET
C*END

```

```

      DIMENSION RHO(*), FANPRS(*), FANVOL(*)
      PARAMETER (MAXIT=50)

```

```

AL = A
AU = A
BL = B
BU = B
CALL FUNCD(AL, FAL, DFAL, RESIS, RHO, DZ, DP, FANPRS, FANVOL, NPTS)
CALL FUNCD(BL, FBL, DFBL, RESIS, RHO, DZ, DP, FANPRS, FANVOL, NPTS)
FBU = FBL
FAU = FAL
DO 10 I = 1, MAXIT
  IF(FAL*FBL.LE.0.0DO)THEN
    XA = AL
    XB = BL
    RETURN
  ENDIF
  AU = BU
  FAU = FBU
  BU = BU + (BU-A)*2.0DO
  CALL FUNCD(BU, FBU, DFBU, RESIS, RHO, DZ, DP, FANPRS, FANVOL, NPTS)
  IF(FAU*FBU.LE.0.0DO)THEN
    XA = AU
    XB = BU
    RETURN
  ENDIF
  BL = AL
  FBL = FAL
  AL = AL - (B-AL)*2.0DO
  CALL FUNCD(AL, FAL, DFAL, RESIS, RHO, DZ, DP, FANPRS, FANVOL, NPTS)
10 CONTINUE
  CALL MSGPRT('ROOT NOT BRACKETED IN BRAKIT', 3)
  RETURN
END
SUBROUTINE RTSAF2(FUNCD,X1,X2,XACC,RTSAFE,
1      RESIS,RHO,DZ,DP,FANPRS,FANVOL,NPTS)
  IMPLICIT DOUBLE PRECISION (A-H,O-Z)
C*BEG
C
C***  RTSAF2  SPECIFIC - THIS ROUTINE FINDS A VALUE OF RTSAFE SUCH
C            THAT FUNCD(RTSAFE) = 0.  IT USES A MODIFICATION OF
C            NEWTON'S METHOD.  THIS SUBROUTINE IS BASED ON
C            A SUBROUTINE NAMED RTSAF FOUND IN:
C
C            NUMERICAL RECIPES, THE ART OF SCIENTIFIC COMPUTING, PRESS,
C            WILLIAM H., FLANNERY, BRAIN P., TEUKOLSKY, SAUL AND
C            VETTERLING, WILLIAM T., CAMBRIDGE, P. 258, 1986
C
C***  SUBROUTINE ARGUMENTS
C
C  INPUT
C  ----
C  FUNCD      FUNCTION WHOSE ZERO IS DESIRED
C  X1         LEFT BRACKET
C  X2         RIGHT BRACKET

```

```

C  XACC      ACCURACY REQUEST
C  RESIS     RESITANCE OF FAN SYSTEM
C  RHO       DENSITY AT INLET AND OUTLET OF FAN [KG/M**3]
C  DZ        DIFFERENCE IN ELEVATION OF THE SYSTEM ENDPOINTS [M]
C  DPS       PRESSURE DIFFERENCE ACROSS THE SYSTEM END POINTS [P]
C  FANPRS    LIST OF NPTS PRESSURES [P]
C  FANVOL    LIST OF NPTS VOLUME FLOWS [M**3/S]
C  NPTS      NUMBER OF POINTS IN FAN CURVE DEFINED BY (FANPRS,FANVOL)
C
C  OUTPUT
C  -----
C
C  RTSAFE    ZERO OF THE FUNCTION FUNCD WITHIN A TOLERANCE OF XACC
C*END

```

```

      DIMENSION RHO(2),FANPRS(*),FANVOL(*)
      PARAMETER (MAXIT=100)

```

```

C
      CALL FUNCD(X1,FL,DF,RESIS,RHO,DZ,DP,FANPRS,FANVOL,NPTS)
      IF(FL.EQ.0.0D0)THEN
        RTSAFE = X1
        GO TO 999
      ENDIF
      CALL FUNCD(X2,FH,DF,RESIS,RHO,DZ,DP,FANPRS,FANVOL,NPTS)
      IF(FH.EQ.0.0D0)THEN
        RTSAFE = X2
        GO TO 999
      ENDIF
      IF(FL*FH.GT.0.)THEN
        CALL MSGPRT(' ROOT MUST BE BRACKETED IN RTSAF2',3)
        RETURN
      ENDIF
      IF(FL.LT.0.)THEN
        XL=X1
        XH=X2
      ELSE
        XH=X1
        XL=X2
        SWAP=FL
        FL=FH
        FH=SWAP
      ENDIF
      RTSAFE=.5*(X1+X2)
      DXOLD=ABS(X2-X1)
      DX=DXOLD
      CALL FUNCD(RTSAFE,F,DF,RESIS,RHO,DZ,DP,FANPRS,FANVOL,NPTS)
      DO 11 J=1,MAXIT
        IF(((RTSAFE-XH)*DF-F)*((RTSAFE-XL)*DF-F).GE.0.
*        .OR. ABS(2.*F).GT.ABS(DXOLD*DF) ) THEN
          DXOLD=DX
          DX=0.5*(XH-XL)
          RTSAFE=XL+DX

```

```

        IF(XL.EQ.RTSAFE)GO TO 999
    ELSE
        DXOLD=DX
        DX=F/DF
        TEMP=RTSAFE
        RTSAFE=RTSAFE-DX
        IF(TEMP.EQ.RTSAFE)GO TO 999
    ENDIF
    IF(ABS(DX).LT.XACC)GO TO 999
    CALL FUNCD(RTSAFE,F,DF,RESIS,RHO,DZ,DP,FANPRS,FANVOL,NPTS)
    IF(F.LT.0.) THEN
        XL=RTSAFE
        FL=F
    ELSE
        XH=RTSAFE
        FH=F
    ENDIF
11  CONTINUE
    CALL MSGPRT(' NON-CONVERGENCE IN RTSAF2',3)
    RETURN
999 CONTINUE
    RETURN
    END

```





FLOG02 - Deposition of Mass, Enthalpy, Oxygen, and Other Product-of-Combustion Flows Passing Between Two Rooms Through a Vertical, Constant-Width Vent or Through a Forced Ventilation System

DESCRIPTION

Consider an instant of time during the simulation of a multi-room compartment fire environment. The flow through a constant-width natural vent located in a wall segment common to any two rooms of the compartment can be characterized generally as having  $N_{SLAB} \leq 6$ , uniform-property, locally-near-horizontal, adjacent slabs of flow. This is depicted in Figure 1. The elevations of the top and bottom of each of these flow slabs is such that it is deposited entirely into only one of the layers of the receiving room. Also, the source of each flow slab is entirely from one of the two layers of the source room.

The flow through a forced-ventilation system connecting two fire-generated room environments can often be characterized as being fully-mixed and uniform-property. In the duct and near the inlet vent this flow would be formed by one or two flow slabs entering the vent from the source room, one flowing from the upper layer and one flowing from the lower layer. This is depicted in Figure 2 for the case of two flow slabs. The duct flow would be ejected at the outlet vent into the receiving room as one or two identical-property flow slabs, one flow slab entering the upper layer and one entering the lower layer.

In the case of the unforced natural vent, each of the flow slabs leaves one room and enters the immediately adjacent room at the same elevation interval. In the case of the forced-ventilation system, a flow slab generally leaves one room and enters the other at two completely different elevations. Indeed, the two rooms joined by the ventilation system will generally be displaced arbitrarily from one another both horizontally and vertically.

It is assumed that a complete description of the flow slabs for the time and natural vent or forced-ventilation system of interest has been obtained and is available. The present algorithm determines the contributions of the flow slabs to the instantaneous rates of flow of mass, enthalpy, oxygen and other products of combustion into the upper and lower layer of each of the two rooms being affected [1]. When this algorithm is used for forced ventilation flows a separate application of the algorithm would be required for each inlet and outlet vent. In the case of an inlet vent, the modifications to the source-room layers would be calculated by the algorithm and implemented, while the calculated effects to the receiving room would be ignored, i.e., the receiving room would be treated as a "dummy" room. In the case of an outlet vent, the modifications to the receiving-room layers would be calculated and implemented while those to the source room would be ignored.

The input descriptions of the flow slabs is assumed here to include their number,  $N_{SLAB}$ , and for each of these: the direction of its flow, i.e., into room 1 or room 2, as identified by  $I_{DIR,N}$ , total mass flow rate,  $\dot{M}_{SLAB,N}$ , enthalpy flow rate,  $\dot{Q}_{SLAB,N}$ , mass flow rate of oxygen,  $\dot{P}_{O2,SLAB,N}$ , flow rates of  $N_{PROD} - 1$  other products of combustion being tracked by the simulation,

$\dot{P}_{K,SLAB,N}$ ,  $K = 2$  to  $N_{PROD}$ , effective heights of the flow slabs in rooms 1 and 2 above a datum elevation,  $Y_{SLAB,1,N}$  and  $Y_{SLAB,2,N}$ , and the absolute temperature,  $T_{SLAB,N}$ , of the slab upon penetrating the receiving room, after deceleration to relatively small velocity, but before significant mixing. Also assumed to be known in each of the two rooms is the absolute temperature of the upper and lower layers (if they are of nonzero thickness),  $T_{U,I}$  and  $T_{L,I}$ , and the elevation relative to the datum elevation of the ceiling, layer interface and floor,  $Y_{CEIL,I}$ ,  $Y_{LAYER,I}$ , and  $Y_{FLOOR,I}$ ,  $I = 1$  or  $2$ , respectively. When one is dealing with slab flows through an unforced vertical vent in a common wall between adjacent rooms the input data should satisfy  $Y_{SLAB,1,N} = Y_{SLAB,2,N}$  for all  $N_{SLAB}$  slabs. As discussed above, this is not generally true for the flow through a forced-ventilation system where the system endpoint elevations are generally different.

FLOG02 determines the source and destination layer(s) of all the slab flows of a vent and the net contributions of these slab flows to the rates of increase of mass,  $\dot{M}_{U,I}$  and  $\dot{M}_{L,I}$ , enthalpy,  $\dot{Q}_{U,I}$  and  $\dot{Q}_{L,I}$ , and products of combustion,  $\dot{P}_{K,U,I}$  and  $\dot{P}_{K,L,I}$ , of the upper and lower layers of the two rooms. The enthalpy-flow-rate calculation includes heat transfer exchanges between the incoming enthalpy flows and the bounding surfaces of the receiving rooms. The algorithm for mass apportionment of a flow slab entering the receiving room is based on an extension to a simple scheme, proposed on pages 24-25 of [2], for depositing into or apportioning between the upper and lower layers of the receiving room, the flow of any particular slab. In determining the net enthalpy flow rate to a receiving layer associated with a given mass flow rate contribution, an algorithm based on an extension of ideas proposed in [3] and used in [4] for heat transfer losses to the walls of the room is adopted.

The rule of apportioning the entering slab flow has the following specific elements:

1. If a) the temperature of the slab is greater than the temperature of the lower layer in the receiving room, but by an amount less than a small user-specified value  $\Delta T_{EPS}$ , and b) the upper layer of the receiving room is of zero thickness, i.e., the layer interface is at the ceiling of the receiving room, then c) the entire slab is deposited in the lower layer and no new upper layer is allowed to grow by virtue of mass additions from the slab flow.
2. If a) the temperature of the slab is less than the temperature of the upper layer in the receiving room, but by an amount less than a small user-specified value  $\Delta T_{EPS}$ , and b) the lower layer of the receiving room is of zero thickness, i.e., the layer interface is at the floor of the receiving room, then c) the entire slab is deposited in the upper layer and no new lower layer is allowed to grow by virtue of mass additions from the slab flow.
3. If neither of the above two elements are relevant, and the temperature of a slab is greater than the temperature of the receiving room upper layer, then the entire slab flow is deposited in the upper layer. If it is less than that of the lower layer,

then the entire slab flow, is deposited in the lower layer. If the slab temperature is between the temperature of the upper and lower layer then the slab flow is divided between the two layers in direct proportion to the temperature differences.

The rule of determining losses of slab enthalpy flow by means of heat transfer to the upper or lower surfaces of the receiving room has the following specific elements:

1. The "excess" enthalpy flow rate of a slab, calculated relative to the temperature of the layer which receives it and the temperature of the slab itself, is determined. This is apportioned along with the "base" enthalpy flow rate, calculated at the temperature of the receiving layer, to the upper layer and/or to the lower layer, in a manner consistent with the above rule for apportioning the mass flow
2. A user-specified fraction,  $\lambda_v$ , of the upper- and lower-layer- portions of the "excess" enthalpy flow rate is assumed to be transferred to the upper and lower layer surfaces of the receiving room. The remaining enthalpy flow rates are added to the upper and lower layer.

Although the following calculations are not actually implemented in the present subroutine, it is noted that there are formulations of the basic model equations where it would be useful to determine explicitly the individual and net contributions of the flow slabs to the group of terms

$$\dot{G}_{U,I} = (C_p T_{U,I} \dot{M}_{U,I} - \dot{Q}_{U,I})$$

$$\dot{G}_{L,I} = (C_p T_{L,I} \dot{M}_{L,I} - \dot{Q}_{L,I})$$

in both rooms I. Reference here is to formulations which use layer densities or temperatures as independent variables in the differential equations governing the room environments. Under such circumstances the above groups not only appear explicitly as source terms in the generic equation set, but they often have the property of being zero identically, even when the separate  $\dot{M}_{U,I}$  and  $\dot{Q}_{U,I}$  or  $\dot{M}_{L,I}$  and  $\dot{Q}_{L,I}$  components of such slab flows are non-zero. For example, in its source room the contribution to these groups of each flow slab will be zero identically. When required, the calculation of the  $\dot{G}_{U,I}$  and  $\dot{G}_{L,I}$  groups can be carried out in a manner as to insure that this fact is properly taken account of, i.e., to insure that unnecessary introduction of roundoff error in the calculation of these groups is eliminated.

#### OUTPUT

$$\dot{G}_{U,I} [\dot{G}_{L,I}]; I = 1 \text{ or } 2$$

(NOTE: THE PRESENT SUBROUTINE DOES NOT PRODUCE THIS OUTPUT)

Total contribution of  $\dot{G}_{U,I} = (C_p T_{U,I} \dot{M}_{U,I} - \dot{Q}_{U,I}) [\dot{G}_{L,I} = (C_p T_{L,I} \dot{M}_{L,I} - \dot{Q}_{L,I})]$  to the upper [lower] layer of room I due to all  $N_{SLAB}$  slabs of the vent flow [W].

$\dot{M}_{U,I}$  [ $\dot{M}_{L,I}$ ]; I = 1 or 2

Total rate of mass flow to the upper [lower] layer of room I due to all  $N_{SLAB}$  slabs of the vent flow [kg/s].

$\dot{P}_{O_2,U,I}$  [ $\dot{P}_{O_2,L,I}$ ]; I = 1 or 2

Total rate of oxygen mass flow to the upper [lower] layer of room I due to all  $N_{SLAB}$  slabs of the vent flow [(kg of  $O_2$ )/s].

$\dot{P}_{K,U,I}$  [ $\dot{P}_{K,L,I}$ ]; K = 2 to  $N_{PROD}$ ; I = 1 or 2

Total rate of flow of product of combustion K to the upper [lower] layer of room I due to all  $N_{SLAB}$  slabs of the vent flow [(unit of product K)/s].

$\dot{Q}_{U,I}$  [ $\dot{Q}_{L,I}$ ]; I = 1 or 2

Total rate of enthalpy flow, based on absolute temperature, to the upper [lower] layer of room I due to all  $N_{SLAB}$  slabs of the vent flow [W].

$\dot{Q}_{SU,I}$  [ $\dot{Q}_{SL,I}$ ]; I = 1 or 2

Total rate of heat transfer to the upper [lower] bounding surfaces of room I due to all  $N_{SLAB}$  slabs of the vent flow [W].

## INPUT

$C_p$

Specific heat at constant pressure of the vent flow [W·s/(kg·K)] (suggest  $10^3$  W·s/(kg·K) for air as default).

$I_{DIR,N}$ ; N = 1 to  $N_{SLAB}$

A measure of the direction of the room 1-to-room 2 flow of slab N. For flow from room 1 to room 2 or from room 2 to room 1, the value of  $I_{DIR,N}$  is 1 or -1, respectively.  $I_{DIR,N}$  is 0 if the room-to-room flow is identically zero.

$\dot{M}_{\text{SLAB},N}; N = 1 \text{ to } N_{\text{SLAB}}$

Total mass flow rate in slab N [kg/s].

$N_{\text{PROD}}$

number of products of combustion, being tracked in the simulation.

$N_{\text{SLAB}}$

number of slabs of flow,  $\leq 6$ , between the bottom and top of the vent.

$\dot{P}_{\text{O}_2,\text{SLAB},N}; N = 1 \text{ to } N_{\text{SLAB}}$

Mass flow rate of oxygen in slab N [(kg of  $\text{O}_2$ )/s].

$\dot{P}_{K,\text{SLAB},N}; K = 2 \text{ to } N_{\text{PROD}}; N = 1 \text{ to } N_{\text{SLAB}}$

Flow rate of product of combustion K in slab N [(unit of product K)/s].

$\dot{Q}_{\text{SLAB},N}; N = 1 \text{ to } N_{\text{SLAB}}$

Total enthalpy flow rate in slab N based on absolute temperature [W].

$T_{\text{SLAB},N}; N = 1 \text{ to } N_{\text{SLAB}}$

Absolute temperature of the flow in slab N as it enters the receiving room [K].

$T_{U,I} [T_{L,I}]; I = 1 \text{ or } 2$

Absolute temperature of the upper [lower] layer in room I [K].

$Y_{SLAB,1,N}$  [ $Y_{SLAB,2,N}$ ];  $N = 1$  to  $N_{SLAB}$

Effective elevation of slab  $N$  in room 1 [room 2] above the datum elevation. In cases of slab flows through natural vents between adjacent rooms the input data should satisfy  $Y_{SLAB,1,N} = Y_{SLAB,2,N}$ . This is not generally true for the flow through a ventilation system. [m]

$Y_{FLOOR,I}$  [ $Y_{CEIL,I}$ ] [ $Y_{LAYER,I}$ ];  $I = 1$  or  $2$

Elevation of the floor [ceiling] [layer interface] in room  $I$  above the datum elevation [m].

$\Delta T_{EPS}$

Minimum temperature difference,  $> 0$ , between  $T_{SLAB,N}$  and  $T_{L,I}$  [ $T_{U,I}$ ] required to initiate growth of an upper [lower] layer (in the receiving room  $I$ ) from a zero thickness [K].

$\lambda_v$

Fraction of the source slab's rate-of-enthalpy inflow, computed relative to the temperature in the receiving room at the effective elevation of the slab, which is assumed to be lost to the bounding walls of the receiving room.

## CALCULATIONS

1. Set equal to zero all  $\dot{G}_{U,I}$ ,  $\dot{G}_{L,I}$ ,  $\dot{M}_{U,I}$ ,  $\dot{M}_{L,I}$ ,  $\dot{P}_{O2,U,I}$ ,  $\dot{P}_{O2,L,I}$ ,  $\dot{P}_{K,U,I}$ ,  $\dot{P}_{K,L,I}$ ,  $\dot{Q}_{U,I}$ ,  $\dot{Q}_{L,I}$ ,  $\dot{Q}_{SU,I}$ ,  $\dot{Q}_{SL,I}$ ,  $K = 2$  to  $N_{PROD}$ ,  $I = 1$  and  $2$ .
2. Consider each of the  $N_{SLAB}$  flow slabs and
  - a. Determine the destination (upper, lower or both layers of room 1 or 2) of any particular slab, slab  $N$ . Let room  $I$  be the receiving room. Using the temperature and the mass, enthalpy, oxygen, and product flow rates of slab  $N$ , adjust  $\dot{G}_{U,I}$ ,  $\dot{G}_{L,I}$ ,  $\dot{M}_{U,I}$ ,  $\dot{M}_{L,I}$ ,  $\dot{P}_{O2,U,I}$ ,  $\dot{P}_{O2,L,I}$ ,  $\dot{Q}_{U,I}$ ,  $\dot{Q}_{L,I}$ ,  $\dot{Q}_{SU,I}$ ,  $\dot{Q}_{SL,I}$ , and  $\dot{P}_{K,U,I}$ ,  $\dot{P}_{K,L,I}$ ,  $K = 2$  to  $N_{PROD}$  according to the flow diagram of Figure 3.

- b. Determine the origin (upper or lower layer of room 1 or 2) of slab N. Let room J be the room of origin. Then according to the temperature and the mass, enthalpy, oxygen and other product flow rates of slab N, adjust appropriately  $\dot{M}_{U,J}$ ,  $\dot{M}_{L,J}$ ,  $\dot{P}_{O_2,U,J}$ ,  $\dot{P}_{O_2,L,J}$ ,  $\dot{Q}_{U,J}$ ,  $\dot{Q}_{L,J}$ , and  $\dot{P}_{K,U,J}$ ,  $\dot{P}_{K,L,J}$ ,  $K = 2$  to  $N_{PROD}$ . Do not adjust  $\dot{G}_{U,J}$ ,  $\dot{G}_{L,J}$ ,  $\dot{Q}_{SU,J}$ ,  $\dot{Q}_{SL,J}$ . There are two possible cases.

Case IIa:  $Y_{SLAB,J,N} > Y_{LAYER,J}$

Slab flow comes from upper layer of room of origin. Subtract  $\dot{M}_{SLAB,N}$  from  $\dot{M}_{U,J}$ ,  $\dot{P}_{O_2,SLAB,N}$  from  $\dot{P}_{O_2,U,J}$ ,  $\dot{Q}_{SLAB,N}$  from  $\dot{Q}_{U,J}$ , and  $\dot{P}_{K,SLAB,N}$  from  $\dot{P}_{K,U,J}$ ,  $K = 2$  to  $N_{PROD}$ . Do not adjust  $\dot{M}_{L,J}$ ,  $\dot{P}_{O_2,L,J}$ ,  $\dot{Q}_{L,J}$ , or the  $\dot{P}_{K,L,J}$ 's.

Case IIb:  $Y_{SLAB,J,N} < Y_{LAYER,J}$

Slab flow comes from lower layer of room of origin. Subtract  $\dot{M}_{SLAB,N}$  from  $\dot{M}_{L,J}$ ,  $\dot{P}_{O_2,SLAB,N}$  from  $\dot{P}_{O_2,L,J}$ ,  $\dot{Q}_{SLAB,N}$  from  $\dot{Q}_{L,J}$ , and  $\dot{P}_{K,SLAB,N}$  from  $\dot{P}_{K,L,J}$ ,  $K = 2$  to  $N_{PROD}$ . Do not adjust  $\dot{M}_{U,J}$ ,  $\dot{P}_{O_2,U,J}$ ,  $\dot{Q}_{U,J}$ , or the  $\dot{P}_{K,U,J}$ 's.

## REFERENCES

- [1] Cooper, L.Y. and Forney, G.P., Section 3.3.5 of Consolidated Compartment Fire Model (CCFM) Computer Code Application CCFM.VENTS - Part I: Physical Basis, NISTIR 90-4342, National Institute of Standards and Technology, Gaithersburg MD, 1990.
- [2] Emmons, H.W., The Flow of Gases Through Vents, Home Fire Project Technical Report 75, Division of Applied Sciences, Harvard University, March 16, 1987.
- [3] Zukoski, E.E., and Kubota, T., Two-Layer Modeling of Smoke Movement in Building Fires, Fire and Materials, 4, 1, 1980.
- [4] Cooper, L.Y., A Mathematical Model For Estimating Available Safe Egress Time in Fires, Fire and Materials, 6, 3/4, pp 135-144, 1982.

## SUBROUTINE VARIABLES

$C_p$	-	CP [W·s/(kg·K)]
$F_U, F_L$	-	FU, FL
$I_{DIR,N}$	-	DIRS12(N)
$\dot{G}_{L,1}, \dot{G}_{L,2}$	-	UFLW2(1,3,1), UFLW2(2,3,1) [W]

$\dot{M}_{SLAB,N}$	-	XMSLAB(N) [kg/s]
$\dot{M}_{U,1}, \dot{M}_{U,2}$	-	UFLW2(1,1,2), UFLW2(2,1,2) [kg/s]
$\dot{M}_{L,1}, \dot{M}_{L,2}$	-	UFLW2(1,1,1), UFLW2(2,1,1) [kg/s]
$N_{PROD}$	-	NPROD
$N_{SLAB}$	-	NSLAB
$\dot{P}_{O_2,SLAB,N}$	-	PSLAB(1,N) [(kg of O <sub>2</sub> )/s]
$\dot{P}_{O_2,U,1}, \dot{P}_{O_2,U,2}$	-	UFLW2(1,3,2), UFLW2(2,3,2) [(kg of O <sub>2</sub> )/s]
$\dot{P}_{O_2,L,1}, \dot{P}_{O_2,L,2}$	-	UFLW2(1,3,1), UFLW2(2,3,1) [(kg of O <sub>2</sub> )/s]
$\dot{P}_{K,SLAB,N}$	-	PSLAB(K,N) [(unit of product K)/s]
$\dot{P}_{K,U,1}, \dot{P}_{K,U,2}$	-	UFLW2(1,2+K,2), UFLW2(2,2+K,2) [(unit of product K)/s]
$\dot{P}_{K,L,1}, \dot{P}_{K,L,2}$	-	UFLW2(1,2+K,1), UFLW2(2,2+K,1) [(unit of product K)/s]
$\dot{Q}_{SLAB,N}$	-	QSLAB(N) [W]
$\dot{Q}_{U,1}, \dot{Q}_{U,2}$	-	UFLW2(1,2,2), UFLW2(2,2,2) [W]
$\dot{Q}_{L,1}, \dot{Q}_{L,2}$	-	UFLW2(1,2,1), UFLW2(2,2,1) [W]
$\dot{Q}_{SU,1}, \dot{Q}_{SU,2}$	-	QSUVNT(1), QSUVNT(2) [W]
$\dot{Q}_{SL,1}, \dot{Q}_{SL,2}$	-	QSLVNT(1), QSLVNT(2) [W]
$T_{SLAB,N}$	-	TSLAB(N) [K]
$T_{U,1}, T_{U,2}$	-	TU(1), TU(2) [K]
$T_{L,1}, T_{L,2}$	-	TL(1), TL(2) [K]
$Y_{CEIL,1}, Y_{CEIL,2}$	-	YCEIL(1), YCEIL(2) [m]
$Y_{FLOOR,1}, Y_{FLOOR,2}$	-	YFLOR(1), YFLOR(2) [m]
$Y_{LAYER,1}, Y_{LAYER,2}$	-	YLAY(1), YLAY(2) [m]
$Y_{SLAB,1,N}, Y_{SLAB,2,N}$	-	YSLAB1(N), YSLAB2(N) [m]
$\Delta T_{EPS}$	-	TEPS [K]
$\lambda_v$	-	RLAM



PREPARED BY

Leonard Y. Cooper and Glenn P. Forney  
March 1988




Figure 1. Possible distributions of the cross-vent pressure-difference distribution,  $\Delta p_{1,2}$ , for a single natural vent and the flow directions of its one-to-six uniform-property flow slabs.






Figure 3. Flow Diagram for FLOG02 Adjustments to  $\dot{G}_{U,I}$ ,  $\dot{G}_{L,I}$ ,  $\dot{M}_{U,I}$ ,  $\dot{M}_{L,I}$ ,  $\dot{P}_{O2,U,I}$ ,  $\dot{P}_{O2,L,I}$ ,  $\dot{P}_{K,U,I}$ ,  $\dot{P}_{K,L,I}$ ,  $\dot{Q}_{U,I}$ ,  $\dot{Q}_{L,I}$ ,  $\dot{Q}_{SU,I}$ ,  $\dot{Q}_{SL,I}$  in the Receiving Room I Due to a Vent Flow Slab  
(Page 1 of 2)




Figure 3. Flow Diagram for FLOG02 Adjust. to  $\dot{G}_{U,I}$ ,  $\dot{G}_{L,I}$ ,  $\dot{M}_{U,I}$ ,  $\dot{M}_{L,I}$ ,  
(Page 2  $\dot{P}_{O2,U,I}$ ,  $\dot{P}_{O2,L,I}$ ,  $\dot{P}_{K,U,I}$ ,  $\dot{P}_{K,L,I}$ ,  $Q_{U,I}$ ,  $Q_{L,I}$ ,  $Q_{SU,I}$ ,  $Q_{SL,I}$  in the  
of 2) Receiving Room I Due to a Vent Flow Slab

## SUBROUTINE

```
      SUBROUTINE FLOGO2(  
        I DIRS12,YSLAB1,YSLAB2,XMSLAB,TSLAB,NSLAB,TU,TL,YFLOR,  
        I YCEIL,YLAY,QSLAB,RLAM,  
        I PSLAB,MXPRD,NPROD,MXSLAB,DTEPS,  
        O QSLVNT,QSUVNT,UFLW2)  
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)  
  
C*BEG  
C***  FLOGO2  GENERIC - DEPOSITION OF MASS, ENTHALPY, OXYGEN, AND OTHER  
C          PRODUCT-OF-COMBUSTION FLOWS PASSING BETWEEN TWO ROOMS  
C          THROUGH A VERTICAL, CONSTANT-WIDTH VENT OR THROUGH A  
C          SIMPLE FAN-RESISTANCE VENTILATION SYSTEM.  
C  
C***  SUBROUTINE ARGUMENTS  
C  
C    INPUT  
C    -----  
C    DIRS12 - A MEASURE OF THE DIRECTION OF THE ROOM 1 TO ROOM  
C              2 FLOW IN EACH SLAB  
C    YSLAB1 - SLAB HEIGHTS IN ROOM 1 ABOVE DATUM ELEVATION [M]  
C    YSLAB2 - SLAB HEIGHTS IN ROOM 1 ABOVE DATUM ELEVATION [M]  
C    XMSLAB - MASS FLOW RATE IN SLABS [KG/S]  
C    TSLAB  - ABSOLUTE TEMPERATURE OF THE FLOW IN SLABS [K]  
C    NSLAB  - NUMBER OF SLABS BETWEEN BOTTOM AND TOP OF VENT  
C    TU     - UPPER LAYER TEMPERATURE IN EACH ROOM [K]  
C    TL     - LOWER LAYER TEMPERATURE IN EACH ROOM [K]  
C    YFLOR  - HEIGHT OF FLOOR IN EACH ROOM ABOVE DATUM ELEVATION [M]  
C    TEPS   - MINIMUM TEMPERATURE DIFFERENCE REQUIRED TO INITIATE  
C              GROWTH OF A NEW LAYER [K]  
C    YCEIL  - HEIGHT OF CEILING IN EACH ROOM ABOVE DATUM ELEVATION [M]  
C    YLAY   - HEIGHT OF LAYER IN EACH ROOM ABOVE DATUM ELEVATION [M]  
C    QSLAB  - ENTHALPY FLOW RATE IN EACH SLAB [W]  
C    RLAM   - FRACTION OF SOURCE SLABS' ENTHALPY INFLOW LOST  
C              TO BOUNDING WALLS  
C    PSLAB  - FLOW RATE OF PRODUCT IN EACH SLAB [(UNIT OF PRODUCT/S)  
C    MXPRD  - MAXIMUM NUMBER OF PRODUCTS CURRENTLY AVAILABLE.  
C    NPROD  - NUMBER OF PRODUCTS  
C    MXSLAB - MAXIMUM NUMBER OF SLABS CURRENTLY AVAILABLE.  
C  
C    OUTPUT  
C    -----  
C    QSLVNT - TOTAL RATE OF HEAT TRANSFER TO LOWER BOUNDING SURFACES  
C              DUE TO ALL SLAB FLOWS OF VENT [W]  
C    QSUVNT - TOTAL RATE OF HEAT TRANSFER TO UPPER BOUNDING SURFACES  
C              DUE TO ALL SLAB FLOWS OF VENT [W]  
C    UFLW2(I,1,J), I=1 OR 2, J=1 OR 2 - MASS FLOW RATE TO UPPER  
C              (J=2) OR LOWER (J=1) LAYER OF ROOM I DUE TO ALL SLAB  
C              FLOWS OF VENT [KG/S]  
C    UFLW2(I,2,J), I=1 OR 2, J=1 OR 2 - ENTHALPY FLOW RATE TO UPPER  
C              (J=2) OR LOWER (J=1) LAYER OF ROOM I DUE TO ALL SLAB  
C              FLOWS OF VENT [W]
```

```

C   UFLW2(I,3,J), I=1 OR 2, J=1 OR 2 - OXYGEN FLOW RATE TO UPPER (J=2)
C       OR LOWER (J=1) LAYER OF ROOM I DUE TO ALL SLAB FLOWS OF
C       VENT [(KG OXYGEN)/S]
C   UFLW2(I,3+K,J), I=1 OR 2, K=2 TO NPROD, J=1 OR 2 - PRODUCT K FLOW
C       RATE TO UPPER (J=2) OR LOWER (J=1) LAYER OF ROOM I DUE
C       TO ALL SLAB FLOWS OF VENT [(UNIT PRODUCT K)/S]
C*END
C
C*** DEFINE INDEXES FOR USE WITH PRODUCT ARRAYS
C
C       INTEGER L,U,M,Q,O,P
C       PARAMETER (L=1,U=2,M=1,Q=2,O=3,P=3)
C
C       INTEGER DIRS12(*)
C       DIMENSION YSLAB1(*),YSLAB2(*),XMSLAB(*),TSLAB(*)
C       DIMENSION QSLAB(*)
C       DIMENSION TU(*),TL(*),YFLOR(*),YCEIL(*),YLAY(*)
C       DIMENSION PSLAB(MXSLAB,*)
C       DIMENSION QSUVNT(*),QSLVNT(*)
C       DIMENSION UFLW2(2,MXPRD+2,2),FF(2)
C
C*** INITIALIZE OUTPUTS
C
C       DO 10 I = 1, 2
C           DO 15 IPROD = 1, NPROD+2
C               UFLW2(I,IPROD,L) = 0.0D0
C               UFLW2(I,IPROD,U) = 0.0D0
C       15    CONTINUE
C           QSUVNT(I) = 0.0D0
C           QSLVNT(I) = 0.0D0
C       10    CONTINUE
C
C*** PUT EACH SLAB FLOW INTO APPROPRIATE LAYER OF ROOM ITO
C       AND TAKE SLAB FLOW OUT OF APPROPRIATE LAYER OF ROOM IFROM
C
C       DO 20 N = 1, NSLAB
C
C*** DETERMINE WHAT ROOM FLOW IS COMING FROM
C
C       IF(DIRS12(N).EQ.1)THEN
C           IFROM = 1
C           ITO = 2
C       ELSE IF(DIRS12(N).EQ.0)THEN
C
C
C*** NO FLOW IN THIS SLAB SO WE CAN SKIP IT
C
C       GO TO 20
C       ELSE IF(DIRS12(N).EQ.-1)THEN
C           IFROM = 2
C           ITO = 1
C       ENDIF
C

```

C\*\*\* APPORTION FLOW BETWEEN LAYERS OF DESTINATION ROOM

C

```
Y_LAYER = Y_LAYER(ITO)
Y_LOW = Y_FLOR(ITO)
Y_UP = Y_CEIL(ITO)
TTS = TSLAB(N)
TTU = TU(ITO)
TTL = TL(ITO)
IF(ITO.EQ.1)THEN
  YSLABT = YSLAB1(N)
  YSLABF = YSLAB2(N)
ELSE
  YSLABT = YSLAB2(N)
  YSLABF = YSLAB1(N)
ENDIF
```

C

C\*\*\* NO UPPER LAYER

C

```
IF(Y_LAYER.GE.Y_UP)THEN
  IF(TTS.GT.TTL+1.0000)THEN
    FU = 1.0000
    TTR = TTL
  ELSE
    FU = 0.0000
    TTR = TTL
  ENDIF
  GO TO 100
ENDIF
```

C

C\*\*\* NO LOWER LAYER

C

```
IF(Y_LAYER.LE.Y_LOW)THEN
  IF(TTS.GE.TTU+1.0000)THEN
    FU = 1.0000
    TTR = TTU
  ELSE
    FU = 0.0000
    TTR = TTU
  ENDIF
  GO TO 100
ENDIF
```

C

C\*\*\* UPPER LAYER TEMP > LOWER TEMP

C

```
IF(TTU.GT.TTL)THEN
```

C

C\*\*\* DETERMINE FU

C

```
IF(TTS.GE.TTU)THEN
  FU = 1.0000
ELSE IF(TTS.LE.TTL)THEN
  FU = 0.0000
```



```

        ELSE
            FU = (TTS - TTL)/(TTU - TTL)
        ENDIF
C
C*** DETERMINE TTR
C
        IF(YSLABT.GT.YLAYER)THEN
            TTR = TTU
        ELSE IF(YSLABT.LT.YLAYER)THEN
            TTR = TTL
        ELSE
            TTR = (TTU+TTL)/2.0D0
        ENDIF
        GO TO 100
C
C*** UPPER LAYER TEMP - LOWER TEMP
C
        ELSE IF(TTU.LE.TTL)THEN
        IF(TTS.GT.TTU)THEN
            FU = 1.0D0
            TTR = TTU
        ELSE IF(TTS.LT.TTU)THEN
            FU = 0.0D0
            TTR = TTL
        ELSE
            IF(YSLABT.GT.YLAYER)THEN
                FU = 1.0D0
                TTR = TTU
            ELSE IF(YSLABT.LT.YLAYER)THEN
                FU = 0.0D0
                TTR = TTL
            ELSE
                FU = .50D0
                TTR = TTL
            ENDIF
            GO TO 100
        ENDIF
    ENDIF
100 CONTINUE
    FL = 1.0D0 - FU
    FF(L) = FL
    FF(U) = FU
    FQ = RLAM*(1.0D0 - TTR/TTS)
    FS = 1.0D0 - FQ
    XMTERM = XMSLAB(N)
    QTERM = QSLAB(N)
    QSUVNT(ITO) = QSUVNT(ITO) + FU*QTERM*FQ
    QSLVNT(ITO) = QSLVNT(ITO) + FL*QTERM*FQ
    DO 110 ILAY = 1, 2
        UFLW2(ITO,M,ILAY) = UFLW2(ITO,M,ILAY) + FF(ILAY)*XMTERM
        UFLW2(ITO,Q,ILAY) = UFLW2(ITO,Q,ILAY) + FF(ILAY)*QTERM*FS
    DO 120 IPROD = 1, NPROD

```

```

        IP = P + IPROD - 1
        UFLW2(ITO,IP,ILAY) = UFLW2(ITO,IP,ILAY)
1      + FF(ILAY)*PSLAB(N,IPROD)
120    CONTINUE
110    CONTINUE
C
C*** CASE 2A  FLOW ENTIRELY FROM UPPER LAYER OF ROOM IFROM
C
        IF(YSLABF.GE.YLAY(IFROM))THEN
            UFLW2(IFROM,M,U) = UFLW2(IFROM,M,U) - XMTERM
            UFLW2(IFROM,Q,U) = UFLW2(IFROM,Q,U) - QTERM
            DO 130 IPROD = 1, NPROD
                IP = P + IPROD - 1
                UFLW2(IFROM,IP,U) = UFLW2(IFROM,IP,U) - PSLAB(N,IPROD)
130        CONTINUE
            ELSE
C
C*** CASE 3A  FLOW ENTIRELY FROM LOWER LAYER OF ROOM IFROM
C
            UFLW2(IFROM,M,L) = UFLW2(IFROM,M,L) - XMTERM
            UFLW2(IFROM,Q,L) = UFLW2(IFROM,Q,L) - QTERM
            DO 140 IPROD = 1, NPROD
                IP = P + IPROD - 1
                UFLW2(IFROM,IP,L) = UFLW2(IFROM,IP,L) - PSLAB(N,IPROD)
140        CONTINUE
            ENDIF
20    CONTINUE
        RETURN
        END

```

PLUGO - Calculation of Mass, Enthalpy, Oxygen and Other Product of  
Combustion Flows to Upper and Lower Layers Due to a Fire-Driven  
Plume

DESCRIPTION

Consider an instant of time during a two-layer zone-type simulation of a room fire environment. This algorithm calculates the rates of flow of mass, enthalpy, oxygen and other products of combustion to the upper and lower layer of the room due directly to a fire in the room and its fire-driven plume [1]. The fire room is depicted in Figure 1.

The total energy-release rate of the fire,  $\dot{Q}$ , is specified. It is assumed that the combustion zone and the plume radiate at an effective rate  $\lambda_R \dot{Q}$ , where  $\lambda_R$  is specified, and that

$$\dot{Q}_{\text{SOURCE}} = (1 - \lambda_R) \dot{Q} \quad (1)$$

is the part of the fire's energy-release-rate which drives the convective buoyant fire plume upward toward the ceiling.

The fire is assumed to be a source of oxygen,  $O_2$ , and of  $(N_{\text{PROD}} - 1)$  other user-specified products of combustion which are taken account of in the simulation. Following the rule of [2] it is assumed that the rate of  $O_2$  generation consumed in the oxidation process,  $\dot{P}_{O_2, \text{SOURCE}}$ , is negative (i.e., oxygen is consumed) and directly proportional to  $\dot{Q}$ .

$$\text{rate of } O_2 \text{ generation} = \dot{P}_{O_2, \text{SOURCE}} = -F \cdot \dot{Q} \quad (2)$$

$$F = 0.076(10^{-6}) [(\text{kg } O_2)/(\text{W} \cdot \text{s})] \quad (3)$$

Note that oxygenated combustibles would have additional contributions to the rate of  $O_2$  generation.

It is assumed that the other products of combustion are generated at specified rates  $\dot{P}_{K, \text{SOURCE}}$ ,  $K = 2$  to  $N_{\text{PROD}}$  [(unit of product K)/s]. Typically, the rate at which mass is generated in the combustion process is negligible compared to other mass flow rates of interest in the simulation, and this is assumed to be the case here.

If the lower layer is of non-zero thickness and if the fire is in the lower layer, then it is assumed that a complete description of the lower-layer room environment at the instant of time of interest is known. This includes the density, absolute temperature, concentration-by-mass of  $O_2$  [(kg of  $O_2$ )/(kg layer)], and concentrations-by-mass of the other  $N_{\text{PROD}}$  products of combustion [(unit of product k)/(kg layer)]. These are designated by  $\rho_L$ ,  $T_L$ ,  $c_{O_2, L}$ , and  $c_{K, L}$ ,  $K = 2$  to  $N_{\text{PROD}}$ , respectively. The  $O_2$  and other product concentrations

would be required only when these concentrations were actually being predicted.

Referring to Figure 1, the elevation of the interface separating the upper and lower layers is  $y_{\text{LAYER}}$ . The fire is located at elevation,  $y_{\text{FIRE}}$ , which is assumed to be between or at the elevation of the room's floor and ceiling,  $y_{\text{FLOOR}}$  and  $y_{\text{CEIL}}$ , respectively. Consistent with the idealized point-source plume model of [3], it is appropriate to choose  $y_{\text{FIRE}}$  as the lowest elevation above the base of the combustion zone where relatively-free, unrestricted, lateral entrainment of the local environment into the combustion zone or plume is possible.

It is assumed that the total fraction of  $\dot{Q}$  which is lost to the bounding surfaces of the room by all modes of heat transfer is given by  $\lambda_T \dot{Q}$ , where  $\lambda_T$  is specified. The remaining portion of  $\dot{Q}$  is  $\dot{Q}_{\text{TOTAL}}$ , i.e.,

$$\dot{Q}_{\text{TOTAL}} = (1 - \lambda_T) \dot{Q} \quad (4)$$

As explained in [1], except certain special configurations this is assumed to be deposited entirely in the upper layer. When the fire is in the upper layer this is all added to the upper layer directly and there is no other enthalpy flow due to the fire. When the fire is in the lower layer, the net rate of enthalpy flow into the upper layer due to the fire is  $\dot{Q}_{\text{TOTAL}}$  plus the rate-of-enthalpy flow associated with the gases entrained into the plume from the lower layer and deposited by the plume into the upper layer.

The algorithm assumes no gas-to-room-surface losses of products of combustion.

#### OUTPUT

$$\dot{M}_U [\dot{M}_L]$$

Rate of mass flow to the upper [lower] layer of the room due to the fire-driven buoyant plume. [kg/s]

$$\dot{P}_{O_2,U} [\dot{P}_{O_2,L}]$$

Rate of  $O_2$  mass flow to the upper [lower] layer of the room due to the fire-driven buoyant plume. [(kg of  $O_2$ )/s]

$$\dot{P}_{K,U} [\dot{P}_{K,L}]; K = 2 \text{ to } N_{\text{PROD}}$$

Rate of flow of product of combustion K to the upper [lower] layer of the room due to the fire-driven buoyant plume. [(unit of product K)/s]

$\dot{Q}_U$  [ $\dot{Q}_L$ ]

Rate of enthalpy flow, based on absolute temperature, to the upper [lower] layer of the room due to the fire-driven buoyant plume. [W]

$\dot{Q}_{TOTAL}$

Net rate of enthalpy flow to the combined upper and lower layers,  $(1 - \lambda_T)\dot{Q}$ . [W]

#### INPUT

$C_p$

Specific heat at constant pressure of the layers (suggest using  $10^3 \text{ W}\cdot\text{s}/(\text{kg}\cdot\text{K})$  which is  $C_p$  of air). [ $\text{W}\cdot\text{s}/(\text{kg}\cdot\text{K})$ ]

$c_{O_2,L}$

Concentration of  $O_2$  in the lower layer. [(kg of  $O_2$ )/(kg of layer)]

$c_{K,L}$ ;  $K = 2$  to  $N_{PROD}$

Concentration of product of combustion  $K$  in the lower layer. [(unit of product  $K$ )/(kg of layer)]

$N_{PROD}$

Number of products of combustion, including oxygen, being tracked in the simulation.

$\dot{P}_{K,SOURCE}$ ;  $K = 2$  to  $N_{PROD}$

Rate of product of combustion  $K$  generated by the fire source. [(unit of product  $K$ )/s]

$\dot{Q}$

Total rate of energy release of the fire. [W]

$T_L$

Absolute temperature of the lower layer. [K]

$y_{CEILING}/y_{LAYER}/y_{FIRE}$

Elevation of the ceiling/layer/fire above the datum elevation. [m]

$\Delta T_{EPS}$

A small temperature difference, between the average plume temperature at the ceiling and  $T_L$ , below which it is assumed that an upper layer will not grow from zero thickness. [K]

$\lambda_R$

Fraction of  $Q$  radiated by the combustion zone and plume,  $0 \leq \lambda_R < 1$ .

$\lambda_T$

Total fraction of  $\dot{Q}$  lost to the bounding surfaces of the room by all modes of heat transfer,  $0 \leq \lambda_T < 1$ .

$\rho_L$

density of the lower layer. [kg/m<sup>3</sup>]

#### CALCULATION

The calculations are presented in the flow diagram of Figure 2.

#### REFERENCES

- [1] Cooper, L.Y. and Forney, G.P., Section 3.2 of Consolidated Compartment Fire Model (CCFM) Computer Code Application CCFM.VENTS - Part I: Physical Basis, NISTIR 90-4342, National Institute of Standards and Technology, Gaithersburg MD, 1990.
- [2] Huggett, C., Estimation of Rate of Heat Release by Means of Oxygen Consumption Measurements, Fire and Materials, Vol. 4, No. 2, (1980).

- [3] Zukoski, E.E., Kubota, T., and Cetegen, B., Entrainment in Fire Plumes, Fire Safety Journal, Vol. 3, pp. 107-121 (1980/81).

# SUBROUTINE VARIABLES

Nomenclature used above correspond to the following nomenclature used in the subroutine:

$C_p$	-	CP [W·s/(kg·K)]
FACTOR	-	0.076(10 <sup>-6</sup> ) [(kg of O <sub>2</sub> )/(W·s)]
g	-	9.8 [m/s <sup>2</sup> ]
$\dot{M}_L$ , $\dot{M}_U$	-	ML, MU [kg/s]
$\dot{M}_{PLUME}$	-	MPLUM [kg/s]
$\dot{P}_{K, SOURCE}$	-	PSOURC(K) [(unit of product K)/s]
$\dot{Q}_L$ , $\dot{Q}_U$	-	QL, QU [W]
$\dot{Q}_{SOURCE}$	-	QSOURC [W]
$\dot{Q}_{TOTAL}$	-	QTOT [W].
$T_L$	-	TL [K]
Y <sub>CEILING</sub>	-	YCEIL [m]
Y <sub>FIRE</sub>	-	YFIRE [m]
Y <sub>LAYER</sub>	-	YLAY [m]
Z	-	Z [m]
$\Delta T_{EPS}$	-	DTEPS [K]
$\Delta T_{PLUME}$	-	DT [K]
$\rho_L$	-	RHOL [kg/m <sup>3</sup> ]
$\lambda_R$ , $\lambda_T$	-	ALAMR, ALAMT

# PREPARED BY

Leonard Y. Cooper and Glenn P. Forney  
March 1988




Figure 1. Fire environment in a room of fire origin






Figure 2. Flowchart of PLUGO calculations

# SUBROUTINE PLUGO

```
      SUBROUTINE PLUGO(
      I CP,CL,NPROD,PSOURC,Q,TL,TU,
      I YCEIL,YLAY,YFIRE,DTEPS,ALAMR,ALAMT,RHOL,
      O XMU,XML,PU,PL,QU,QL,QTOT)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)

C*BEG
C
C***  PLUGO  GENERIC - ROUTINE TO CALCULATE THE MASS, ENTHALPY AND PRODUCT
C          FLOWS DUE TO ONE PARTICULAR PLUME.
C
C***  SUBROUTINE ARGUMENTS
C
C  INPUT
C  -----
C  CP      SPECIFIC HEAT [W*S/(KG*K)]
C  CL      CONCENTRATION OF EACH PRODUCT IN LOWER LAYER
C           [UNIT OF PRODUCT/(KG LAYER)]
C  NPROD   NUMBER OF PRODUCTS OF COMBUSTION
C  PSOURC  AMOUNT OF PRODUCT RELEASED (OR ABSORBED) BY THE FIRE
C  Q        ENERGY RELEASE RATE OF FIRE [W]
C  TL      TEMPERATURE OF LOWER LAYER
C  TU      TEMPERATURE OF UPPER LAYER
C  YCEIL   HEIGHT OF CEILING ABOVE DATUM ELEVATION [M]
C  YLAY    HEIGHT OF LAYER ABOVE DATUM ELEVATION [M]
C  YFIRE   HEIGHT OF FIRE ABOVE DATUM ELEVATION [M]
C  DTEPS   MINIMUM TEMPERATURE OF FLOW THAT WILL CAUSE FORMATION
C           OF AN UPPER LAYER [K]
C  ALAMR   FRACTION OF Q RADIATED BY THE COMBUSTION ZONE AND PLUME
C  ALAMT   TOTAL FRACTION OF Q LOST TO THE BOUNDING SURFACES OF THE ROOM
C           BY ALL MODES OF HEAT TRANSFER
C  RHOL    DENSITY OF LOWER LAYER [KG/M**3]
C
C  OUTPUT
C  -----
C  XMU     MASS FLOW RATE INTO UPPER LAYER
C  XML     MASS FLOW RATE INTO LOWER LAYER
C  PU      PRODUCT FLOW RATE INTO UPPER LAYER
C  PL      PRODUCT FLOW RATE INTO LOWER LAYER
C  QU      ENTHALPY FLOW RATE INTO UPPER LAYER
C  QL      ENTHALPY FLOW RATE INTO LOWER LAYER
C  QTOT    TOTAL ENTHALPY FLOW RATE
C
C
C*END

      PARAMETER (G=9.8D0,FACTOR=.076D-6)
      DIMENSION PU(*), PL(*), PSOURC(*), CL(*)

C
      QTOT = (1.0D0 - ALAMT)*Q
      QSRC = (1.0D0 - ALAMR)*Q
      Z = YLAY - YFIRE
      IF(NPROD.GE.1)PSOURC(1) = -FACTOR*Q
```

```

      IF(Z.LT.0.0D0.OR.(Z.EQ.0.0D0.AND.YFIRE.LT.YCEIL))THEN
C
C   FIRE IN UPPER LAYER; NO PLUME CALCULATIONS REQUIRED;
C   NO CHANGE TO LOWER LAYER
C
      XML = 0.0D0
      QL = 0.0D0
      XMU = 0.0D0
      DO 10 IPROD = 1, NPROD
        PL(IPROD) = 0.0D0
        PU(IPROD) = PSOURC(IPROD)
10    CONTINUE
      QU = QTOT
      RETURN
    ENDIF
    IF(Z.EQ.0.0D0.AND.YFIRE.GE.YCEIL)GO TO 300
C
C   FIRE IN LOWER LAYER; AND BELOW INTERFACE;
C   USE PLUME CALCULATIONS FOR XMPLUM, TPLUME
C
      GZ = (G*Z**5)**.5D0
      QSTAR = QSRC/(RHOL*TL*CP*GZ)
      XMPLUM = .21D0*RHOL*GZ*QSTAR**(1.0D0/3.0D0)
      IF(YCEIL.LE.YLAY.AND.XMPLUM.NE.0.0D0)THEN
        DTPLUM = QSRC/(CP*XMPLUM)
        IF(DTPLUM.LT.DTEPS)GO TO 300
      ENDIF
      XMU = XMPLUM
      DO 20 K = 1, NPROD
        PU(K) = CL(K)*XMPLUM + PSOURC(K)
        PL(K) = -CL(K)*XMPLUM
20    CONTINUE
      QU = CP*XMPLUM*TL + QTOT
      XML = -XMPLUM
      QL = -CP*XMPLUM*TL
      RETURN
300 CONTINUE
C
C   NO CHANGE TO UPPER LAYER SINCE FIRE IS IN LOWER LAYER AND:
C   1) IT IS IMMEDIATELY BELOW BOTH CEILING AND INTERFACE, OR
C   2) INTERFACE IS AT CEILING AND PLUME IS TOO WEAK TO START NEW LAYER
C
      XMU = 0.0D0
      QU = 0.0D0
      DO 30 K = 1, NPROD
        PL(K) = PSOURC(K)
        PU(K) = 0.0D0
30    CONTINUE
      XML = 0.0D0
      QL = QTOT
      RETURN
    END

```



**VENTF - Calculate Flow Through a Fan/Duct Forced Ventilation System Between Two Rooms**

**DESCRIPTION**

Consider an instant of time during the two-layer, zone-type simulation of a multi-room compartment fire environment. This algorithm, VENTF, calculates the direction and rate of flow of mass, enthalpy, oxygen and products of combustion through a fan/duct forced ventilation system connecting any two rooms, or a room and the outside environment [1]. The calculation uses a separate algorithm, FANRES [2], which performs the system flow simulation in cases where the flow rate is not specified.

Designate the two rooms or the room and the outside environment as rooms 1 and 2 and refer to the Figure 1. Define room 1 [room 2] as the room connected to the system inlet [outlet]. Here, the system inlet [outlet] is defined as the side of the fan/duct flow system which corresponds to the flow inlet [outlet] under normal fan operating conditions. Normal operating conditions lead to what will be referred to here as flow in the normal or forward direction.

The elevation of the inlet and outlet end-point of the fan/duct system are characterized by the elevation of their midpoints,  $y_{\text{SYSTEM}, I}$ , in room  $I = 1$  or 2, above a datum elevation. The vent openings at the duct end-points are assumed to be rectangular with sides aligned vertically and horizontally.  $H_I$ , the depth of these vents in room  $i$ , are also specified. These are required in order to determine the effect on the fire environment of the system flow at instants of time when flow through the system is drawn partially from, or supplied partially to both the upper and lower layer of room  $I$ .

A complete characterization of the instantaneous environment in each of the two rooms is provided as input. Using this, the algorithm first computes the densities of the flow in the ventilation system duct near its entry region under the conditions of either forward flow (flow enters at the designed system inlet) or backward flow (flow enters at the designed system outlet). Using these two densities, certain specified characteristics of the fan/duct system, and FANRES [2], the algorithm estimates the instantaneous direction of flow through the system, indicated by the parameter  $I_{\text{DIR}}$ , and the mass-flow rate,  $\dot{M}_{\text{DUCT}}$ . Assuming an adiabatic duct flow, the algorithm then calculates: the temperature of the duct flow,  $T_{\text{DUCT}}$ , and the rates of flow through the duct of enthalpy, oxygen, and  $N_{\text{PROD}} - 1$  other products of combustion,  $\dot{Q}_{\text{DUCT}}$ ,  $\dot{P}_{\text{DUCT}, \text{O}_2}$ ,  $\dot{P}_{\text{DUCT}, K}$ ,  $K = 2$  to  $N_{\text{PROD}}$ , respectively. Finally, the algorithm computes: the number and rates of flow of uniform-property flow slabs leaving the source room (one or two flow slabs depending on whether the flow is extracted from one or both of the layers) and the number and rates of flow of identically-uniform-property flow slabs entering the receiving room. If there is only one of these latter slabs entering the receiving room, then it enters the upper or lower layer. If there are two such flow slabs, then one enters the upper layer and the other the lower layer.

The specified characteristics of the fan/duct system include pairs of values,

$[\dot{V}_{F,J}, \Delta p_{F,J}]$ ,  $J = 1$  to  $JTERM$ , which identify points on the fan operating curve.

If  $JTERM = 1$ , then the single  $\dot{V}_{F,1}$  input is a specified volume-flow-rate. Under these circumstances, the corresponding  $\Delta p_{F,1}$  input is not required and FANRES is not used.

If  $JTERM > 1$ , then the specified  $[\dot{V}_{F,J}, \Delta p_{F,J}]$  values indicate the volume-flow-rate delivery of the fan,  $\dot{V}_{F,J}$ , at discrete cross-fan pressure differences,  $\Delta p_{F,J}$ . Referring to Figure 2,  $\Delta p_{F,J}$  is defined as the cross-fan pressure difference,  $\Delta p_F$ , at point  $J$  on the fan curve where

$$\Delta p_F = \delta p_{F,2} - \delta p_{F,1} \quad (1)$$

and where  $\delta p_{F,1}$  and  $\delta p_{F,2}$  are the pressures that would be measured at the fan inlet and outlet sections, respectively, relative to the datum pressure. In specifying the  $[\dot{V}_{F,J}, \Delta p_{F,J}]$  data, the  $\Delta p_{F,J}$  are required to be monotonically decreasing with increasing  $\dot{V}_{F,J}$ . A sketch of such a generic fan operating curve is presented in Figure 2. In the calculations, the fan curve is approximated by linear interpolation between and linear extrapolation beyond the fan-curve data points.

If  $JTERM > 1$ , then effective flow resistances of the duct system,  $R_1$  and  $R_2$  for forward and backward flow, respectively, must be specified. (The definitions for the  $R_i$ 's are presented in the list of inputs.) Finally, the midpoint elevation of the two vents at the system endpoints are specified. Further discussion on the fan/duct system characteristics and the basis of the FANRES algorithm [2] are presented in [3].

As mentioned above, flow through the system is determined by the instantaneous environments local to the vents at the two system endpoints. As depicted in Figure 1 it is assumed that these two environments are described within the context of a two-layer model of the rooms' fire environments. In each of the rooms  $I$ ,  $I = 1$  or  $2$ , VENTF computes the local environment from specified values of: pressure at the floor elevation,  $\delta p_{FLOOR,I}$ , relative to a datum pressure; elevations of the floor and smoke layer interface,  $y_{FLOOR,I}$  and  $y_{LAYER,I}$ , above a datum elevation; absolute temperature and density of the upper and lower layer,  $T_{U,I}$ ,  $T_{L,I}$  and  $\rho_{U,I}$ ,  $\rho_{L,I}$ , respectively; and concentrations of oxygen and products of combustion in the upper and lower layer,  $c_{U,O2,I}$ ,  $c_{L,O2,I}$  and  $c_{U,K,I}$ ,  $c_{L,K,I}$ ,  $K = 2$  to  $N_{PROD}$ , respectively.

As discussed in [1] and [2], at some instants of time in a fire scenario the existence of two solutions to the fan/duct flow equations, a forward- and backward-flow solution, is possible. However, at a particular instant of time in a simulation, it not known apriori whether or not there is a single or multiple solution. For this reason a final input variable,  $I_{DIROLD}$ , is required which identifies, in the event of a multiple solution, the preferred direction of the flow. When carrying out simulations of time-dependent fire environments it is reasonable to choose the preferred flow direction for each fan/duct system as the flow direction computed at the previously simulated instant of time (or, as the normal flow direction, if the previous flow rate is zero identically).

Characterizing the Entering Duct Flow When the Elevation of the Layer Interface is Between the Top and Bottom of the Vent Opening

When the layer interface is not between the top and bottom of the vent opening the properties of the entering duct flow are identical to the specified properties of the lower layer (if  $y_{\text{LAYER, IFR}} \geq y_{\text{SYSTEM, IFR}} + H_{\text{IFR}}/2$ ) or upper layer (if  $y_{\text{LAYER, IFR}} \leq y_{\text{SYSTEM, IFR}} - H_{\text{IFR}}/2$ ) where IFR, standing for IFROM, is the number (1 or 2) of the source room from which the flow enters the ventilation system. (ITO will be used below as the number of the receiving room into which the ventilation flow is deposited.) When the layer interface is between the top and bottom of the vent opening special considerations are required.

Figure 3 is an illustration of the flow entering the duct in room IFR when the layer interface is between the top and bottom of the vent opening. Define  $\delta p_{\text{DUCT}}$  as the pressure relative to the datum pressure at a location somewhat inside the duct, upstream of significant duct-flow development and mixing between the upper- and lower-layer flow contributions.

Assume that the flow through the vent and into the entry region of the duct is similar to the flow through an orifice or nozzle. Assume also that  $H_{\text{IFR}}$  is small enough so that the absolute value of the hydrostatic pressure difference over the height of the duct is small compared to the absolute value of the pressure drop,  $|\Delta p_{\text{ENTRY}}| = |\delta p_{\text{DUCT}} - \delta p_{\text{SYSTEM, IFR}}|$ , across the duct entrance region. Then the velocity of the upper and lower layer flow-slab components of the flow in the entrance region of the duct are both proportional to  $|\Delta p_{\text{ENTRY}}|^{1/2}$ . In particular, the mass-flow-rate of the two flow slabs from the upper and lower layer,  $\dot{M}_{\text{SLABU, IFR}}$  and  $\dot{M}_{\text{SLABL, IFR}}$ , respectively, are given by

$$\begin{aligned}\dot{M}_{\text{SLABU, IFR}} &= b C_D h_{\text{U, IFR}} (|\Delta p_{\text{ENTRY}}| \rho_{\text{U, IFR}})^{1/2} \\ \dot{M}_{\text{SLABL, IFR}} &= b C_D h_{\text{L, IFR}} (|\Delta p_{\text{ENTRY}}| \rho_{\text{L, IFR}})^{1/2}\end{aligned}\quad (2)$$

In the above,  $b$  is the width of the vent and  $C_D$  is the flow coefficient. Also, as illustrated in Figure 3,  $h_{\text{U, I}}$  [ $h_{\text{L, I}}$ ] is the absolute value of the difference in elevation in room I between the top [bottom] of the vent and the layer interface.

$$\begin{aligned}h_{\text{U, I}} &= y_{\text{SYSTEM, I}} + H_{\text{I}}/2 - y_{\text{LAYER, I}} \\ h_{\text{L, I}} &= H_{\text{I}} - h_{\text{U, I}} = y_{\text{LAYER, I}} - (y_{\text{SYSTEM, I}} - H_{\text{I}}/2)\end{aligned}\quad (3)$$

From Eq. (2) it follows that

$$\dot{M}_{\text{SLABU, IFR}} / \dot{M}_{\text{SLABL, IFR}} = (h_{\text{U, IFR}} / h_{\text{L, IFR}}) (\rho_{\text{U, IFR}} / \rho_{\text{L, IFR}})^{1/2}\quad (4)$$

Assume that once in the duct the two streams mix adiabatically to form a uniform-property duct flow of temperature,  $T_{\text{DUCT}}$ , and density,  $\rho_{\text{DUCT}}$ , where

the upper- and lower-layer inlet flows can all be modeled as the same perfect gas, i.e.,

$$\rho_{DUCT} T_{DUCT} = \rho_{U, IFR} T_{U, IFR} = \rho_{L, IFR} T_{L, IFR} \quad (5)$$

Adiabatic mixing requires

$$\begin{aligned} \dot{M}_{SLABU, IFR} T_{U, IFR} + \dot{M}_{SLABL, IFR} T_{L, IFR} &= |\dot{M}_{DUCT}| T_{DUCT} \\ &= (\dot{M}_{SLABU, IFR} + \dot{M}_{SLABL, IFR}) T_{DUCT} \end{aligned} \quad (6)$$

Now Eqs. (4)-(6) lead to

$$\begin{aligned} \rho_{DUCT} = \rho_{DUCT, IFR} = \rho_{L, IFR} [1 + (h_{U, IFR}/h_{L, IFR})(\rho_{U, IFR}/\rho_{L, IFR})^{1/2}] / \\ [1 + (h_{U, IFR}/h_{L, IFR})(\rho_{L, IFR}/\rho_{U, IFR})^{1/2}] \end{aligned} \quad (7)$$

where  $\rho_{DUCT, IFR}$  is the density of the duct flow when the source room is room IFR. Thus, if the flow direction (i.e., the value of IFR) is known, the above model for flow entry into the duct yields an Eq.-(7) estimate of  $\rho_{DUCT, IFR}$  without specific knowledge of the value of  $\dot{M}_{DUCT}$  or its entry flow-slab components,  $\dot{M}_{SLABU, IFR}$  and  $\dot{M}_{SLABL, IFR}$ . This result will be used below in the FANRES algorithm which requires values of  $\rho_{D, IFR}$  under possible conditions of both forward or backward flow through the ventilation system, i.e., for both possibilities IFR = 1 or 2.

With the solution  $\rho_{DUCT, IFR}$  for both possibilities IFR = 1 and 2, it is possible to obtain the correct solution for IFR and the correct associated value for the flow rate. If JTERM > 1, this will require use of the algorithm/subroutine FANRES [2].

With the correct value of IFR, the solution to  $\rho_{DUCT}$  is determined in Eq. (7). Then, Eq. (5) leads to  $T_{DUCT}$

$$T_{DUCT} = \rho_{U, IFR} T_{U, IFR} / \rho_{DUCT} \quad (8)$$

Once  $\dot{M}_{DUCT}$  is available, Eqs. (4), (6), and (8) lead to  $\dot{M}_{SLABU, IFR}$  and  $\dot{M}_{SLABL, IFR}$

$$\begin{aligned} \dot{M}_{SLABL, IFR} &= |\dot{M}_{DUCT}| T_{DUCT} / \{ T_{U, IFR} [(h_{U, IFR}/h_{L, IFR})(\rho_{U, IFR}/\rho_{L, IFR})^{1/2}] \\ &\quad + T_{L, IFR} \} \end{aligned} \quad (9)$$



$$\begin{aligned} \dot{M}_{\text{SLABU, IFR}} &= |\dot{M}_{\text{DUCT}}| - \dot{M}_{\text{SLABL, IFR}} \\ &= |\dot{M}_{\text{DUCT}}| T_{\text{DUCT}} / (T_{\text{L, IFR}} [(h_{\text{L, IFR}}/h_{\text{U, IFR}})(\rho_{\text{L, IFR}}/\rho_{\text{U, IFR}})^{1/2}] \\ &\quad + T_{\text{U, IFR}}) \end{aligned} \quad (10)$$

### The Flow in the Duct and the Flow Slabs Entering the Recieving Room

The above ideas lead to the solution to the duct-entry flow problem. With this, the assumed fully-mixed duct-flow condition leads first to a solution for all of the duct-flow properties, and then to a solution for the number (1 or 2) and properties of the flow slabs which enter the recieving room.

### OUTPUT

$I_{\text{DIR}}$

A measure of the direction of the room 1-to-room 2 system flow. For flow from room 1 to room 2 (normal inlet to normal outlet) or from room 2 to room 1 (normal outlet to normal inlet, i.e., a condition of fan backflow), the value of  $I_{\text{DIR}}$  is 1 or -1, respectively.  $I_{\text{DIR}}$  is 0 if the room-to-room flow is identically zero.

$\dot{M}_{\text{SLABF, N}}; N = 1 \text{ to } N_{\text{SLABF}} \leq 2$  [ $\dot{M}_{\text{SLABT, N}}; N = 1 \text{ to } N_{\text{SLABT}} \leq 2$ ]

Total mass flow rate in slab N leaving the source room, room IFR [entering the recieving room, room ITO]. [kg/s]

$\dot{M}_{\text{DUCT}}$

Mass flow rate through the duct. [kg/s]

$N_{\text{PROD}}$

Number of products of combustion being tracked in the simulation.

$N_{SLABF}$

Number of slabs of flow, 1 or 2, leaving the source room, room IFR, and entering the system duct. If  $N_{SLABF} = 1$  then the flow is totally from the upper or lower layer. If  $N_{SLABF} = 2$  then each layer supplies one of the slabs.

$N_{SLABT}$

Number of slabs of flow, 1 or 2, supplied to the receiving room, room ITO, and leaving the system duct. If  $N_{SLABT} = 1$  then all of the flow enters either the upper or lower layer. If  $N_{SLABT} = 2$  then one of the slabs will enter the upper layer and the other slab will enter the lower layer.

$\dot{P}_{DUCT,K}$ ,  $K = 2$  to  $N_{PROD}$

Product K flow rate through the duct. [(unit of product K)/s]

$\dot{P}_{DUCT,O_2}$

Mass flow rate of oxygen flowing through the duct. [(kg of oxygen)/s]

$\dot{P}_{SLABF,K,N}$ ;  $N = 1$  to  $N_{SLABF} \leq 2$ ,  $K = 2$  to  $N_{PROD}$   
[ $\dot{P}_{SLABT,K,N}$ ;  $N = 1$  to  $N_{SLABT} \leq 2$ ,  $K = 2$  to  $N_{PROD}$ ]

Product K flow rate in slab N leaving the source room, room IFR [entering the receiving room, room ITO]. [(unit of product K)/s]

$\dot{P}_{SLABF,O_2,N}$ ;  $N = 1$  to  $N_{SLABF} \leq 2$  [ $\dot{P}_{SLABT,O_2,N}$ ;  $N = 1$  to  $N_{SLABT} \leq 2$ ]

Oxygen flow rate in slab N leaving the source room, room IFR [entering the receiving room, room ITO]. [(kg of  $O_2$ )/s]

$\dot{Q}_{DUCT}$

Enthalpy flow rate through the duct based on absolute temperature  $T_{DUCT}$ . [W]

$\dot{Q}_{SLABF,N}; N = 1 \text{ to } N_{SLABF} \leq 2$  [ $\dot{Q}_{SLABT,N}; N = 1 \text{ to } N_{SLABT} \leq 2$ ]

Enthalpy flow rate in slab N leaving the source room, room IFR, based on the absolute temperature,  $T_{SLABF,N}$  [entering the receiving room, room ITO, based on the absolute temperature,  $T_{SLABT,N}$ ]. [W]

$T_{DUCT}$

Absolute temperature of the flow in the duct. [K]

$T_{SLABF,N}; N = 1 \text{ to } N_{SLABF} \leq 2$  [ $T_{SLABT,N}; N = 1 \text{ to } N_{SLABT} \leq 2$ ]

Absolute temperature of the flow in slab N leaving the source room, room IFR [entering the receiving room, room ITO]. [K]

$y_{SLABF,N}; N = 1 \text{ to } N_{SLABF} \leq 2$  [ $y_{SLABT,N}; N = 1 \text{ to } N_{SLABT} \leq 2$ ]

Elevations above the datum elevation of the midpoints of slab N leaving the source room, room IFR [entering the receiving room, room ITO]. [m]

## INPUT

ACC

Absolute error tolerance for the fan/duct system's flow rate calculation. This parameter is not used when the flow rate is specified, i.e., when  $JTERM = 1$ .

$c_{L,K,I} [c_{U,K,I}]; K = 2 \text{ to } N_{PROD}; I = 1 \text{ or } 2$

Concentration of product of combustion K in lower [upper] layer of room I. [(unit of product K)/(kg of layer)]

$c_{L,O2,I} [c_{U,O2,I}]; I = 1 \text{ or } 2$

Concentration of oxygen in lower [upper] layer of room I. [(kg of oxygen)/(kg of layer)]

$C_p$

Specific heat at constant pressure of the vent flow.  
[W·s/(kg·K)] (suggest  $10^3$  W·s/(kg·K) for air as  
default)

$I_{DIROLD}$

A measure of the preferred direction for the solution  
flow in the event of multiple flow solutions, i.e.,  
the preferred value for  $I_{DIR}$ .

$H_I$ ;  $I = 1$  or  $2$

Height of the vent in room  $I$ . Note that  $H_I$  must  
satisfy:  $H_I \leq 2(Y_{CEILING} - Y_{SYSTEM,I})$  and  $H_I \leq$   
 $2(Y_{SYSTEM,I} - Y_{FLOOR,I})$ . [m]

$JTERM \geq 1$

Number of pairs of input data [ $\dot{V}_{F,J}$ ,  $\Delta p_{F,J}$ ] used to  
approximate the fan curve.

$N_{PROD}$

Number of products of combustion being tracked in the  
simulation.

$R_1$  [ $R_2$ ]

(Required if  $JTERM > 1$ .) Effective resistance of the  
duct for forward [backward] flow. If  $\dot{M}_{DUCT,I}$  is the  
mass flow rate through the fan/duct system, then  $R_I$  is  
defined as

$$R_I = |\Delta p_F + (\delta p_{SYSTEM,1} - \delta p_{SYSTEM,2})| \\ + \rho_{DUCT,I} g (Y_{SYSTEM,1} - Y_{SYSTEM,2})^{1/2} / |\dot{M}_{DUCT,I}|$$

where  $\Delta p_F$  is defined in Eq. (1), and  $\delta p_{SYSTEM,1}$  and  
 $\delta p_{SYSTEM,2}$  are the pressures in rooms 1 and 2,  
relative to the datum pressure, at the elevations,  
 $Y_{SYSTEM,1}$  and  $Y_{SYSTEM,2}$  [1]. [ $\text{Pa}^{1/2} \text{kg}^{-1} \text{s} =$   
 $(\text{kg}/(\text{m} \cdot \text{s}^2))^{1/2} \text{kg}^{-1} \text{s}$ ]

$T_{U,I}$  [ $T_{L,I}$ ];  $I = 1$  or  $2$

Absolute temperature of the upper [lower] layer in room  $I$ . [K]

$\dot{V}_{F,J}$ ,  $J = 1$  to  $JTERM$

Volume-flow-rate delivered by the fan when cross-fan pressure difference is at the specified value  $\Delta p_{F,J}$ . A  $\dot{V}_{F,J}$  value is positive when flow is in the normal direction of fan operation. The pairs of ( $\dot{V}_{F,J}$ ,  $\Delta p_{F,J}$ ) data would be obtained from fan operating characteristics provided typically by the fan manufacturer. The  $\Delta p_{F,J}$  are required to be monotonically decreasing with increasing  $\dot{V}_{F,J}$ . The  $\Delta p_F$  corresponding to a given volume-flow-rate through the fan,  $\dot{V}_F$ , is calculated by the FANRES algorithm/subroutine [2]. This is done by interpolating between the ( $\dot{V}_{F,J}$ ,  $\Delta p_{F,J}$ ) data points or by extrapolating beyond these for  $\dot{V}_F < \dot{V}_{F,1}$  and  $\dot{V}_F < \dot{V}_{F,JTERM}$ . If  $JTERM = 1$ , then the  $\dot{V}_{F,1}$  input is the specified volume-flow-rate, the corresponding  $\Delta p_{F,1}$  input is not required, and FANRES is not used. [ $m^3/s$ ]

$y_{FLOOR,I}$ ;  $I = 1$  or  $2$

Elevation of the floor in room  $I$  above the datum elevation. [m]

$y_{LAYER,I}$ ;  $I = 1$  or  $2$

Elevation of layer in room  $I$  above the datum elevation. [m]

$y_{SYSTEM,I}$ ;  $I = 1$  or  $2$

Elevation of the middle of the system endpoint in room  $I$  above the datum elevation. [m]

$\Delta p_{F,J}$ ,  $J = 1$  to  $JTERM$

Values of the cross-fan pressure difference at the  $JTERM$  points used to approximate the fan operating curve (volume flow-rate vs cross-fan pressure difference). See the above  $\dot{V}_{F,J}$  entry. [ $Pa = kg/(m \cdot s^2)$ ]

$\delta P_{\text{FLOOR},I}$ ;  $I = 1$  or  $2$

Pressure, above the datum pressure, at the floor in room  $I$ . [ $\text{Pa} = \text{kg}/(\text{m} \cdot \text{s}^2)$ ]

$\rho_{L,I}$  [ $\rho_{U,I}$ ];  $I = 1$  or  $2$

Density of lower [upper] layer in room  $I$ . [ $\text{kg}/\text{m}^3$ ]

### CALCULATIONS

1. If  $J\text{TERM} = 1$  and  $\dot{V}_{F,1} = 0$ , then there is no forced vent flow and no adjustments are made to the environments in rooms 1 and 2. If  $J\text{TERM} = 1$  and  $\dot{V}_{F,1}$  is non-zero, or if  $J\text{TERM} > 1$ , then continue with calculation step 2.

2. For  $I = 1$  and  $2$ :

- a. Calculate  $\delta P_{\text{SYSTEM},I}$ :

- i. if  $y_{\text{SYSTEM},I} > y_{\text{LAYER},I}$ :

$$\begin{aligned} \delta P_{\text{SYSTEM},I} = & P_{\text{FLOOR},I} - \rho_{L,I}g(y_{\text{LAYER},I} - y_{\text{FLOOR},I}) \\ & - \rho_{U,I}g(y_{\text{SYSTEM},I} - y_{\text{LAYER},I}) \quad (10) \end{aligned}$$

- ii. if  $y_{\text{SYSTEM},I} \leq y_{\text{LAYER},I}$ :

$$\begin{aligned} \delta P_{\text{SYSTEM},I} = & P_{\text{FLOOR},I} \\ & - \rho_{L,I}g(y_{\text{SYSTEM},I} - y_{\text{FLOOR},I}) \quad (11) \end{aligned}$$

- b. Compute  $h_{U,I}$  and  $h_{L,I}$  from Eq. (2).

- c. Although the flow direction is not yet known, the density in the duct near the room IFR flow entry is computed for the possible flow condition  $\text{IFR} = I$ , for  $I = 1$  or  $2$ .

- i. if  $y_{\text{SYSTEM},I} + H_I/2 \leq y_{\text{LAYER},I}$ :

the duct flow is entirely from the lower layer and  
 $\rho_{\text{DUCT},I} = \rho_{L,I}$

- ii. if  $y_{\text{SYSTEM},I} - H_I/2 \geq y_{\text{LAYER},I}$ :

the duct flow is entirely from the upper layer and  
 $\rho_{\text{DUCT},I} = \rho_{U,I}$

iii. else:

the duct flow is from both the upper and lower layers  
and  $\rho_{\text{DUCT},I}$  is computed from Eq. (7)

3. Determine  $\dot{M}_{\text{DUCT}}$ ,  $I_{\text{DIR}}$ , IFR, and ITO. If  $J\text{TERM} > 1$ , then use the FANRES algorithm. If  $J\text{TERM} = 1$ , then  $\dot{M}_{\text{DUCT}}$  is obtained from the specified volumetric flow rate,  $\dot{V}_{F,1}$ .

a. if  $J\text{TERM} = 1$ :

i. if  $\dot{V}_{F,1} > 0$ :

$$\dot{M}_{\text{DUCT}} = \rho_{\text{DUCT},1} \dot{V}_{F,1}, \quad I_{\text{DIR}} = 1, \text{ and } \text{IFR} = 1$$

ii. if  $\dot{V}_{F,1} < 0$ :

$$\dot{M}_{\text{DUCT}} = -\rho_{\text{DUCT},2} \dot{V}_{F,1}, \quad I_{\text{DIR}} = -1, \text{ and } \text{IFR} = 2$$

b. if  $J\text{TERM} > 1$  calculate:

$$\Delta y_{\text{SYSTEM}} = y_{\text{SYSTEM},1} - y_{\text{SYSTEM},2}$$

$$\Delta p_{\text{SYSTEM}} = \delta p_{\text{SYSTEM},1} - \delta p_{\text{SYSTEM},2}$$

and then:

$$\text{CALL FANRES}(\dot{M}_{\text{SOLN},N}, N_{\text{SOLN}}, \text{ACC}, J\text{TERM}, R_I, \dot{V}_{F,J}, \Delta p_{F,J},$$

$$\Delta p_{\text{SYSTEM}}, \Delta y_{\text{SYSTEM}}, \rho_{\text{DUCT},I})$$

i. if  $N_{\text{SOLN}} = 1$ , then  $\dot{M}_{\text{DUCT}} = |\dot{M}_{\text{SOLN},1}|$  and

if  $\dot{M}_{\text{SOLN},1} > 0$ :  $I_{\text{DIR}} = \text{IFR} = 1$  and  $\text{ITO} = 2$

if  $\dot{M}_{\text{SOLN},1} < 0$ :  $I_{\text{DIR}} = -1$  and  $\text{IFR} = 2$  and  $\text{ITO} = 1$

if  $\dot{M}_{\text{SOLN},1} = 0$ :  $I_{\text{DIR}} = 0$ , there is no forced vent flow, and no adjustments are made to the environments in rooms 1 and 2

ii. if  $N_{\text{SOLN}} = 2$ , then  $I_{\text{DIR}} = I_{\text{DIROLD}}$  and

if  $I_{\text{DIR}} = \dot{M}_{\text{SOLN},1}/|\dot{M}_{\text{SOLN},1}|$ , then  $\dot{M}_{\text{DUCT}} = |\dot{M}_{\text{SOLN},1}|$

if  $I_{\text{DIR}} = \dot{M}_{\text{SOLN},2}/|\dot{M}_{\text{SOLN},2}|$ , then  $\dot{M}_{\text{DUCT}} = |\dot{M}_{\text{SOLN},2}|$

else:

if  $I_{\text{DIR}} = 1$ , then  $\text{IFR} = 1$  and  $\text{ITO} = 2$

else  $\text{IFR} = 2$  and  $\text{ITO} = 1$

4. Determine the number and properties of the flow slabs leaving the room IFR and the properties of the duct flow.

- a. if  $y_{\text{SYSTEM, IFR}} + H_{\text{IFR}}/2 \leq y_{\text{LAYER, IFR}}$ , then the duct flow is entirely from the lower layer. Set  $N_{\text{SLABF}} = 1$ . Then:

$$\dot{M}_{\text{SLABF}, 1} = \dot{M}_{\text{DUCT}}$$

$$\dot{P}_{\text{SLABF}, 02, 1} = \dot{P}_{\text{DUCT}, 02} = \dot{M}_{\text{DUCT}} \cdot c_{L, 02, \text{IFR}}$$

$$\dot{P}_{\text{SLABF}, K, 1} = \dot{P}_{\text{DUCT}, K} = \dot{M}_{\text{DUCT}} \cdot c_{L, K, \text{IFR}}, K = 2 \text{ to } N_{\text{PROD}}$$

$$\dot{Q}_{\text{SLABF}, 1} = \dot{Q}_{\text{DUCT}} = \dot{M}_{\text{DUCT}} \cdot c_p T_{L, \text{IFR}}$$

$$T_{\text{SLABF}, 1} = T_{\text{DUCT}} = T_{L, \text{IFR}}$$

$$y_{\text{SLABF}, 1} = y_{\text{SYSTEM, IFR}}$$

- b. if  $y_{\text{SYSTEM, IFR}} - H_{\text{IFR}}/2 \geq y_{\text{LAYER, IFR}}$ , then the duct flow is entirely from the upper layer. Set  $N_{\text{SLABF}} = 1$ . Then:

$$\dot{M}_{\text{SLABF}, 1} = \dot{M}_{\text{DUCT}}$$

$$\dot{P}_{\text{SLABF}, 02, 1} = \dot{P}_{\text{DUCT}, 02} = \dot{M}_{\text{DUCT}} \cdot c_{U, 02, \text{IFR}}$$

$$\dot{P}_{\text{SLABF}, K, 1} = \dot{P}_{\text{DUCT}, K} = \dot{M}_{\text{DUCT}} \cdot c_{U, K, \text{IFR}}, K = 2 \text{ to } N_{\text{PROD}}$$

$$\dot{Q}_{\text{SLABF}, 1} = \dot{Q}_{\text{DUCT}} = \dot{M}_{\text{DUCT}} \cdot c_p T_{U, \text{IFR}}$$

$$T_{\text{SLABF}, 1} = T_{\text{DUCT}} = T_{U, \text{IFR}}$$

$$y_{\text{SLABF}, 1} = y_{\text{SYSTEM, IFR}}$$

- c. else the duct flow is from both the upper and lower layers. Set  $N_{\text{SLABF}} = 2$  and let slab 1 come from the upper layer and slab 2 come from the lower layer. Compute  $T_{\text{DUCT}}$  from Eq. (8);  $\dot{M}_{\text{SLABL, IFR}}$  from Eq. (9); and  $\dot{M}_{\text{SLABL, IFR}}$  from Eq. (10). Then:

$$\dot{M}_{\text{SLABF}, 1} = \dot{M}_{U, \text{IFR}}$$

$$\dot{P}_{\text{SLABF}, 02, 1} = \dot{M}_{U, \text{IFR}} \cdot c_{U, 02, \text{IFR}}$$

$$\dot{P}_{\text{SLABF}, K, 1} = \dot{M}_{U, \text{IFR}} \cdot c_{U, K, \text{IFR}}, K = 2 \text{ to } N_{\text{PROD}}$$

$$\dot{Q}_{\text{SLABF}, 1} = \dot{M}_{U, \text{IFR}} \cdot c_p T_{U, \text{IFR}}$$

$$T_{\text{SLABF}, 1} = T_{U, \text{IFR}}$$

$$y_{\text{SLABF}, 1} = y_{\text{SYSTEM, IFR}} + h_{U, \text{IFR}}/2$$

$$\dot{M}_{\text{SLABF}, 2} = \dot{M}_{L, \text{IFR}}$$



$$\begin{aligned}
\dot{P}_{SLABF,O2,2} &= \dot{M}_{L,IFR} \cdot C_{L,O2,IFR} \\
\dot{P}_{SLABF,K,1} &= \dot{M}_{L,IFR} \cdot C_{L,K,IFR}, \quad K = 2 \text{ to } N_{PROD} \\
\dot{Q}_{SLABF,2} &= \dot{M}_{L,IFR} \cdot C_p T_{L,IFR} \\
T_{SLABF,2} &= T_{L,IFR} \\
Y_{SLABF,2} &= Y_{SYSTEM,IFR} - h_{L,IFR}/2 \\
\dot{P}_{DUCT,O2} &= \dot{P}_{SLABF,O2,1} + \dot{P}_{SLABF,O2,2} \\
\dot{P}_{DUCT,K} &= \dot{P}_{SLABF,K,1} + \dot{P}_{SLABF,K,2}, \quad K = 2 \text{ to } N_{PROD} \\
\dot{Q}_{DUCT} &= \dot{Q}_{SLABF,1} + \dot{Q}_{SLABF,2} \\
T_{DUCT} &= T_{DUCT,IFR}
\end{aligned}$$

5. Determine the number and properties of the flow slabs entering room ITO.

- a. if  $Y_{SYSTEM,IFR} + H_{IFR}/2 \leq Y_{LAYER,IFR}$  or  $Y_{SYSTEM,IFR} - H_{IFR}/2 \geq Y_{LAYER,IFR}$ , then the duct flow enters entirely the upper or lower layer. Set  $N_{SLABT} =$  Then:

$$\begin{aligned}
\dot{M}_{SLABT,1} &= \dot{M}_{DUCT} \\
\dot{P}_{SLABT,O2,1} &= \dot{P}_{DUCT,O2} \\
\dot{P}_{SLABT,K,1} &= \dot{P}_{DUCT,K}, \quad K = 2 \text{ to } N_{PROD} \\
\dot{Q}_{SLABT,1} &= \dot{Q}_{DUCT} \\
T_{SLABT,1} &= T_{DUCT} \\
Y_{SLABT,1} &= Y_{SYSTEM,ITO}
\end{aligned}$$

- b. else the duct flow is split between the upper and lower layer. Set  $N_{SLABF} = 2$ . Let slab 1 enter the upper layer and slab 2 enter the lower layer. Let  $F_U$  and  $F_L$  be the fraction of the duct flow going to the upper and lower layer, respectively. Then:

$$\begin{aligned}
F_U &= h_{U,ITO}/H_{ITO} \\
F_L &= 1 - F_U \\
Y_{SLABT,1} &= Y_{SYSTEM,ITO} + h_{U,ITO}/2 \\
Y_{SLABT,2} &= Y_{SYSTEM,ITO} - h_{L,ITO}/2 \\
\dot{M}_{SLABT,1} &= F_U \cdot \dot{M}_{DUCT}
\end{aligned}$$

$$\begin{aligned} \dot{P}_{SLABT,02,1} &= F_U \cdot \dot{P}_{DUCT,02} \\ \dot{P}_{SLABT,K,1} &= F_U \cdot \dot{P}_{DUCT,K}, K = 2 \text{ to } N_{PROD} \\ \dot{Q}_{SLABT,1} &= F_U \cdot \dot{Q}_{DUCT} \\ T_{SLABT,1} &= T_{DUCT} \\ \dot{M}_{SLABT,2} &= F_U \cdot \dot{M}_{DUCT} \\ \dot{P}_{SLABT,02,2} &= F_U \cdot \dot{P}_{DUCT,02} \\ \dot{P}_{SLABT,K,2} &= F_U \cdot \dot{P}_{DUCT,K}, K = 2 \text{ to } N_{PROD} \\ \dot{Q}_{SLABT,2} &= F_U \cdot \dot{Q}_{DUCT} \\ T_{SLABT,2} &= T_{DUCT} \end{aligned}$$

### SUBROUTINES USED

FANRES [2]

### REFERENCES

- [1] Cooper, L.Y. and Forney, G.P., Section 3.4 of Consolidated Compartment Fire Model (CCFM) Computer Code Application CCFM.VENTS - Part I: Physical Basis, NISTIR 90-4342, National Institute of Standards and Technology, Gaithersburg MD, 1990.
- [2] Cooper, L.Y., Forney, G.P., and Klote, J.H., FANRES - Calculate the Flow Through a Fan/Duct Forced Ventilation System Between Two Rooms; Consolidated Compartment Fire Model (CCFM) Computer Code Application CCFM.VENTS - Part III: Algorithms and Subroutines, Cooper, L.Y., and Forney, G.P., Eds., NISTIR 90-4344, National Institute of Standards and Technology, Gaithersburg MD, 1990.

### SUBROUTINE VARIABLES

All nomenclature in the subroutine is identical to the nomenclature used above except for:

$c_{L,K,1}, c_{L,K,2}; K = 2 \text{ to } N_{PROD}$

-  $CL(K,1), CL(K,2); K = 2 \text{ to } N_{PROD}$

$c_{U,K,1}, c_{U,K,2}; K = 2 \text{ to } N_{PROD}$

-  $CU(K,1), CU(K,2); K = 2 \text{ to } N_{PROD}$

$c_{L,O2,1}, c_{L,O2,2}$	-	CL(1,1), CL(1,2)
$c_{U,O2,1}, c_{U,O2,2}$	-	CU(1,1), CU(1,2)
$C_p$	-	CP [W·s/(kg·K)]
$g$	-	G [m/s <sup>2</sup> ]
$I_{DIR}$	-	DIRS12
$\dot{M}_{DUCT}$	-	MDUCT [kg/s]
$\dot{M}_{SLABF,N}, \dot{M}_{SLABT,N}$	-	MSLBF(N), MSLBT(N) [kg/s]
$\dot{M}_{DUCT}$	-	MFLOW [kg/s]
$N_{SLABF}, N_{SLABT}$	-	NSLBF, NSLBT
$\dot{P}_{DUCT,O2}$	-	PDUCT(1) [(kg of oxygen)/s]
$\dot{P}_{DUCT,K}$	-	PDUCT(K) [(unit of product K)/s]
$\dot{P}_{SLABF,K,N}, \dot{P}_{SLABT,K,N}$	-	PSLBF(N,K), PSLBT(N,K) [(unit of product K)/s]
$\dot{P}_{SLABF,O2,N}, \dot{P}_{SLABT,O2,N}$	-	PSLBF(N,1), PSLBT(N,1) [(kg of oxygen)/s]
$\dot{Q}_{DUCT}$	-	QDUCT [W]
$\dot{Q}_{SLABF,N}, \dot{Q}_{SLABT,N}$	-	QSLBF(N), QSLBT(N) [W]
$T_{DUCT}$	-	TDUCT [K]
$T_{SLABF,N}, T_{SLABT,N}$	-	TSLBF(N), TSLBT(N) [K]
$T_{U,1}, T_{U,2}$	-	TU(1), TU(2) [K]
$T_{L,1}, T_{L,2}$	-	TL(1), TL(2) [K]
$\dot{V}_{F,J}$	-	V(J) [m <sup>3</sup> /s]
$Y_{FLOOR,1}, Y_{FLOOR,2}$	-	YFLOR(1), YFLOR(2) [m]
$Y_{LAYER,1}, Y_{LAYER,2}$	-	YLAY(1), YLAY(2) [m]
$Y_{SLABF,N}, Y_{SLABT,N}$	-	YSLBF(N), YSLBT(N) [m]
$Y_{SYSTEM,1}, Y_{SYSTEM,2}$	-	YSYS(1), YSYS(2) [m]

$\Delta P_{F,J}$	-	DP(J) [Pa = kg/(m·s <sup>2</sup> )]
$\delta P_{FLOOR,1}, \delta P_{FLOOR,2}$	-	PFLOR(1), PFLOR(2) [Pa = kg/(m·s <sup>2</sup> )]
$\delta P_{SYSTEM,1}, \delta P_{SYSTEM,2}$	-	PS(1), PS(2) [Pa = kg/(m·s <sup>2</sup> )]
$\rho_{L,1}, \rho_{L,2}$	-	DENL(1), DENL(2) [kg/m <sup>3</sup> ]
$\rho_{U,1}, \rho_{U,2}$	-	DENU(1), DENU(2) [kg/m <sup>3</sup> ]

**PREPARED BY**

Leonard Y. Cooper and Glenn P. Forney  
January 1989




Figure 1. The fire-generated environment in two rooms connected by a fan/duct system.




Figure 2. Flow entering the duct in room IFR when the layer interface there is between the top and bottom of the vent opening.




Figure 3. A sketch of a generic fan-operating curve which increases monotonically in volume-flow-rate with decreasing cross-fan pressure difference.

## SUBROUTINE

```
      SUBROUTINE VENTF(
I      ACC, FANPRS, FANVOL, JTERM, CONL, CONU,
I      CP, NPROD, MXPRD, PFLOR, RESIS, TL, TU, YFLOR,
I      YLAY, YSYS, RHOL, RHOU, Hfvnt, LSTSOL,
O      NSLBF, NSLBT, DIRS12,
O      XMSLBF, XMSLBT, XMDUCT, QSLBF, QSLBT, QDUCT,
O      PSLBF, PSLBT, PDUCT, TSLBF, TSLBT, TDUCT,
O      YSLBF, YSLBT
      &
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
C*BEG
C
C***  VENTF  GENERIC - THIS ROUTINE CALCULATES THE FLOWS (MASS, ENTHALPY
C              AND PRODUCT OF COMBUSTION) THROUGH A FAN DUCT SYSTEM.
C
C
C***  SUBROUTINE ARGUMENTS
C
C  INPUT
C  -----
C  ACC      - ACCURACY OF CALCULATION
C  FANPRS,FANVOL - ARRAYS OF PRESSURES [P], VOLUME FLOWS [M**3/S] THAT
C              DEFINE THE FAN CURVE
C  JTERM    - NUMBER OF TERMS IN THE FAN CURVE
C  CONL     - CONCENTRATION OF ALL PRODUCTS FOR LOWER LAYERS
C  CONU     - CONCENTRATION OF ALL PRODUCTS FOR UPPER LAYERS
C  CP       - SPECIFIC HEAT AT CONSTANT PRESSURE [W*S/(KG*K)]
C  NPROD    - ACTUAL NUMBER OF PRODUCTS
C  MXPRD    - MAXIMUM NUMBER OF PRODUCTS
C  PFLOR    - PRESSURE AT FLOOR [P]
C  RESIS    - RESISTANCE OF FAN SYSTEM
C  TL,TU    - TEMPERATURE OF LOWER/UPPER LAYERS [K]
C  YFLOR    - HEIGHT OF FLOORS ABOVE DATUM ELEVATION [M]
C  YLAY     - HEIGHT OF LAYERS ABOVE FLOOR [M]
C  YSYS     - HEIGHT OF FAN SYSTEM ABOVE FLOOR [M]
C  RHOL     - DENSITY OF LOWER LAYERS [KG/M**3]
C  RHOU     - DENSITY OF UPPER LAYERS [KG/M**3]
C  Hfvnt    - WIDTH OF INLET/OUTLET OF FAN SYSTEM [M]
C  LSTSOL   - LAST SOLUTION FOR THIS FAN SYSTEM; LSTSOL-1 -->
C              PREVIOUS SOLUTION WAS IN FORWARD DIRECTION, LSTSOL - 2
C              --> PREVIOUS SOLUTION WAS IN REVERSE DIRECTION
C
C  OUTPUT
C  -----
C  NSLBF,NSLBT - NUMBER OF SLABS (1 OR 2) IN INLET/OUTLET
C  DIRS12     - DIRECTION OF FLOW FOR EACH SLAB
C  XMSLBF,XMSLBT,XMDUCT - MASS FLOW FOR FROM/TO, BOTH SLABS
C  QSLBF,QSLBT,QDUCT - ENTHALPY FLOW FOR FROM/TO, BOTH SLABS
C  PSLBF,PSLBT,PDUCT - PRODUCT FLOW FOR FROM/TO, BOTH SLABS
C  TSLBF,TSLBT,TDUCT - TEMPERATURE FOR FROM/TO, BOTH SLABS [K]
C  YSLBF,YSLBT - HEIGHT OF FROM/TO SLAB
```



C\*END

```
DIMENSION FANPRS(*), FANVOL(*), CONL(MXPRD,2), CONU(MXPRD,2)
DIMENSION PFLOR(2), TL(2), TU(2)
DIMENSION YFLOR(2), YLAY(2), YSYS(2), RHOL(2), RHOU(2)
DIMENSION PS(2), HFVNT(2)
DIMENSION VFANMS(2)
DIMENSION XMSLBF(2), XMSLBT(2)
DIMENSION QSLBF(2), QSLBT(2)
DIMENSION PSLBF(2,MXPRD), PSLBT(2,MXPRD), PDUCT(MXPRD)
DIMENSION TSLBF(2), TSLBT(2), YSLBF(2), YSLBT(2)
DIMENSION YVBOT(2), YVTOP(2)
DIMENSION RESIS(2)
INTEGER DIRS12
```

C

```
DIMENSION RHODCT(2), TDCT(2), FL(2), FU(2), HL(2), HU(2)
PARAMETER (G=9.80D0)
PARAMETER (R=400.0D0/1.40D0)
```

C

```
DO 100 IROOM = 1, 2
  YVTOP(IROOM) = YSYS(IROOM) + HFVNT(IROOM)/2.0D0
  YVBOT(IROOM) = YSYS(IROOM) - HFVNT(IROOM)/2.0D0
```

C

C\*\*\* CALCULATE PS,I

C

```
      IF(YSYS(IROOM).GT.YLAY(IROOM))THEN
        PS(IROOM) = PFLOR(IROOM)
$      -RHOL(IROOM)*G*(YLAY(IROOM)-YFLOR(IROOM))
$      -RHOU(IROOM)*G*(YSYS(IROOM)-YLAY(IROOM))
      ELSE
        PS(IROOM) = PFLOR(IROOM)
$      -RHOL(IROOM)*G*(YSYS(IROOM)-YFLOR(IROOM))
      ENDIF
```

C

C\*\*\* COMPUTE RHODCT

C

```
      IF(YVTOP(IROOM).LE.YLAY(IROOM))THEN
        FL(IROOM) = 1.0D0
        FU(IROOM) = 0.0D0
      ELSE IF(YVBOT(IROOM).GE.YLAY(IROOM))THEN
        FL(IROOM) = 0.0D0
        FU(IROOM) = 1.0D0
      ELSE
        HU(IROOM) = YVTOP(IROOM) - YLAY(IROOM)
        HL(IROOM) = YLAY(IROOM) - YVBOT(IROOM)
        A = HL(IROOM)*SQRT(RHOU(IROOM))
        B = HU(IROOM)*SQRT(RHOL(IROOM))
        FL(IROOM) = A/(A+B)
        FU(IROOM) = B/(A+B)
      ENDIF
      TDCT(IROOM) = FL(IROOM)*TL(IROOM) + FU(IROOM)*TU(IROOM)
      RHODCT(IROOM) = RHOU(IROOM)*TU(IROOM)/TDCT(IROOM)
```

```

100 CONTINUE
C
C*** CALCULATE MASS GOING THROUGH FAN
C
  IF(JTERM.EQ.1)THEN
    IF(FANVOL(1).GE.0)THEN
      XMDUCT = RHODCT(1)*FANVOL(1)
    ELSE
      XMDUCT = RHODCT(2)*FANVOL(1)
    ENDIF
  ELSE
    ACC = 1.0D-8
    DPS = PS(1) - PS(2)
    DYS = YSYS(1) - YSYS(2)
    CALL FANRES(
I      ACC,FANPRS,FANVOL,JTERM,DPS,RESIS,DYS,RHODCT,
O      VFANMS,NSOLN
+      )
    IF(NSOLN.EQ.1)THEN
      XMDUCT = VFANMS(1)
    ELSE
      XMDUCT = VFANMS(LSTSOL)
      CALL MSGPRT('MULTIPLE SOLUTIONS IN FANRES',0)
    ENDIF
  ENDIF
C
C*** DETERMINE DIRECTION OF FLOW, (THE FROM ROOM)
C
  IF(XMDUCT.GT.0.0D0)THEN
    IFROM = 1
    DIRS12 = 1
  ELSE IF(XMDUCT.LT.0.0D0)THEN
    IFROM = 2
    DIRS12 = -1
  ELSE
    IFROM = 1
    DIRS12 = 0
  RETURN
  ENDIF
  ITO = 3 - IFROM
  XMDUCT = ABS(XMDUCT)
C
C*** DETERMINE THE NUMBER AND PROPERTIES OF THE FLOW SLABS LEAVING
C THE ROOM IFROM AND THE PROPERTIES OF THE DUCT FLOW
C
  IF(YVTOP(IFROM).LE.YLAY(IFROM))THEN
    NSLBF = 1
    XMSLBF(1) = XMDUCT
    QDUCT = XMDUCT*CP*TL(IFROM)
    QSLBF(1) = QDUCT
    TDUCT = TL(IFROM)
    TSLBF(1) = TDUCT

```

```

        YSLBF(1) = YSYS(IFROM)
        DO 210 IPROD = 1, NPROD
            PDUCT(IPROD) = XMDUCT*CONL(IPROD,IFROM)
            PSLBF(1,IPROD) = PDUCT(IPROD)
210    CONTINUE
        ELSE IF(YVBOT(IFROM).GE.YLAY(IFROM))THEN
            NSLBF = 1
            XMSLBF(1) = XMDUCT
            QDUCT = XMDUCT*CP*TU(IFROM)
            QSLBF(1) = QDUCT
            TDUCT = TU(IFROM)
            TSLBF(1) = TDUCT
            YSLBF(1) = YSYS(IFROM)
            DO 220 IPROD = 1, NPROD
                PDUCT(IPROD) = XMDUCT*CONU(IPROD,IFROM)
                PSLBF(1,IPROD) = PDUCT(IPROD)
220    CONTINUE
        ELSE
            NSLBF = 2
            XMLIFR = XMDUCT*FL(IFROM)
            XMUIFR = XMDUCT*FU(IFROM)
            TDUCT = TDCT(IFROM)
            XMSLBF(1) = XMUIFR
            QSLBF(1) = XMUIFR*CP*TU(IFROM)
            TSLBF(1) = TU(IFROM)
            YSLBF(1) = YLAY(IFROM) + HU(IFROM)/2.0D0
C
            XMSLBF(2) = XMLIFR
            QSLBF(2) = XMLIFR*CP*TL(IFROM)
            TSLBF(2) = TL(IFROM)
            YSLBF(2) = YLAY(IFROM) - HL(IFROM)/2.0D0
            QDUCT = QSLBF(1) + QSLBF(2)
            DO 250 IPROD = 1, NPROD
                PSLBF(1,IPROD) = XMUIFR*CONU(IPROD,IFROM)
                PSLBF(2,IPROD) = XMLIFR*CONL(IPROD,IFROM)
                PDUCT(IPROD) = PSLBF(1,IPROD) + PSLBF(2,IPROD)
250    CONTINUE

        ENDIF
C
C*** DETERMINE THE NUMBER AND PROPERTIES OF THE FLOW SLABS ENTERING
C THE ROOM ITO = 3 - IFROM AND THE PROPERTIES OF THE DUCT FLOW
C
        ITO = 3 - IFROM
        IF(YVTOP(ITO).LE.YLAY(ITO).OR.YVBOT(ITO).GE.YLAY(ITO))THEN
            NSLBT = 1
            XMSLBT(1) = XMDUCT
            QSLBT(1) = QDUCT
            TSLBT(1) = TDUCT
            YSLBT(1) = YSYS(ITO)
            DO 310 IPROD = 1, NPROD
                PSLBT(1,IPROD) = PDUCT(IPROD)

```

```

310    CONTINUE
      ELSE
        NSLBT = 2
        FFU = HU(ITO)/HFVNT(ITO)
        FFL = 1.0D0 - FFU
        YSLBT(1) = YLAY(ITO) + HU(ITO)/2.0D0
        YSLBT(2) = YLAY(ITO) - HL(ITO)/2.0D0
        XMSLBT(1) = FFU*XMDUCT
        XMSLBT(2) = FFL*XMDUCT
        QSLBT(1) = FFU*QDUCT
        QSLBT(2) = FFL*QDUCT
        TSLBT(1) = TDUCT
        TSLBT(2) = TDUCT
        DO 320 IPROD = 1, NPROD
          PSLBT(1,IPROD) = FFU*PDUCT(IPROD)
          PSLBT(2,IPROD) = FFL*PDUCT(IPROD)
320    CONTINUE
      ENDIF
      RETURN
      END

```

VENTHP - Calculation of the Flow of Mass, Enthalpy, Oxygen and Other  
Products of Combustion Through a Vertical, Constant-Width Vent in  
a Wall Segment Common to Two Rooms

DESCRIPTION

Consider an instant of time during the simulation of a multi-room compartment fire environment. This algorithm can be used to calculate characteristics of the flow of mass, enthalpy, oxygen, and other products of combustion through a constant-width vent located in a wall segment common to any two rooms of the compartment [1].

Designate the two rooms of interest as rooms 1 and 2 and refer to Figure 1 which depicts the room fire environments. It is assumed that the temperature, density, concentration of oxygen and of other products of combustion in the upper and lower layer of each room are specified. Also specified in each room are: the elevation above a datum elevation of the floor, and the smoke-layer interface; the pressure at the floor of each room above a specified datum pressure and the absolute pressure there.

In general, within the elevation interval of the common wall segment the room 1-to-room 2 pressure difference can be described by at most three, piecewise-continuous, linear functions of elevation. These will have endpoints at the bottom and top of the common wall segment and at the elevations of the interfaces which separate the upper and lower layers of the two rooms (provided these interfaces also lie in the elevation interval of the wall segment). This is depicted in Figure 2. As illustrated in this figure, it is possible for each of the pieces of the piecewise linear functions to change sign somewhere within its operable elevation range. The elevations (no more than four) and corresponding pressure differences at the endpoints of these pieces are required for the present calculation. Also required are the elevations (no more than three in addition to the above) where the pieces change sign (i.e., elevations where the room 1-to-room 2 pressure difference is zero).

It is assumed that the number and values of all of the above elevations and the corresponding absolute pressures in each room as well as the pressure differences have been determined previously and that these are available as input data (e.g., from use of COMWL1 [2]).  $N_{WELEV} \leq 7$  is the number of the unique elevations and  $y_{WELEV,N}$ ,  $P_{W1,N}$ ,  $P_{W2,N}$ , and  $\Delta P_{W1,W2,N} = P_{W1,N} - P_{W2,N}$ ,  $N = 1$  to  $N_{WELEV}$ , are the values of the elevations, increasing uniformly with  $N$ , the corresponding absolute pressures in room 1 and 2, and the pressure differences, respectively.

Consider a vent located in the common wall segment. The type of vent considered here is of constant width and has a specified top and bottom elevation,  $y_{VTOP}$  and  $y_{VBOT}$ , respectively, where  $y_{VTOP} - y_{VBOT} > 0$ . This algorithm first computes the pressures and room 1-to-room 2 pressure difference at  $y_{VTOP}$  and  $y_{VBOT}$  and forms new arrays  $y_N$ ,  $P_{1,N}$ ,  $P_{2,N}$ ,  $\Delta P_{1,2,N}$  by sorting the results into the original input arrays,  $y_{WELEV,N}$ ,  $P_{W1,N}$ ,  $P_{W2,N}$ , and  $\Delta P_{W1,W2,N}$ . All elements of the new  $y_N$  above  $y_{VTOP}$  and below  $y_{VBOT}$ , along

with corresponding  $P_{1,N}$ ,  $P_{2,N}$ , and  $\Delta P_{1,2,N}$  are then eliminated. The result are the uniquely determined elevations of the endpoints and zeros of the piecewise-continuous, linear, pressure-difference function of elevation bounded by  $y_{VTOP}$  and  $y_{VBOT}$ .  $N_{VELEV} \leq 7$ , the total number of these endpoints and zeros, are found. The result is given as the output,  $y_N$ ,  $P_{1,N}$ ,  $P_{2,N}$ ,  $\Delta P_{1,2,N}$ ,  $N = 1$  to  $N_{VELEV}$ . These are the elevations, increasing uniformly with  $N$ , and corresponding absolute pressures in room 1 and 2 and pressure differences, respectively. Note that  $y_1 = y_{VBOT}$  and  $y_{NELEV} = y_{VTOP}$ .

The gases in all layers of the two rooms are modeled as the same perfect gas. Recommended properties are those of air.

Associated with each  $y_N$ ,  $y_{N+1}$  pair is a "slab" of flow, designated as slab  $N$ , which is uniform in composition and which, by virtue of the cross-vent pressure difference, is driven between the two rooms. The number of slabs of flow passing through the vent is  $NSLAB = (N_{VELEV} - 1) \leq 7$ . Flow is from room 1 to room 2, or from room 2 to room 1 depending on whether either of  $\Delta P_{1,2,N}$  or  $\Delta P_{1,2,N+1}$  is positive or negative, respectively (one of the pair can be zero, but it is impossible for them to be of different signs). The direction of the flow is determined and identified by the value  $I_{DIR,N}$ . For flow from room 1 to room 2 or from room 2 to room 1, the value of  $I_{DIR,N}$  is 1 or -1, respectively.  $I_{DIR,N}$  is 0 if the room-to-room flow is identically zero (this can only occur if both  $\Delta P_{1,2,N}$  and  $\Delta P_{1,2,N+1}$  are identically zero).

Depicted in Figure 3 are possible distributions of cross-vent pressure for a single vent and the direction of the velocity in each of its one-to-six uniform-property flow slabs. The details of such a plot for a particular vent at a given instant of time is determined from the arrays  $y_N$ ,  $\Delta P_{1,2,N}$ , and  $I_{DIR,N}$ .

For each flow slab,  $N$ , the algorithm finds the density,  $\rho_{SLAB,N}$ , total mass flow rate,  $\dot{M}_{SLAB,N}$ , absolute temperature,  $T_{SLAB,N}$ , enthalpy flow rate,  $\dot{Q}_{SLAB,N}$ , oxygen concentration,  $c_{SLAB,O2,N}$ , oxygen mass flow rate,  $\dot{P}_{O2,SLAB,N}$ , and the concentrations and flow rates of the rest of the NPROD products of combustion being tracked by the simulation,  $c_{SLAB,K,N}$  and  $\dot{P}_{K,SLAB,N}$ , respectively,  $K = 2$  to NPROD.

$T_{SLAB,N}$  and  $\rho_{SLAB,N}$  are the properties of the slab as it enters the receiving room. These are computed with the assumption that the flow process, from stagnation in the source room to the point where the flow has decelerated in the receiving room (without significant mixing), is one of constant enthalpy. Thus,  $T_{SLAB,N}$  is the absolute temperature of the slab's source layer.

Also determined by the algorithm are the elevations above the datum,  $y_{SLAB,N}$ , of the centroids of the distributions of the flow slabs' entering horizontal-velocity, or -momentum profiles, i.e.,

$$y_{SLAB,N} = \frac{\int_{y_N}^{y_{N+1}} \rho v y dy}{\int_{y_N}^{y_{N+1}} \rho v dy} \approx \frac{\int_{y_N}^{y_{N+1}} v y dy}{\int_{y_N}^{y_{N+1}} v dy}, \quad N = 1 \text{ to } N_{SLAB} \quad (1)$$

where  $y$  is elevation above a common datum, and  $\rho$  and  $v = v(y)$  are the density and velocity of the flow as it enters the receiving room. In Eq. (1), the variation of  $\rho$  with velocity is neglected and  $\rho$  is approximated as being uniform through the slab. To the extent that the horizontal penetration of a slab into the receiving room is dependent on its horizontal momentum,  $y_{SLAB,N}$  may be preferable to other characteristic elevations as a measure of the elevation of its flow as it is projected into the receiving room by its horizontal momentum.

The present algorithm is applicable for arbitrary values of room pressures and cross-vent pressure differences, i.e., whether or not compressibility plays a significant role in the vent flow. The flow calculation used here is from [3]. This is an extension to the incompressible vent flow calculation presented, for example, in [4].

### OUTPUT

$c_{SLAB,O2,N}$ ;  $N = 1$  to  $N_{SLAB}$

Concentration of oxygen in slab  $N$  between the elevations  $y_N$  and  $y_{N+1}$ . [(kg of oxygen)/(kg of slab flow)]

$c_{SLAB,K,N}$ ;  $N = 1$  to  $N_{SLAB}$ ;  $K = 2$  to  $NPROD$

Concentration of product of combustion  $K$  in slab  $N$  between the elevations  $y_N$  and  $y_{N+1}$ . [(unit of product)/(kg of slab flow)]

$I_{DIR,N}$ ;  $N = 1$  to  $N_{SLAB}$

A measure of the direction of the room 1-to-room 2 flow in slab  $N$ , between  $y_N$  and  $y_{N+1}$ . For flow from room 1 to room 2 or from room 2 to room 1, the value of  $I_{DIR,N}$  is 1 or -1, respectively.  $I_{DIR,N}$  is 0 if the room-to-room flow is identically zero.

$\Delta p_{1,2,N}$ ;  $N = 1$  to  $N_{VELEV}$

$p_{1,N} - p_{2,N}$ , i.e., pressure in room 1 - pressure in room 2 at elevation  $y_N$ . If  $\Delta p_{1,2,N} > 0$ ,  $= 0$ , or  $< 0$  then flow through a vent at this elevation will be from room 1 to room 2, zero, or from room 2 to room 1, respectively. [ $N/m^2 = kg/(m \cdot s^2) = \text{pascal}$ ]

$\dot{M}_{SLAB,N}$ ;  $N = 1$  to  $N_{SLAB}$

Total mass flow rate in slab  $N$  between the elevations  $y_N$  and  $y_{N+1}$ . [kg/s]

$N_{SLAB}$

Number of slabs of flow,  $\leq 6$ , between the bottom and top of the vent.

$N_{VELEV}$

Number of unique elevations,  $\leq 7$ , which identify: 1) endpoints of the continuous, linear pieces of the room 1-to-room 2 pressure difference vs elevation; and/or 2) elevations of zero room 1-to-room 2 pressure difference, where all these elevations are at or between the top and bottom of the vent.

$\dot{P}_{O_2,SLAB,N}$ ;  $N = 1$  to  $N_{SLAB}$

Mass flow rate of oxygen in slab  $N$  between the elevations  $y_N$  and  $y_{N+1}$ . [(kg of oxygen)/s]

$\dot{P}_{K,SLAB,N}$ ;  $N = 1$  to  $N_{SLAB}$ ;  $K = 2$  to  $N_{PROD}$

Flow rate of product of combustion  $K$  in slab  $N$  between the elevations  $y_N$  and  $y_{N+1}$ . [(unit of product  $K$ )/s]

$P_{1,N}$  [ $P_{2,N}$ ];  $N = 1$  to  $N_{VELEV}$

Absolute hydrostatic pressure in room 1 [room 2] at elevation  $y_N$ . [Pa = kg/(m·s<sup>2</sup>)]

$\dot{Q}_{SLAB,N}$ ;  $N = 1$  to  $N_{SLAB}$

Total enthalpy flow rate in slab  $N$  based on absolute temperature between the elevations  $y_N$  and  $y_{N+1}$ . [W]

$\rho_{SLAB,N}$ ;  $N = 1$  to  $N_{SLAB}$

Density of the flow in slab  $N$  between the elevations  $y_N$  and  $y_{N+1}$  as it enters the receiving room [kg/m<sup>3</sup>].



$T_{\text{SLAB},N}$ ;  $N = 1$  to  $N_{\text{SLAB}}$

Absolute temperature of the flow in slab  $N$  between the elevations  $y_N$  and  $y_{N+1}$  as it enters the receiving room. [K]

$y_{\text{SLAB},N}$ ;  $N = 1$  to  $N_{\text{SLAB}}$ ; and  $y_{\text{SLAB},N} < y_{\text{SLAB},N+1}$

Elevations above the datum of the centroids of the distribution of slab  $N$ 's entering horizontal-velocity, or -momentum profiles as determined by Eq. (1). These are useful as a single measure of the elevation of slab  $N$  as it is projected into the receiving room. [m]

$y_N$ ;  $N = 1$  to  $N_{\text{VELEV}}$ ;  $y_N < y_{N+1}$

The elevations above the datum elevation which identify the bounds of the  $N_{\text{VELEV}}$  uniform-composition slabs of flow through a vent. [m]

#### INPUT

$A_v$

Area of the vent [ $\text{m}^2$ ].

$c_{L,K,I}$  [ $c_{U,K,I}$ ];  $K = 2$  to  $N_{\text{PROD}}$ ;  $I = 1$  or  $2$

Concentration of product of combustion  $K$  in lower [upper] layer of room  $I$ . [(unit of product  $k$ )/(kg of layer)]

$c_{L,O_2,I}$  [ $c_{U,O_2,I}$ ];  $I = 1$  or  $2$

Concentration of oxygen in lower [upper] layer of room  $I$ . [(kg of oxygen)/(kg of layer)]

$C_p$

Specific heat at constant pressure of the vent flow [ $\text{W}\cdot\text{s}/(\text{kg}\cdot\text{K})$ ] (suggest  $10^3 \text{ W}\cdot\text{s}/(\text{kg}\cdot\text{K})$  for air as default).

$\Delta P_{W1,W2,N}$ ;  $N = 1$  to  $N_{WELEV}$

$P_{W1,N} - P_{W2,N}$ , i.e., pressure in room 1 - pressure in room 2 at elevation  $y_{WELEV,N}$ . If  $\Delta P_{W1,W2,N} > 0$ , = 0, or  $< 0$  then flow through a vent at this elevation will be from room 1 to room 2, zero, or from room 2 to room 1, respectively. [ $\text{Pa} = \text{kg}/(\text{m} \cdot \text{s}^2)$ ]

$N_{WELEV}$

Number of unique elevations,  $\leq 7$ , which identify: 1) endpoints of the continuous, linear pieces of the room 1-to-room 2 pressure difference vs elevation; and/or 2) elevations of zero room 1-to-room 2 pressure difference, where all these elevations are at or between the horizontal extremes of the wall segment common to rooms 1 and 2.

$N_{PROD}$

Number of products of combustion, including oxygen, being tracked in the simulation.

$P_{W1,N}$  [ $P_{W2,N}$ ];  $N = 1$  to  $N_{WELEV}$

Absolute hydrostatic pressure in room 1 [room 2] at elevation  $y_{WELEV,N}$ . [ $\text{Pa} = \text{kg}/(\text{m} \cdot \text{s}^2)$ ]

$P_{DATUM}$

Datum absolute pressure. [ $\text{Pa} = \text{kg}/(\text{m} \cdot \text{s}^2)$ ]

$\delta P_{FLOOR,I}$ ;  $I = 1$  or  $2$

Pressure at floor in room 1 [room 2] above the datum pressure. [ $\text{Pa} = \text{kg}/(\text{m} \cdot \text{s}^2)$ ]

$T_{U,I}$  [ $T_{L,I}$ ];  $I = 1$  or  $2$

Absolute temperature of the upper [lower] layer in room I. [K]

$y_{WELEV,N}$ ;  $N = 1$  to  $N_{WELEV}$ ; and  $y_{WELEV,N} < y_{WELEV,N+1}$

Values of the  $N_{WELEV}$  elevations above the datum elevation. [m]

$y_{\text{FLOOR},I}$  [ $y_{\text{LAYER},I}$ ];  $I = 1$  or  $2$

Elevation of floor [layer] in room  $I$  above the datum elevation. [m]

$y_{\text{VTOP}}$  [ $y_{\text{VBOT}}$ ]

Elevation of the top [bottom] of the vent above the datum elevation. [m]

$\rho_{L,I}$  [ $\rho_{U,I}$ ];  $I = 1$  or  $2$

Density of lower [upper] layer in room  $I$ . [ $\text{kg}/\text{m}^3$ ]

$\epsilon_p$

Error tolerance for  $\delta p_{\text{FLOOR},I}$ . If  $p_{\text{ERROR},I}$  is the uncertainty in  $\delta p_{\text{FLOOR},I}$ ,  $I = 1$  or  $2$ , then  $p_{\text{ERROR},I}$  satisfies

$$|p_{\text{ERROR},I}| < \epsilon_p p_0 + |\delta p_{\text{FLOOR},I}| \epsilon_p$$

where  $p_0 = 1.0$  Pa. The first term is based on an absolute error tolerance and dominates the above error bound when  $|\delta p_{\text{FLOOR},I}|$  is less than 1.0 Pa. The second term is a relative error tolerance and dominates when  $|\delta p_{\text{FLOOR},I}|$  is greater than 1.0 Pa. The error tolerance,  $\epsilon_p$ , is chosen to be consistent with the tolerance specified for the pressure equation in an ODE solver that calls this routine.

#### CALCULATION

1. Place  $y_{\text{WELEV},N}$ ,  $p_{w1,N}$ ,  $p_{w2,N}$ ,  $\Delta p_{w1,w2,N}$  into  $y_N$ ,  $p_{1,N}$ ,  $p_{2,N}$ ,  $\Delta p_{1,2,N}$ , eliminating all array sets with  $y_N$  below  $y_{\text{VBOT}}$  and above  $y_{\text{VTOP}}$ .
2. Add  $y_{\text{VBOT}}$  and  $y_{\text{VTOP}}$  to  $y_N$ .
3. Using the algorithm/subroutine DELP [5], calculate the absolute pressures in the rooms 1 and 2 and their pressure differences at the elevations  $y_{\text{VBOT}}$  and  $y_{\text{VTOP}}$ . Store the results at appropriate locations in the arrays  $p_{1,N}$ ,  $p_{2,N}$ ,  $\Delta p_{1,2,N}$ .
4. Using the algorithms/subroutines HSORT [6,7] and RMVDUP [8], sort the height/pressure-pair/pressure-difference list and remove any entries with duplicate elevations.

5. Set the number of slabs,  $N_{SLAB}$ , to one less than the number of elevations in  $y_N$ .

6. Determine:

$$I_{DIR,N} = -1, 0, 1 \text{ according to whether} \quad (2)$$

$$\text{sign}[\Delta p_{1,2,N} + \Delta p_{1,2,N+1}] < 0, = 0, \text{ or } > 0, \text{ respectively.}$$

7. Determine:

$$T_{SLAB,N} = \text{absolute temperature of slab N's source layer} \quad (3)$$

$$\rho_{SLAB,N} = \frac{(\text{density of slab N's source layer}) \cdot (\text{absolute pressure at slab base in receiving room})}{(\text{absolute pressure at slab base in source room})} \quad (4)$$

$$c_{SLAB,O_2,N} = \text{concentration of oxygen in slab N's source layer} \quad (5)$$

$$c_{SLAB,K,N} = \text{concentration of product of combustion K in slab N's source layer, } K = 2 \text{ to } N_{PROD} \quad (6)$$

8. For nonzero-flow slabs, i.e., where  $\max(|\Delta p_{1,2,N}|, |\Delta p_{1,2,N+1}|) > 0$ , set:

$$\left. \begin{aligned} \rho &= \rho_{SLAB,N}; \\ y_T &= y_{N+1}; \quad y_B = y_N; \\ p_T &= |\Delta p_{1,2,N+1}|; \quad p_B = |\Delta p_{1,2,N}| \\ p_{ABST} &= \max[p_{1,N+1}, p_{2,N+1}] \\ p_{ABSB} &= \max[p_{1,N}, p_{2,N}] \end{aligned} \right\} \quad (7)$$

where the subscripts B and T refer to the bottom and top of the slab, respectively. Then define

$$\epsilon = (p_T + p_B)/(p_{ABST} + p_{ABSB}); \quad x = 1 - \epsilon \quad (8)$$

Following [3] compute the flow coefficient  $C(x)$  from

$$C(x) = 0.85 - 0.17x = 0.68 + 0.17\epsilon, \quad (9)$$

and  $w(x)$  from

$$w(x) = \begin{cases} 1 - [3/(4\gamma)]\epsilon & \text{if } 0 < \epsilon \leq 10^{-5} \\ f(x)/[2\epsilon]^{1/2} & \text{if } 1 \geq \epsilon > 10^{-5} \end{cases} \quad (10)$$

$$f(x) = \begin{cases} ([2\gamma/(\gamma-1)]x^{2/\gamma}[1-x^{(\gamma-1)/\gamma}])^{1/2} & \text{if } \epsilon < 1 - [2/(\gamma+1)]^{\gamma/(\gamma-1)} \\ \{\gamma[2/(\gamma+1)]^{(\gamma+1)/(\gamma-1)}\}^{1/2} & \text{if } \epsilon \geq 1 - [2/(\gamma+1)]^{\gamma/(\gamma-1)} \end{cases} \quad (11)$$

where  $\gamma$ , the ratio of specific heats of the vent flow gas, is taken to be that of air, 1.40.

Define  $p$  and  $p_{CUT}$  by

$$p^{1/2} = \begin{cases} 0 & \text{if } p_T^{1/2} + p_B^{1/2} = 0 \\ [p_T + (p_B p_T)^{1/2} + p_B]/[p_T^{1/2} + p_B^{1/2}] & \text{otherwise} \end{cases} \quad (12)$$

$$p_{CUT}^{1/2} = [\epsilon_p \cdot \text{MAX}(1.0 \text{ pascal}, |\delta p_{FLOOR,1}|, |\delta p_{FLOOR,2}|)]^{1/2} \quad (13)$$

Compute the numerical damping factor,  $F_{NOISE}$ , from

$$F_{NOISE} = 1.0 - \exp(-p^{1/2}/p_{CUT}^{1/2}) \quad (14)$$

a. Determine (from Eq. (17) of [3]):

$$\dot{M}_{SLAB,N} = F_{NOISE} C(x) w(x) [(8\rho)^{1/2}/3] A_v \cdot ([y_T - y_B]/[y_{VTOP} - y_{VBOT}]) \cdot p^{1/2} \quad (15)$$

Note that  $\epsilon \rightarrow 0$  corresponds to the incompressible limit and that under this limit the mass flow rate result of Eq. (15) is identical to the traditional mass flow-rate result of Eq. (33) of [4] (when, as recommended in [4], the vent flow coefficient there is taken to be 0.68).

The term,  $F_{\text{NOISE}}$ , is designed to damp out the numerical noise (error) in the calculated value for  $p$  that would otherwise be dominant in Eq. (15) when the cross-vent pressure difference terms  $p_T, p_B$  are small relative to the maximum of 1.0 pascal and the calculated floor pressures,  $\delta p_{\text{FLOOR},1}, \delta p_{\text{FLOOR},2}$ . The term,  $p_{\text{CUT}}$ , is an estimate of how small the maximum of  $(p_T, p_B)$  must be to retain a few digits of accuracy in the calculation of  $p$ . When the calculated value of  $p$  is smaller than  $p_{\text{CUT}}$ , this value and, therefore, the value of  $\dot{M}_{\text{SLAB},N}$  will likely contain noise which should be damped.  $F_{\text{NOISE}}$  is constructed to tend towards 1 when  $p_T$  or  $p_B$  are large relative to  $p_{\text{CUT}}$  and tends towards 0 when  $p_T$  and  $p_B$  are small relative to  $p_{\text{CUT}}$ .

As an example of the importance of  $F_{\text{NOISE}}$  consider  $p_B$  for a vent whose bottom is at the floor and connects two rooms with equal floor height. Then for the lowest vent flow slab  $p_B$  is given by

$$p_B = |\delta p_{\text{FLOOR},1} - \delta p_{\text{FLOOR},2}| = |\Delta p_{1,2,1}|$$

Both  $\delta p_{\text{FLOOR},1}$  and  $\delta p_{\text{FLOOR},2}$  are quantities that are typically estimated by an ODE solver to only a certain user-specified accuracy, say  $\epsilon_p = 10^{-6}$ . In this case, if the first 6 digits of  $\delta p_{\text{FLOOR},1}$  and  $\delta p_{\text{FLOOR},2}$  match then the difference,  $p_B$ , will consist entirely of random noise or error. Under the not-unreasonable worst-case scenario, the same level of error can be expected in the calculated value of  $p_T$ . The numerical errors present in  $p_B$  and  $p_T$  leads to a similar error in  $p$  which is amplified when computing  $\dot{M}_{\text{SLAB},N}$  in Eq. (15). This is amplified further by  $C_p \cdot T_{\text{SLAB},N}$  when calculating  $\dot{Q}_{\text{SLAB},N}$  per Eq. (18) below. The term  $\dot{Q}_{\text{SLAB},N}$  is a component of the differential equation for pressure. So, if the noise in  $\dot{Q}_{\text{SLAB},N}$  dominates the pressure equation the ODE solver will reduce the time step resulting in solution inefficiency or worse, a solver failure.

b. With  $\dot{M}_{\text{SLAB},N}$ , now calculate

$$\dot{p}_{\text{O}_2, \text{SLAB},N} = c_{\text{SLAB}, \text{O}_2, N} \cdot \dot{M}_{\text{SLAB},N} \quad (16)$$

$$\dot{p}_{K, \text{SLAB},N} = c_{\text{SLAB}, K, N} \cdot \dot{M}_{\text{SLAB},N}, \quad K = 2 \text{ to } N_{\text{PROD}} \quad (17)$$

$$\dot{Q}_{\text{SLAB},N} = C_p \cdot \dot{M}_{\text{SLAB},N} \cdot T_{\text{SLAB},N} \quad (18)$$

- c. From indicated integrations in Eq. (1) determine (see Section 3.3.4.5 of [1]):

$$y_{SLAB,N} = [(3/5)(p_T^{5/2} - p_B^{5/2})(y_T - y_B)/(p_T^{3/2} - p_B^{3/2}) + (p_T y_B - p_B y_T)]/(p_T - p_B) \quad (19)$$

In order to increase accuracy when calculating the above bracketed term when  $p_T$  is close to  $p_B$ , the subroutine computes  $y_{SLAB,N}$  according to:

$$y_{SLAB,N} = \begin{cases} (y_B + y_T)/2 & \text{if } |p_B - p_T| \leq 0.001 \\ [(3/5)(p_T^2 + p_T^{3/2}p_B^{1/2} + p_T p_B + p_T^{1/2}p_B^{3/2} + p_B^2)(y_T - y_B)/ \\ (p_T + p_T^{1/2}p_B^{1/2} + p_B) + (p_T y_B - p_B y_T)]/ \\ (p_T - p_B) & \text{otherwise} \end{cases} \quad (20)$$

#### SUBROUTINES USED

DELP [5]

HSORT [6,7]

RMVDUP [8]

#### REFERENCES

- [1] Cooper, L.Y. and Forney, G.P., Section 3.3 of Consolidated Compartment Fire Model (CCFM) Computer Code Application CCFM.VENTS - Part I: Physical Basis, NISTIR 90-4342, National Institute of Standards and Technology, Gaithersburg MD, 1990.

- [2] Cooper, L.Y. and Forney, G.P., COMWL1 - Setup for Calculation of the Flow Through Natural Vents in the Wall Segment Common to Two Rooms; Consolidated Compartment Fire Model (CCFM) Computer Code Application CCFM.VENTS - Part III: Algorithms and Subroutines, Cooper, L.Y., and Forney, G.P., Eds., NISTIR 90-4344, National Institute of Standards and Technology, Gaithersburg MD, 1990. 1989.
- [3] Cooper, L.Y., Calculating Flows Through Vertical Vents in Zone Fire Models, Combustion Science and Technology, Vol. 63, Nos. 1-3, pp. 43-50, 1989.
- [4] Emmons, H.W., Vent Flows, Section 1/Chapter 8 of the SFPE Handbook of Fire Protection Engineering, Society of Fire Protection Engineers, Boston MA, pp. 1-130-1-137, 1988.
- [5] Cooper, L.Y., and Forney, G.P., DELP - Calculation of the Absolute Hydrostatic Pressures at a Specified Elevation in Each of Two Adjacent Rooms and the Pressure Difference; Consolidated Compartment Fire Model (CCFM) Computer Code Application CCFM.VENTS - Part III: Algorithms and Subroutines, Cooper, L.Y., and Forney, G.P., Eds., NISTIR 90-4344, National Institute of Standards and Technology, Gaithersburg MD, 1990.
- [6] Forney, G.P., Utility Routine: HSORT; Consolidated Compartment Fire Model (CCFM) Computer Code Application CCFM.VENTS - Part III: Algorithms and Subroutines, Cooper, L.Y., and Forney, G.P., Eds, NISTIR 90-4344, National Institute of Standards and Technology, Gaithersburg MD, 1990.
- [7] Press, W.H. et al, Numerical Recipes - the Art of Scientific Computing, Cambridge Press, pp. 231-232 (1986).
- [8] Forney, G.P., Utility Routine: RMVDUP; Consolidated Compartment Fire Model (CCFM) Computer Code Application CCFM.VENTS - Part III: Algorithms and Subroutines, Cooper, L.Y., and Forney, G.P., Eds., NISTIR 90-4344, National Institute of Standards and Technology, Gaithersburg MD, 1990.

#### SUBROUTINE VARIABLES

All nomenclature in the subroutine is identical to the nomenclature used above except for:

$A_v$	-	AVENT [m <sup>2</sup> ]
$C_p$	-	CP [W·s/(kg·K)]
$C_{SLAB,O_2,N}$	-	CSLAB(1) [(kg oxygen)/(kg of layer)]
$C_{SLAB,K,N}$	-	CSLAB(K) [(unit of product K)/(kg of layer)]



$c_{L,K,1}, c_{L,K,2}$	-	CONL(K,1), CONL(K,2) [(unit of product K)/(kg of layer)]
$c_{U,K,1}, c_{U,K,2}$	-	CONU(K,1), CONU(K,2) [(unit of product K)/(kg of layer)]
$c_{L,O2,1}, c_{L,O2,2}$	-	CONL(1,1), CONL(1,2) [(kg of oxygen)/(kg of layer)]
$c_{U,O2,1}, c_{U,O2,2}$	-	CONU(1,1), CONU(1,2) [(kg of oxygen)/(kg of layer)]
$f$	-	FF [dimensionless]
$I_{DIR,N}$	-	DIRS12 [dimensionless]
$\dot{M}_{SLAB,N}$	-	XMSLAB [kg/s]
$N_{SLAB}, N_{VELEV}, N_{WELEV}$	-	NSLAB, NVELEV, NMELEV [dimensionless]
$\dot{P}_{K,SLAB,N}$	-	PSLAB(K,N) [(unit of product K)/s]
$\dot{P}_{O2,SLAB,N}$	-	PSLAB(1,N) [(kg oxygen)/s]
$P_{DATUM}$	-	PDATM [Pa = kg/(m·s <sup>2</sup> )]
$P_{W1,N}, P_{W2,N}$	-	PAB1, PAB2 [Pa = kg/(m·s <sup>2</sup> )]
$P_{1,N}, P_{2,N}$	-	PVAB1, PVAB2 [Pa = kg/(m·s <sup>2</sup> )]
$P_{ABST}, P_{ABSB}$	-	PABST, PABSB [Pa = kg/(m·s <sup>2</sup> )]
$P_B, P_T$	-	P1,P2 [Pa = kg/(m·s <sup>2</sup> )]
$\dot{Q}_{SLAB,N}$	-	QSLAB [W]
$T_{SLAB,N}$	-	TSLAB [K]
$T_{U,1}, T_{U,2}$	-	TU(1), TU(2) [K]
$T_{L,1}, T_{L,2}$	-	TL(1), TL(2) [K]
$w$	-	W [dimensionless]
$x$	-	X [dimensionless]
$y_{FLOOR,1}, y_{FLOOR,2}$	-	YFLOR(1), YFLOR(2) [m]
$y_{LAYER,1}, y_{LAYER,2}$	-	YLAY(1), YLAY(2) [m]
$y_{SLAB,N}$	-	YSLAB [m]

$Y_{VTOP}, Y_{VBOT}$	-	YVTOP, YVBOT [m]
$Y_{WELEV}$	-	YELEV [dimensionless]
$Y_B, Y_T$	-	Y1, Y2 [m]
$Y_N$	-	YVELEV [m]
$\gamma$	-	1.40
$\Delta P_{W1, W2, N}$	-	DP1M2 [Pa = kg/(m·s <sup>2</sup> )]
$\Delta P_{1, 2, N}$	-	DPV1M2 [Pa = kg/(m·s <sup>2</sup> )]
$\delta P_{FLOOR, 1}, \delta P_{FLOOR, 2}$	-	PFLOR(1), PFLOR(2) [Pa = kg/(m·s <sup>2</sup> )]
$\epsilon$	-	EPS [dimensionless]
$\epsilon_p$	-	EPSP [dimensionless]
$\rho$	-	DEN [kg/m <sup>3</sup> ]
$\rho_{L, 1}, \rho_{L, 2}$	-	DENL(1), DENL(2) [kg/m <sup>3</sup> ]
$\rho_{U, 1}, \rho_{U, 2}$	-	DENU(1), DENU(2) [kg/m <sup>3</sup> ]
$\rho_{SLAB, N}$	-	RSLAB [kg/m <sup>3</sup> ]

PREPARED BY

Leonard Y. Cooper and Glenn P. Forney  
March 1988




Figure 1. The fire-generated environment in two adjacent rooms.




Figure 2. Possible distributions of cross-wall pressure difference,  $\Delta p_{1,2}$ , for a common-wall segment.




Figure 3. Possible distributions of cross-vent pressure difference,  $\Delta p_{1,2}$ , for a single vent and flow directions of its one-to-six uniform-property flow slabs.

## SUBROUTINE

```
      SUBROUTINE VENTHP(
      I YFLOR,YLAY,
      I TU,TL,DENL,DENU,PFLOR,YVTOP,YVBOT,AVENT,YELEV,DP1M2,NELEV,
      I CP,PDATM,PAB1,PAB2,CONL,CONU,
      I NPROD,MXPRD,MXSLAB,EPSP,
      OCSLAB,PSLAB,PV1,PV2,QLAB,
      O DIRS12,DPV1M2,RSLAB,TSLAB,YSLAB,YVELEV,XMSLAB,NVELEV,NSLAB)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)

C*BEG
C***  VENTHP  GENERIC - CALCULATION OF THE FLOW OF MASS, ENTHALPY, OXYGEN
C              AND OTHER PRODUCTS OF COMBUSTION THROUGH A VERTICAL,
C              CONSTANT-WIDTH VENT IN A WALL SEGMENT COMMON TO TWO ROOMS.
C              THE SUBROUTINE USES INPUT DATA DESCRIBING THE TWO-LAYER
C              ENVIRONMENT IN EACH OF THE TWO ROOMS AND OTHER INPUT DATA
C              CALCULATED IN SUBROUTINE COMWL1.
C***  SUBROUTINE ARGUMENTS
C
C  INPUT
C  -----
C  YFLOR - HEIGHT OF FLOOR ABOVE DATUM ELEVATION [M]
C  YLAY  - HEIGHT OF LAYER ABOVE DATUM ELEVATION [M]
C  TU    - UPPER LAYER TEMPERATURE [K]
C  TL    - LOWER LAYER TEMPERATURE [K]
C  DENL  - LOWER LAYER DENSITY [KG/M**3]
C  DENU  - UPPER LAYER DENSITY [KG/M**3]
C  PFLOR - PRESSURE AT FLOOR ABOVE DATUM PRESSURE
C          [KG/(M*S**2) - PASCAL]
C  YVTOP - ELEVATION OF TOP OF VENT ABOVE DATUM ELEVATION [M]
C  YVBOT - ELEVATION OF BOTTOM OF VENT ABOVE DATUM ELEVATION [M]
C  AVENT - AREA OF THE VENT [M**2]
C  YELEV - VALUES OF THE NELEV ELEVATIONS ABOVE DATUM ELEVATION [M]
C  DP1M2 - PRESSURE IN ROOM 1 - PRESSURE IN ROOM 2 AT
C          ELEVATIONS YELEV [KG/(M*S**2) - PASCAL]
C  NELEV - NUMBER OF ELEVATIONS DELINEATING SLABS
C  CP    - SPECIFIC HEAT [W*S/(KG*K)]
C  PDATM - DATUM ABSOLUTE PRESSURE [KG/(M*S**2) - PASCAL]
C  PAB1  - ABSOLUTE PRESSURE IN ROOM 1 AT HEIGHT YELEV
C          [KG/(M*S**2) - PASCAL]
C  PAB2  - ABSOLUTE PRESSURE IN ROOM 2 AT HEIGHT YELEV
C          [KG/(M*S**2) - PASCAL]
C  CONL  - CONCENTRATION OF EACH PRODUCT IN LOWER LAYER
C          [UNIT OF PRODUCT/(KG LAYER)]
C  CONU  - CONCENTRATION OF EACH PRODUCT IN UPPER LAYER
C          [UNIT OF PRODUCT/(KG LAYER)]
C  NPROD - NUMBER OF PRODUCTS IN CURRENT SCENARIO
C  MXPRD - MAXIMUM NUMBER OF PRODUCTS CURRENTLY AVAILABLE
C  MXSLAB- MAXIMUM NUMBER OF SLABS CURRENTLY AVAILABLE
C  EPSP  - ERROR TOLERANCE FOR PRESSURES AT FLOOR
C
```

```

C  OUTPUT
C  -----
C  CSLAB  - CONCENTRATION OF OTHER PRODUCTS IN EACH SLAB
C           [UNIT PRODUCT/(KG SLAB)]
C  PSLAB  - AMOUNT OF OTHER PRODUCTS IN EACH SLAB [UNIT OF
C           PRODUCT/S]
C  PVAB1  - ABSOLUTE PRESSURE IN ROOM 1 AT ELEVATION YVELEV
C           [KG/(M*S**2) - PASCAL]
C  PVAB2  - ABSOLUTE PRESSURE IN ROOM 2 AT ELEVATION YVELEV
C           [KG/(M*S**2) - PASCAL]
C  QSLAB  - ENTHALPY FLOW RATE IN EACH SLAB [W]
C  DIRS12 - A MEASURE OF THE DIRECTION OF THE ROOM 1 TO ROOM
C           2 FLOW IN EACH SLAB
C  DPV1M2 - PRESSURE IN ROOM 1 - PRESSURE IN ROOM 2 AT
C           ELEVATIONS YVELEV [KG/(M*S**2) - PASCAL]
C  RSLAB  - DENSITY OF THE FLOW IN EACH SLAB [KG/M**3]
C  TSLAB  - ABSOLUTE TEMPERATURF OF THE FLOW IN EACH SLAB [K]
C  YSLAB  - ELEVATIONS ABOVE THE DATUM ELEVATION OF THE
C           CENTROIDS OF MOMENTUM OF EACH SLAB [M]
C  YVELEV - ELEVATIONS ABOVE THE DATUM ELEVATIONS OF VENT
C           BOUNDARIES, LAYERS, AND NEUTRAL PLANES [M]
C  XMSLAB - MAGNITUDE OF THE MASS FLOW RATE IN SLABS [KG/S]
C  NVELEV - NUMBER OF UNIQUE ELEVATIONS DELINEATING SLABS
C  NSLAB  - NUMBER OF SLABS BETWEEN BOTTOM AND TOP OF THE VENT
C
C***  ROUTINES CALLED
C           DELP,HSORT,RMVDUP
C
C
C
C*END

      DIMENSION YFLOR(*), YLAY(*), TU(*), TL(*), DENL(*), DENU(*)
      DIMENSION PAB1(*), PAB2(*), PFLOR(*)
      DIMENSION PSLAB(MXSLAB,*), PV1(*), PV2(*)
      DIMENSION CSLAB(MXSLAB,*), CONL(MXPRD,2), CONU(MXPRD,2)
      DIMENSION YELEV(*), DP1M2(*), DPV1M2(*), XMSLAB(*)
      DIMENSION RSLAB(*), TSLAB(*), YSLAB(*), YVELEV(*)
      DIMENSION QSLAB(*)
      DIMENSION VP1RT(4),VP2RT(4)
      INTEGER DIRS12(*)
      CHARACTER*132 MSSG
      PARAMETER (GAM=1.40D0)
      DATA GAMCUT,GAMMAX/.471718212,.684731456/

C
      NVELEV = 0

C
C*** COPY VALUES OF YELEV(I) INTO YV THAT LIE BETWEEN THE BOTTOM AND TOP
C OF THE VENT
C
      YMIN = MAX(YVBOT,YELEV(1))
      YMAX = MIN(YVTOP,YELEV(NELEV))
      DO 10 I = 1, NELEV

```

```

        IF(YMIN.LE.YELEV(I).AND.YELEV(I).LE.YMAX)THEN
            NVELEV = NVELEV + 1
            YVELEV(NVELEV) = YELEV(I)
            DPV1M2(NVELEV) = DP1M2(I)
        ENDIF
10 CONTINUE
C
C*** PUT YVTOP INTO YVELEV IF IT LIES BETWEEN YMIN AND YMAX
C AND CALCULATE THE PRESSURE AT YVTOP
C
        IF(YMIN.LE.YVBOT.AND.YVBOT.LE.YMAX)THEN
            NVELEV = NVELEV + 1
            YVELEV(NVELEV) = YVBOT
            CALL DELP(
I          YVELEV(NVELEV),YFLOR,YLAY,DENL,DENU,PFLOR,PDATM,
O          DPV1M2(NVELEV),PAB1(NVELEV),PAB2(NVELEV))
        ENDIF
C
C*** PUT YVBOT INTO YVELEV IF IT LIES BETWEEN YMIN AND YMAX
C AND CALCULATE THE PRESSURE AT YVBOT
C
        IF(YMIN.LE.YVTOP.AND.YVTOP.LE.YMAX)THEN
            NVELEV = NVELEV + 1
            YVELEV(NVELEV) = YVTOP
            CALL DELP(
I          YVELEV(NVELEV),YFLOR,YLAY,DENL,DENU,PFLOR,PDATM,
O          DPV1M2(NVELEV),PAB1(NVELEV),PAB2(NVELEV))
        ENDIF
C
C*** SORT THE LIST AND REMOVE DUPLICATES
C
        CALL HSORT(YVELEV,DPV1M2,PAB1,PAB2,NVELEV)
        CALL RMVDUP(YVELEV,DPV1M2,PAB1,PAB2,NVELEV)
C
C*** THE NUMBER OF SLABS IS 1 LESS THAN THE NUMBER OF HEIGHTS
C
        NSLAB = NVELEV - 1
        DO 30 N = 1, NSLAB
C
C*** DETERMINE WHETHER TEMPERATURE AND DENSITY PROPERTIES
C SHOULD COME FROM ROOM 1 OR ROOM 2
C
        PTEST = DPV1M2(N+1) + DPV1M2(N)
        IF(PTEST.GT..0)THEN
            JROOM = 1
            DIRS12(N) = 1
            PRAT = PAB2(N)/PAB1(N)
        ELSE IF(PTEST.LT.0.0)THEN
            DIRS12(N) = -1
            JROOM = 2
            PRAT = PAB1(N)/PAB2(N)
        ELSE

```



```

        DIRS12(N) = 0
        JROOM = 1
        PRAT = 0.0D0
    ENDIF
C
C*** DETERMINE WHETHER TEMPERATURE AND DENSITY PROPERTIES
C     SHOULD COME FROM UPPER OR LOWER LAYER
C
        YMID = (YVELEV(N)+YVELEV(N+1))/2.0D0
        IF(YMID.LE.YLAY(JROOM))THEN
            TSLAB(N) = TL(JROOM)
            RSLAB(N) = DENL(JROOM)*PRAT
            DO 200 IPROD = 1, NPROD
                CSLAB(N,IPROD) = CONL(IPROD,JROOM)
200          CONTINUE
            ELSE
                TSLAB(N) = TU(JROOM)
                RSLAB(N) = DENU(JROOM)*PRAT
                DO 210 IPROD = 1, NPROD
                    CSLAB(N,IPROD) = CONU(IPROD,JROOM)
210          CONTINUE
            ENDIF
            IF(RSLAB(N).LT.0.)THEN
                WRITE(MSSG,220)RSLAB(N)
220          FORMAT('IN VENTHP: RSLAB=',E11.4,'<0.')
                CALL MSGPRT(MSSG,3)
            ENDIF
C
C*** FOR NONZERO-FLOW SLABS DETERMINE XMSLAB(N) AND YSLAB(N)
C
        XMSLAB(N) = 0.0D0
        QSLAB(N) = 0.D0
        DO 11 IPROD = 1, NPROD
            PSLAB(N,IPROD) = 0.0D0
11          CONTINUE
            YSLAB(N) = YMID
            P1 = ABS(DPV1M2(N))
            P2 = ABS(DPV1M2(N+1))
            P1RT = SQRT(P1)
            P2RT = SQRT(P2)
C
C*** IF BOTH CROSS PRESSURES ARE 0 THEN THEN THERE IS NO FLOW
C
            IF(P1.GT.0.OR.P2.GT.0.)THEN
                R1 = RSLAB(N)
                Y2 = YVELEV(N+1)
                Y1 = YVELEV(N)
                PABST = PAB1(N+1)
                IF(PAB2(N+1).GT.PABST)PABST = PAB2(N+1)
                PABSB = PAB1(N)
                IF(PAB2(N).GT.PABSB)PABSB = PAB2(N)
                EPS = (P1+P2)/(PABST+PABSB)

```

```

XX = 1.0D0 - EPS
CC = .68D0+.17D0*EPS
X0 = 1.0D0
EPSCUT = EPSP*MAX(X0,PFLOR(1),PFLOR(2))
EPSCUT = SQRT(EPSCUT)
PSUM = P1RT + P2RT
IF(PSUM.EQ.0.0D0)THEN
  WP = 0.0D0
ELSE
  WP = (P1 + P1RT*P2RT + P2)/PSUM
ENDIF
ZZ = 1.0D0
IF(ABS(WP/EPSCUT).LE.130)ZZ = 1.D0 - EXP(-WP/EPSCUT)

C
C*** GAMCUT = (2./(GAM+1))*(GAM/(GAM-1.))
C GAMMAX = SQRT( GAM*(2./(GAM+1))*((GAM+1)/(GAM-1)) )
C
  IF(EPS.LE.1.E-5)THEN
    WW = 1.0D0 - .75D0*EPS/GAM
  ELSE
    IF(EPS.LT.GAMCUT)THEN
      GG = XX*(1.0D0/GAM)
      FF = SQRT( (2.0D0*GAM/(GAM-1.0D0))*GG*GG*
        §      (1.0D0-XX/GG) )
    ELSE
      FF = GAMMAX
    ENDIF
    WW = FF/SQRT(EPS+EPS)
  ENDIF
  XMSLAB(N) = CC*WW*SQRT(8*R1)*AVENT*(Y2-Y1)/(YMAX-YMIN)*
1      (P2+P1RT*P2RT+P1)/(P2RT + P1RT)
  XMSLAB(N) = XMSLAB(N)*ZZ
  XMSLAB(N) = XMSLAB(N)/3.0D0
  QSLAB(N) = CP*XMSLAB(N)*TSLAB(N)
  DO 12 IPROD = 1, NPROD
    PSLAB(N,IPROD) = CSLAB(N,IPROD)*XMSLAB(N)
12 CONTINUE

C
C*** IF PRESSURES ARE CLOSE TOGETHER THEN SET SLAB HEIGHT AT MID POINT
C
  IF(ABS(P1-P2).GT..001)THEN

C
C      A**5 - B**5      A**4 + A**3*B + A**2*B**2 + A*B**3 + B**4
C      ----- = -----
C      A**3 - B**3      A**2 + A*B + B**2
C
C
C
C      WHEN A AND B ARE "ALMOST" THE SAME THE RHS ABOVE IS A BETTER WAY
C      OF PERFORMING THE DESIRED CALCULATION SINCE IT AVOIDS CANCELLATION.
C
    VP1RT(1) = P1RT
    VP2RT(1) = P2RT

```

```

DO 100 I = 2, 4
    VP1RT(I) = VP1RT(I-1)*P1RT
    VP2RT(I) = VP2RT(I-1)*P2RT
100 CONTINUE
RNUM = VP2RT(4) + VP1RT(4)
DO 110 I = 1, 3
    RNUM = RNUM + VP1RT(I)*VP2RT(4-I)
110 CONTINUE
FACT1 = RNUM/(P1 + P1RT*P2RT + P2)
YSLAB(N) = (.6D0*FACT1*(Y2-Y1)+(P2*Y1-P1*Y2))/(P2-P1)
ELSE
    YSLAB(N) = YMID
ENDIF
ENDIF
30 CONTINUE
RETURN
END

```



## **APPENDIX B - Utility Algorithms/Subroutines**

The subroutines in this appendix are utility routines used by some of the algorithms presented in Appendix A. Their usage is explained in the comments found at the beginning of each subroutine. These routines do not calculate physical quantities but manipulate data structures used by physical routines. For example, HSORT, sorts arrays used by the subroutines COMWLL and VENTHP.



# HSORT - Arranges an Array in Ascending Order Using a Heap Sorting Algorithm

```
SUBROUTINE HSORT(RA,RB,RC,RD,N)
IMPLICIT DOUBLE PRECISION (A-H,O-Z)
C
C*BEG
C*** HSORT  GENERIC - THIS ROUTINE SORTS THE ARRAY, RA, IN ASCENDING
C              ORDER USING A HEAP SORT.  CORRESPONDING CHANGES ARE MADE
C              IN THE ARRAYS RB, RC AND RD.  THIS SUBROUTINE IS BASED ON
C              A SUBROUTINE NAMED SORT2 FOUND IN:
C
C              NUMERICAL RECIPES, THE ART OF SCIENTIFIC COMPUTING, PRESS,
C              WILLIAM H., FLANNERY, BRAIN P., TEUKOLSKY, SAUL AND
C              VETTERLING, WILLIAM T., CAMBRIDGE, PP. 231-232, 1986
C
C*END
C
  DIMENSION RA(*),RB(*),RC(*),RD(*)
  L = N/2 + 1
  IR = N
10  CONTINUE
    IF(L.GT.1)THEN
      L = L - 1
      RRA = RA(L)
      RRB = RB(L)
      RRC = RC(L)
      RRD = RD(L)
    ELSE
      RRA = RA(IR)
      RRB = RB(IR)
      RA(IR) = RA(1)
      RB(IR) = RB(1)
      RRC = RC(IR)
      RRD = RD(IR)
      RC(IR) = RC(1)
      RD(IR) = RD(1)
      IR = IR - 1
      IF(IR.EQ.1)THEN
        RA(1) = RRA
        RB(1) = RRB
        RC(1) = RRC
        RD(1) = RRD
        RETURN
      ENDIF
    ENDIF
  I = L
  J = L + L
20  IF(J.LE.IR)THEN
    IF(J.LT.IR)THEN
      IF(RA(J).LT.RA(J+1))J = J + 1
    ENDIF
```

```
      IF(RRA.LT.RA(J))THEN
        RA(I) = RA(J)
        RB(I) = RB(J)
        RC(I) = RC(J)
        RD(I) = RD(J)
        I = J
        J = J + J
      ELSE
        J = IR + 1
      ENDIF
    GO TO 20
  ENDIF
  RA(I) = RRA
  RB(I) = RRB
  RC(I) = RRC
  RD(I) = RRD
  GO TO 10
END
```



## INTERP - Interpolates A Table of Numbers

```
      SUBROUTINE INTERP(X,Y,N,T,YINT,YDERIV,ICODE)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
C*BEG
C
C***  INTERP  GENERIC - THIS ROUTINE INTERPOLATES A TABLE OF
C          NUMBERS FOUND IN THE ARRAYS, X AND Y.
C
C***  SUBROUTINE ARGUMENTS
C
C  INPUT
C  -----
C  X,Y      -  ARRAYS OF SIZE N TO BE INTERPOLATED AT X-T
C  ICODE    -  CODE TO SELECT HOW TO EXTRAPOLATE VALUES IF T
C              IS LESS THAN X(1) OR GREATER THAN X(N).
C              IF ICODE = 1 THEN YINT = Y(1) FOR T < X(1)
C                  AND YINT = Y(N) FOR T > X(N).
C              IF ICODE = 2 THEN YINT IS EVALUATED BY INTERPOLATION
C              IF X(1) < T < X(N) AND BY EXTRAPOLATION IF T < X(1)
C              OR      T > X(N)
C
C  OUTPUT
C  -----
C  YINT     -  INTERPOLATED VALUE OF THE Y ARRAY AT T
C  YDERIV   -  DERIVATIVE OF Y AT T. (SLOPE OF LINE SEGMENT
C              THAT BRACKETS T
C*END
```

```
      DIMENSION X(*), Y(*)
      SAVE
      DATA ILAST /1/
      IF(N.EQ.1)THEN
        YINT = Y(1)
        YDERIV = 0.0D0
        RETURN
      ENDIF
      IF(T.LE.X(1))THEN
        IF(ICOD .EQ. 1)THEN
          YINT = Y(1)
          YDERIV = 0.0D0
          RETURN
        ELSE
          IMID = 1
          GO TO 100
        ENDIF
      ENDIF
      IF(T.GE.X(N))THEN
        IF(ICODE.EQ.1)THEN
          YINT = Y(N)
          YDERIV = 0.0D0
```

```

        RETURN
    ELSE
        IMID = N - 1
        GO TO 100
    ENDIF
ENDIF
IF(ILAST+1.LE.N)THEN
    IMID = ILAST
    IF(X(IMID).LE.T.AND.T.LE.X(IMID+1))GO TO 100
ENDIF
IF(ILAST+2.LE.N)THEN
    IMID = ILAST + 1
    IF(X(IMID).LE.T.AND.T.LE.X(IMID+1))GO TO 100
ENDIF
IA = 1
IZ = N-1
1 CONTINUE
    IMID = (IA+IZ)/2
    IF(T.LT.X(IMID))THEN
        IZ = IMID - 1
        GO TO 1
    ENDIF
    IF(T.GE.X(IMID+1))THEN
        IA = IMID + 1
        GO TO 1
    ENDIF
100 CONTINUE
    DYDX = (Y(IMID+1)-Y(IMID))/(X(IMID+1)-X(IMID))
    YINT = Y(IMID) + DYDX*(T-X(IMID))
    YDERIV = DYDX
    ILAST = IMID
    RETURN
END

```

# RMVDUP - Remove Duplicate Entries From a List of Data

```
      SUBROUTINE RMVDUP(X,Y,Z,W,N)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
C*BEG
C***  RMVDUP  SPECIFIC - THIS ROUTINE REMOVES DUPLICATE ENTRIES FROM
C          THE SORTED DATA QUAD-RUPPLES (X(I),Y(I),Z(I),W(I) I=1 ... N)
C          THE PARAMETER N IS THEN ADJUSTED TO THE NEW NUMBER OF ENTRIES.
C          TWO ENTRIES I AND I+1 ARE DUPLICATES IF X(I)=X(I+1)
C*END

      DIMENSION X(*), Y(*), Z(*), W(*)
      II = 1
      DO 10 I = 2, N
      DIFF = X(I) - X(II)
      IF(DIFF.LT.0.)DIFF = -DIFF
      IF(DIFF.NE.0.)THEN
        II = II + 1
        X(II) = X(I)
        Y(II) = Y(I)
        Z(II) = Z(I)
        W(II) = W(I)
      ENDIF
10  CONTINUE
      N = II
      RETURN
      END
```



**RMVDUP            Remove Duplicate Entries From a List of Data**

```
SUBROUTINE RMVDUP(X,Y,Z,W,N)
C
C      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
C
C*BEG
C***  RMVDUP  SPECIFIC - THIS ROUTINE REMOVES DUPLICATE ENTRIES FROM
C              THE SORTED DATA QUAD-RUPPLES (X(I),Y(I),Z(I),W(I) I=1 ... N)
C              THE PARAMETER N IS THEN ADJUSTED TO THE NEW NUMBER OF ENTRIES.
C              TWO ENTRIES I AND I+1 ARE DUPLICATES IF X(I)=X(I+1)
C
C*END
C
C***  LOCAL/DUMMY VARIABLE DECLARATIONS
C
C      DIMENSION X(*), Y(*), Z(*), W(*)
C      II = 1
C      DO 10 I = 2, N
C      DIFF = X(I) - X(II)
C      IF(DIFF.LT.0.)DIFF = -DIFF
C      IF(DIFF.NE.0.)THEN
C          II = II + 1
C          X(II) = X(I)
C          Y(II) = Y(I)
C          Z(II) = Z(I)
C          W(II) = W(I)
C      ENDIF
C  10 CONTINUE
C      N = II
C      RETURN
C      END
```



NIST-114A (REV. 3-90)		U.S. DEPARTMENT OF COMMERCE NATIONAL INSTITUTE OF STANDARDS AND TECHNOLOGY		1. PUBLICATION OR REPORT NUMBER NISTIR 4344	
<b>BIBLIOGRAPHIC DATA SHEET</b>				2. PERFORMING ORGANIZATION REPORT NUMBER	
				3. PUBLICATION DATE July 1990	
4. TITLE AND SUBTITLE The Consolidated Compartment Fire Model (CCFM) Computer Code Application CCFM.VENTS - Part III: Catalog of Algorithms and Subroutines					
5. AUTHOR(S) Leonard Y. Cooper and Glenn P. Forney, Editors					
6. PERFORMING ORGANIZATION (IF JOINT OR OTHER THAN NIST, SEE INSTRUCTIONS) U.S. DEPARTMENT OF COMMERCE NATIONAL INSTITUTE OF STANDARDS AND TECHNOLOGY GAITHERSBURG, MD 20899				7. CONTRACT/GRANT NUMBER	
				8. TYPE OF REPORT AND PERIOD COVERED	
9. SPONSORING ORGANIZATION NAME AND COMPLETE ADDRESS (STREET, CITY, STATE, ZIP) Sponsored in part by: Naval Sea Systems Command Department of the Navy Washington, DC 20362					
10. SUPPLEMENTARY NOTES					
11. ABSTRACT (A 200-WORD OR LESS FACTUAL SUMMARY OF MOST SIGNIFICANT INFORMATION. IF DOCUMENT INCLUDES A SIGNIFICANT BIBLIOGRAPHY OR LITERATURE SURVEY, MENTION IT HERE.)  A project was carried out at The National Institute of Standards and Technology (NIST) to study the feasibility of developing a new-generation, multi-room, compartment fire model computer code, called the Consolidated Compartment Fire Model (CCFM) computer code. The idea was that such a code would consolidate past progress in zone-type compartment fire modeling, and allow readily for integration of future advances with the greatest possible flexibility. The project lead to the development of a prototype multi-room CCFM product called CCFM.VENTS.  This is Part III of a four-part report which documents CCFM.VENTS. It is a catalog of all the modular algorithms and associated computer subroutines used to simulate the physical phenomena in CCFM.VENTS. Each physical algorithm entry includes a description of the phenomenon simulated, a concise presentation of the calculation procedure used, identification of all input and output parameters, and a listing of the subroutine. The catalog entries have been developed and are presented as modular, stand-alone products. The stand-alone design feature allows the catalog entries to be used both in CCFM and in any other modular, zone-type, compartment fire model computer code.  The other three parts of this report are: Part I: Physical Basis; Part II: Software Reference Guide; and Part IV: User Reference Guide.					
12. KEY WORDS (6 TO 12 ENTRIES; ALPHABETICAL ORDER; CAPITALIZE ONLY PROPER NAMES; AND SEPARATE KEY WORDS BY SEMICOLONS) building fires; compartment fires; computer models; fire models; mathematical models; vents; zone models					
13. AVAILABILITY <input checked="" type="checkbox"/> UNLIMITED <input type="checkbox"/> FOR OFFICIAL DISTRIBUTION. DO NOT RELEASE TO NATIONAL TECHNICAL INFORMATION SERVICE (NTIS). <input type="checkbox"/> ORDER FROM SUPERINTENDENT OF DOCUMENTS, U.S. GOVERNMENT PRINTING OFFICE, WASHINGTON, DC 20402. <input checked="" type="checkbox"/> ORDER FROM NATIONAL TECHNICAL INFORMATION SERVICE (NTIS), SPRINGFIELD, VA 22161.				14. NUMBER OF PRINTED PAGES 126	
				15. PRICE A07	







